


# Simulation of ecological processes

C. I. Wit and J. Goudriaan



## Simulation Monographs

Simulation monographs is a series on  
computer simulation in agriculture and  
its supporting sciences

# Simulation of ecological processes

C.T.deWit and J.Goudriaan



Wageningen

Centre for Agricultural Publishing and Documentation

1978

Books already published in this series:

- C.T. de Wit and H. van Keulen. Simulation of transport processes in soils (2nd ed., revised). 1975. 109 pp.  
ISBN 90-220-0591-7. Price Dfl. 15.00
- J. Beek and M.J. Frissel. Simulation of nitrogen behaviour in soils. 1973. 76 pp. ISBN 90-220-0440-6. Price Dfl. 12.50
- H.G. Fransz. The functional response to prey density in an acarine system. 1974. 149 pp. ISBN 90-220-0509-7.  
Price Dfl. 20.00
- M.J. Frissel and P. Reiniger. Simulation of accumulation and leaching in soils. 1974. 124 pp. ISBN 90-220-0530-5.  
Price Dfl. 17.50
- H. van Keulen. Simulation of water use and herbage growth in arid regions. 1975. 184 pp. ISBN 90-220-0557-7.  
Price Dfl. 27.50
- G.F. Makkink and H.D.J. van Heemst. Simulation of the water balance of arable land and pastures. 1975. 87 pp.  
ISBN 90-220-0566-6. Price Dfl. 15.00
- G.W. Arnold and C.T. de Wit (Eds). Critical evaluation of systems analysis in ecosystems research and management. 1976. 114 pp.  
ISBN 90-220-0593-3. Price Dfl. 20.00
- J. van den Bos and R. Rabbinge. Simulation of the fluctuations of the grey larch bud moth. 1976. 91 pp. ISBN 90-220-0589-5.  
Price Dfl. 16.50
- R. Rabbinge. Biological control of fruit-tree red spider mite. 1976. 234 pp. ISBN 90-220-590-8. Price Dfl. 37.50
- J. Goudriaan. Crop micrometeorology: a simulation study. 1977. 257 pp. ISBN 90-220-0614-X, Price Dfl. 37.50
- E. van Elderen. Heuristic strategy for scheduling farm operations. 1977. 233 pp. ISBN 90-220-0612-3. Price Dfl. 22.50
- P. Kampmeijer and J.C. Zadoks. EPIMUL, a simulator of foci and epidemics in mixtures of resistant and susceptible plants, mosaics and multilines. 1977. 56 pages. ISBN 90-220-0636-0.  
Price Dfl. 10.00
- T. Kozai, J. Goudriaan and M. Kimura. Light transmission and photosynthesis in greenhouses. 1978. 105 pp.  
ISBN 90-220-0646-8. Price Dfl. 16.50
- K.R. Christian et al. Simulation of grazing systems. 1978. 121 pp. ISBN 90-220-0645-X. Price Dfl. 18.00

Not for sale in the U.S.A., Canada and Latin America  
2nd ed. (revised and extended)  
ISBN 90-220-0652-2

© Centre for Agricultural Publishing and Documentation,  
Wageningen, the Netherlands, 1978

Printed in the Netherlands

# Contents

1	Introduction	1
1.1	Purpose	1
1.2	Some terminology	2
1.3	Electrical analogues	3
1.4	State determined systems	4
1.5	Simulation languages	6
2	Exponential growth	8
2.1	Analytical and numerical integration	8
2.2	Simulation	10
2.3	Time constant and time interval of integration	16
2.4	Integration method and time interval	18
3	The growth of yeast	22
3.1	Description of the system	22
3.2	Relational diagrams	23
3.3	Simulation	26
3.4	Logistic growth	29
4	Interference of plants	31
4.1	Replacement series	31
4.2	Density of sowing	34
4.3	Simulation of plant interference	39
4.4	Further modelling aspects	43
4.4.1	The INDEX and MACRO feature	43
4.4.2	The INITIAL and DYNAMIC section	47
5	Growth and competition of Paramecium	49
5.1	Description of the system	49
5.2	A simulation program	51
5.3	Preliminary estimation of parameters	54
5.4	Final determination of parameters	56
5.5	Stochastic aspects	59
5.6	The programming of probability functions	61
6	Modelling of development, dispersion and diffusion	66
6.1	Introduction	66
6.2	Physiological age and development stage	68
6.3	Demographic models	70

6.3.1	Age-classes	70
6.3.2	Errors of approximation	72
6.3.3	The matrix method	76
6.4	Germination models	78
6.4.1	Boxcar train without dispersion	78
6.4.2	Boxcar train with constant relative dispersion	80
6.4.3	Boxcar train with controlled dispersion	84
6.4.4	Refinements	86
6.5	The flow of heat in soils	87
7	Growth and development of <i>Helminthosporium maydis</i>	91
7.1	Introduction	91
7.2	The weather	92
7.3	Appearance and growth of lesions	94
7.4	Sporophore or stalk formation	96
7.5	Sporulation of green stalks	99
7.6	Sporulation of dried stalks	101
7.7	Dispersal of spores	102
7.8	Germination of spores and penetration of germ tubes	106
7.9	Timing, initialization and output organization	108
7.10	Results and sensitivity analyses	110
8	Population models for fruit-tree red spider mite and predatory mites	115
8.1	Fruit-tree red spider mite	115
8.2	Predatory mite, <i>Amblyseius potentillae</i>	121
8.3	Relations between predator and prey	122
8.4	Preference as a competitive process	123
8.5	Verification	126
9	Solutions of the exercises	130
	References	167
	Appendix A	169
	Appendix B	171
	Index	173

This book is the result of work of students and research workers, interested in the simulation of ecological processes.

These are:

Agricultural University of Wageningen:

C.T. de Wit, J. Goudriaan, R. Rabbinge, G.J. van Hoof,  
C. de Jonge, Ms. J.G. Blijenburg, M.W. Sabelis, F.H. Rijsdijk

University of Nijmegen:

J.G.M. Janssen (Chapter 6)

Hebrew University, Jerusalem, Israel:

Graduate students and research workers of the workshop  
"Simulation in Ecology", organized by the late N.H. Tadmor  
(April 1973)

Connecticut Agricultural Experiment Station, New Haven, USA:

P.E. Waggoner (Chapter 7)

The second and revised edition was improved by the valuable remarks of:

T. Kozai, Chiba University, Japan

J. Putter, Volcani Center, Bet-Dagan, Israel

# 1 Introduction

## *1.1 Purpose*

'In recent years there has been extensive study of the behaviour of complex interacting systems in such fields as engineering, physiology and economics. Drawing on and building upon this diverse body of experience, progress has been made over the past ten years in the development of methods for understanding the dynamics of ecosystems and the impact of stresses upon them, including stresses generated by man. These methods are subsumed under the heading of systems ecology. Systems ecology is based on the assumption that the state of an ecosystem at any particular time can be expressed quantitatively and that changes in the system can be described by mathematical terms,' to quote the expert panel on the role of system analyses of the Man and Biosphere Program (MAB, 1972).

Whether this basic assumption of systems ecology can be made operative or not, the approach raises considerable interest among natural scientists. However, many outsiders venturing in systems ecology are confused because they are exposed to mathematical and computer techniques, and to the treatment of complex systems at a too rapid rate. This is a pity because the systems approach has its merits. However, these can only come to the fore if a dialogue can be maintained between system ecologists and their more experimentally inclined colleagues. The confusion may be reduced and the necessary dialogue stimulated by introducing the motivated ecologist stepwise to one of the main aspects of system ecology: the analyses and simulation of state determined systems. This is done in this book by treating in detail various ecological systems, ranging from simple exponential growth to plant epidemics of considerable complexity. Ecological, mathematical and programming aspects are interwoven during the treatment, and exercises have been set that are an integral part of the text on a second reading. In this way the only thing that may be new in simulation is the emphasis that is placed on the quantifying of the underlying processes, the iterative use of information and the use of suitable 'simulation languages' as a means of communication, not only between man and machine, but also between man and man.

## 1.2 Some terminology

System ecologists use and misuse many words and terms and in order not to add to the confusion of the reader it is necessary to define the common concepts: model, system and simulation. There are many models. A simple mathematical model is the age-old relation of velocity ( $v$ ) and the distance ( $s$ ) covered by a falling apple, depending on the gravitational acceleration ( $g$ ) and the time from the moment of release ( $t$ ):  $v=gt$  and  $s=0.5gt^2$ . An example of a non-mathematical model is a map. It is a simplification of reality. It contains relevant information and allows measurements. Dependent upon the purpose of the map railways, lines of equal rainfall or soil types are presented. A scale model of a designed ship enables prediction of its resistance in the water. To maintain the original relations between viscosity, density, velocity, length etc., the laws of scaling must be satisfied. Of course, the internal structure of the ship is not modelled. A system may be defined as a limited part of reality with related elements. The totality of relations within a system is called the structure of a system: both models and systems have a structure and it follows from the definitions that a model is system. The reverse does not seem true. However, it may be argued that a piece of art is a model of a conception in the artist's mind or that an engine is a model of the conception of its creator. A system  $M$  is a model of system  $O$ , provided that the structure is partly overlapping or isomorphic. Which parts of  $O$  are presented in  $M$  is determined by the requirements of relevance imposed on the model. Which part should not be considered follows also from the requirement that a model must be lucid and easy to handle.

Examples of a system are a cell, a plant, an animal, a field with a crop, a forest and a farm. It is better to choose the boundary between system and environment such that the system is isolated. Often such a boundary cannot be found and then it should be chosen so that the environment influences the system, but the system itself does affect the environment as little as possible. To achieve this goal, one must often consider a system larger than seems necessary for the purpose. If for instance, the influence of temperature on the growth of plants is studied in a climate room, this climate room is part of the environment when its construction is so good that temperature, moisture content and light intensity do not depend on the size of the plants. In most climate rooms this requirement is not met so that it may be wise to treat the room itself as part of the system. If the resulting model is unwieldy, it may be necessary to characterize the interaction between plant system and environment by continuous measurements at the interface, for

example, of the light intensity at the leaf surfaces, and of the temperature and humidity between plants. This approach erodes the generality of the model, but may enable better evaluation. A system has a pattern of behaviour which implies that the system changes with time, that it is dynamic. A simplified representation of a dynamic system is a dynamic model. An operational definition of simulation is the building of a dynamic model and the study of its behaviour. Simulation is useful if it increases one's insight of reality by extrapolation and analogy, if it leads to the design of new experiments and if the model accounts for most relevant phenomena. It should contain as few false assumptions as possible. This requirement does not seem stringent enough, but unfortunately false assumptions are almost inevitable, although more so for analytical solution techniques than for simulation techniques. Therefore with simulation our attention can be shifted from solution techniques to the study of behaviour of model and system. There has been a tendency among statisticians to restrict the term simulation to the study and modelling of stochastic processes. However, then this term cannot be used in the field of the engineering sciences and what is even more unacceptable, the common usage of the word becomes restricted.

### *1.3 Electrical analogues*

Many systems may be modelled by means of electrical analogues. For instance, a model of a falling apple might consist of an apparatus with two capacitors. The first one is charged with a current that is considered analogous to the gravitational acceleration and thus its potential is analogous to velocity. The second capacitor is charged with a current that is proportional to the potential of the first and thus its potential is analogous to the covered distance.

---

#### *Exercise 1*

Write down the differential equations that relate the rate of change of the velocity ( $v$ ) of a falling apple to the gravitational acceleration ( $g$ ) and of its distance ( $s$ ) from the point of release to the velocity. Also write down the differential equation that relates the potential ( $e$ ) of a capacitor to its capacitance ( $c$ ) and the charging current ( $i$ ). Find the expressions for the charging currents  $i_1$  and  $i_2$  of two capacitors so that the potential of the first capacitor corresponds to  $v$  and of the second capacitor to  $s$ .

---

The integration in the capacitors takes place continuously and

simultaneously, as do velocity and distance in reality. At any moment the condition of the system is fully determined by the potentials of the capacitors. The analogous computers that have been developed from this principle are very useful for simulating such continuous systems.

There are, however, several problems with their use. The user should adapt the scale of variables to be modelled to the range of potential of the circuit elements and has to accept their inaccuracies. The resulting difficulties rapidly increase with increasing size and complexity of the models and with preciser requirements of accuracy.

These problems do not show up in simulation with digital computers and it is very illustrative that the first 'languages' to simulate continuous systems on digital computers were developed to control the result of analogous machines and to facilitate the assessment of the scaling factors.

Digital machines with proper simulation languages are also preferred to analogous machines when there are discontinuous elements, and when many empirical relationships are used. A large drawback, however, is that digital machines operate sequentially and discretely, whereas in many systems, continuous processes operate in parallel.

In the future the disadvantages of both machines may be eliminated and the advantages combined by hybrid computers in which digital and analogous computers are amalgamated.

#### *1.4 State determined systems*

As has been said, systems ecology is based on the assumption that the state of an ecosystem at any particular time can be expressed quantitatively and that changes in the system can be described by mathematical terms. This assumption leads to the formulation of state determined models in which state variables, driving or forcing variables, rate variables, auxiliary variables and output variables can be distinguished.

State variables characterize and quantify all properties that describe the current state of the system. Examples of state variables are amount of biomass, number of animals, content of mineral elements in various parts of the system, amount of food, amount of poison, number of niches, water content, temperature of the soil and so on. The values of all state variables have to be known at the onset of simulation. In mathematical terms they are quantified by the contents of integrals.

Driving or forcing variables are those that are not affected by processes within the system but characterize the influence from outside. These may be macrometeorological variables, the amount of food added in course of time and so on. It should be

realized that depending on the boundaries of the system to be simulated, the same variables may be classified either as state or driving variables.

Rate variables quantify the rate of change of the state variables. Their values are determined by the state variables and the driving variables according to rules formulated from the knowledge of the underlying ecological, physiological and physical processes. These processes may be so complicated that the calculation process becomes much more lucid when use is made of properly chosen intermediate or auxiliary variables. Output variables are the quantities which the simulation produces for the user. Sometimes they are state variables, sometimes rates and sometimes auxiliary variables that may be calculated especially for the purpose.

A model of the type considered in this book consists mainly of a set of prescriptions for calculating the rate variables. At any particular time, each rate depends only on the current values of the state variables and driving variables, and can therefore be calculated independently of all other rates. The only deviations from this rule that are allowed are merely labour-saving devices; thus, when the nature of the model implies that a rate  $R_2$  happens to be equal to a function of the rate  $R_1$ , the prescription for calculating  $R_2$  may be written in terms of  $R_1$ . However, strict care must be taken to avoid any formulation in which two or more rates are mutually dependent, leading to several equations with several unknowns.

This is in accordance with experience. In a mixture of yeasts, the rates of growth do not depend on each other, but each one separately upon the state of the system, characterized by its own amount, the concentration of food and waste products and forcing variables such as temperature. The interaction between the growth of the yeasts in the mixture evolves in time because of the consumption of the same food source or production of the same waste products. Another illustration is a chemical reaction where compound C is formed from compounds A and B, the rate of formation of C not depending on the rate of formation of A and B but only on the amounts or concentrations of compound A, B and C that are present and the reaction constants. If, however, the rate of formation of A is slow, this compound is depleted in due course to such a level that the rate of formation of C adjusts to that of A. In other words, the observation that the rate of formation of A and C is equal is a consequence of the operation of the system and does not reflect a direct relation between both rates, that is to be modelled. Another illustration is presented in the form of an exercise.

---

### Exercise 2

Ask two children, who are not allowed to communicate with each other, to stand on one side of a room and tell them to walk to the other side of the room at the same speed, carrying out independently of each other the following instructions on command:

1: close eyes, 2: decide on step size, 3: take a step, 4: open eyes, 5: compare positions, 1: close eyes, 2: decide on step size, and so on. Mark the position of each child on the floor after each cycle of instructions. Do the children stay practically side by side? What are the variables of state? What are their 'rates' of change? In how many independent decision processes are these decided upon? It may be observed that the subsequent step sizes of each child are more variable in the beginning than at the end. Why is this so?

---

### 1.5 Simulation languages

Although digital computers may memorize data easily, these sequential and digital instruments seem most unsuitable for the simulation of continuous and parallel dynamic systems. The main feature of simulation languages, or simulation programming systems, is to overcome these disadvantages.

The principle that rates of change are not mutually dependent but depend independently on state variables and driving variables, allows all rates of change to be calculated in any order at any instant of time. After all rates are calculated at one instant, they are used to integrate the state variables over a small time-interval. In this way, the model operates in semi-parallel fashion. The simplest way of integration is by the Eulerian or rectilinear method in which the new value of a state variable at time  $t+\Delta t$  equals the old value at time  $t$  plus the calculated rate of change at time  $t$  times the constant time-interval  $\Delta t$ .

Another feature of simulation programming systems is that all processes and processing details may be presented in conceptual rather than computational order. The programming system itself contains a sorting routine which orders all calculations and integrations in a proper algorithm. The advantages of this procedure are that the simulation program may be presented much more clearly, that a considerable variety of programming and conceptual errors may be detected by the programming system and that sub-models are easily assembled in a larger model. Apart from this, all simulation languages contain subroutines that execute operations that are often dealt with in modelling and facilitate the organization of data input and output.

In this way programming systems have been developed that enable sophisticated use of computers by research workers without much training in advanced programming techniques and with a minimum of formal mathematical knowledge. As has been said, these programming systems are not only intended to improve the communication between man and computer but also between research workers themselves. Since many simulation programs that appear at present can be used only by the designer himself or by very motivated programmers, this latter aspect of simulation languages is very important and needs to be developed still more. However, during the last 15 years so many different systems of simulation programming have become available that communication between scientists has not improved. But recently all languages seem to be gravitating towards one concept: the Continuous System Simulation Language (CSSL), originally defined by a working party of the Simulation Council (Brennan & Silberberg, 1968). The most widely used CSSL version is at present CSMP III, the Continuous System Modelling Program III (IBM). This language as defined in IBM Users Manual SH19-7001-2 is used in the subsequent text. Array handling has become much simpler in CSMP III than in the older version of CSMP, but still some knowledge of FORTRAN is required. Therefore we use a preprocessor for array handling.

In Appendix A a short user's guide is given for the use of the preprocessor. This preprocessor was developed by the computer centre of the Agricultural University in Wageningen and is available on request. The text of this book is self-explanatory and written to be read without the use of a manual, but for actual programming it is advisable to have a manual at hand.

## 2 Exponential growth

### 2.1 Analytical and numerical integration

The growth rate of many populations may be proportional to the size of the population, either expressed in number of individuals or total biomass. With simple organisms such as bacteria, growth is often also continuous. The growth rate at any moment can then be expressed by the equation:

$$GR = RGR \times A \quad (2.1)$$

in which RGR is the relative growth rate and A the amount of organisms.

---

#### Exercise 3

If weight is expressed in grams and time in hours, what are then the dimensions of A, RGR and GR? Give at least three environmental conditions that must always be satisfied to achieve a situation where the relative growth rate is independent of the amount of organisms and time.

---

In differential notation, Eqn (2.1) is written as

$$dA/dT = RGR \times A \quad (2.2)$$

The integrated form of this equation or the analytical solution when RGR is constant, is

$$A = IA \times e^{RGR \times T} \quad (2.3)$$

in which IA is the amount of organisms that appear to be present at time zero and e the base of the natural logarithm. Under these circumstances the amount of organisms increases exponentially with time.

---

#### Exercise 4

Calculate with a slide rule, with tabulated values of the function  $e^x$  or with a pocket calculator, the value of A after 0, 2, 4 up to 10 hours for RGR equal to  $0.1 \text{ hour}^{-1}$  and IA equal

to 1 gram. Represent the results by a graph with time along the horizontal axis and the amount A along the vertical axis and connect the points with a smooth line. Plot the results also on a graph with time along the horizontal axis and connect the points also by a line.

What do you observe about the straightness of the second line? Show that this observation is mathematically correct.

---

The amount of organisms as a function of time may be found also by a recursive process. If, at a certain time T the amount of organisms equals A, the rate of growth at that moment equals  $RGR \times A$ . During a short time-interval  $\Delta t$  (DELTA), this rate of growth hardly changes, so that at time  $T + \Delta t$  the amount of organisms equals approximately  $A + RGR \times A \times \Delta t$ . With this new value, the rate of growth at time  $T + \Delta t$  can be calculated and so the amount of organisms at time  $T + 2 \times \Delta t$ , and so on.

---

#### Exercise 5

Calculate the values of A after 0, 2, 4 up to 10 hours for  $RGR = 0.1 \text{ hour}^{-1}$  and  $A = 1$  gram at time zero. Use time-intervals of 2 hours and apply the following scheme:

TIME	A	RGR x A	RGR x A x DELT
0	1	0.1	0.2
2	1.2	etc	

Plot the results on the graphs of Exercise 4 and connect the points by straight line segments.

---

A comparison of this stepwise solution and the analytical solution shows that the size of the population is underestimated by the use of the recursive solution because it is wrongly assumed that the growth rate remains the same during each time-interval DELTA, even though for continuous growth, the amount of organisms increases. It is to be expected that the discrepancy between the recursive and analytical solution decreases with decreasing time-intervals.

---

#### Exercise 6

Plot the results of a recursive calculation for time steps of 1 and 0.5 hours on the graphs of Exercise 4. Derive a formula that gives the value of A directly after n time-intervals of size DELTA and convert this function of n to a function of time. What can you say about this function when the time-interval

DELT approaches zero?

---

## 2.2 Simulation

Calculations with even shorter time-intervals are very tedious and are better done by formulating the problem in CSMP and using a computer. In its elementary form, this simulation is the same as numerical integration of a set of differential equations.

The problem in CSMP reads as follows on punched cards:

```
TITLE EXPONENTIAL GROWTH
      A=INTGRL(IA,GR)
      GR=RGR*A
INCON IA=1.
PARAMETER RGR=0.1
TIMER FINTIM=10.,OUTDEL=0.5,DELT=0.1
OUTPUT A
METHOD RECT
END
STOP
ENDJOB
```

The first card mentions the TITLE, which is repeated on every page of output. The card with the INTGRL function states that A equals IA at time zero and that its current value at any time is found by integrating GR. The fourth and fifth card give the value of the only INITIAL CONSTANT (IA) and of the only PARAMETER (RGR). The TIMER card ensures that the simulation is finished after 10 time-units (FINTIM), that output is given at every 0.5 time-unit (OUTDEL) and that intervals of 0.1 unit (DELT) are used for the numerical integration.

---

### Exercise 7

Which variable governs the unit of time?

---

It is stated on the OUTPUT card that the value of A has to be plotted against time and that its numerical value has to be given also in a table. The METHOD card indicates that the integration must be done according to the RECTilinear method of Euler. This is the method that was used in the previous exercises. The END card defines the end of the simulation model and the STOP card the end of the simulation program. If the computation is to be repeated with a relative growth rate of 0.2, it suffices to insert the cards PARAMETER RGR = 0.2 and

END between the END and STOP cards in the above program.  
The ENDJOB card is needed as the final card of the CSMP program.

---

### Exercise 8

Punch the lines of the program on cards and urge your computer centre to install one of the CSSL-type languages and to inform you on the deviation in notation between this language and the CSMP version used in this example. Then carry out the program.

---

Some readers may not have access to a suitable computer centre so that a facsimile of the program and results are given in Fig. 1. The first page contains the program, and some additional information, the third page the FORTRAN subroutine 'UPDATE' created by CSMP and the fourth page the answers. The third page is mainly of interest for those readers that know some FORTRAN and wish to follow the way CSMP organizes the work. They may have also some use for several pages with information on the organization of the program that are not reproduced here.

---

### Exercise 9

Plot the results also on the graph of Exercise 4 and compare the numerical results with those of the analytical solution. Explain why the simulated results are still underestimates.

---

For simple exponential growth where the relative growth rate

```
1 $$$CONTINUOUS SYSTEM MODELING PROGRAM III  VIM3  TRANSLATOR OUTPUT$$$
```

```
TITLE EXPONENTIAL GROWTH
      A=INTGPL(IA,GR)
      GR=RGR*A
INCON IA=1,
PARAMETER RGR=0.1
TIMER FINTIM=10.,OUTDEL=0.5,DELT=0.1
OUTPUT A
METHOD RECT
END
STOP
```

```
***  OUTPUT VARIABLE SEQUENCE  ***
GR      A
```

\$\$\$ TRANSLATION TABLE CONTENTS \$\$\$	CURRENT	MAXIMUM
MACRO AND STATEMENT OUTPUTS	8	600
STATEMENT INPUT WORK AREA	35	1900
INTEGRATORS+MEMORY BLOCK. OUTPUTS	1 + 0	300
PARAMETERS+FUNCTION GENERATORS	5 + 0	400
STORAGE VARIABLES+INTEGRATOR ARRAYS	0 + 0/2	50
HISTORY AND MEMORY BLOCK NAMES	21	50
MACRO DEFINITIONS AND NESTED MACPOS	6	50
MACRO STATEMENT STORAGE	13	125

```

LITERAL CONSTANT STORAGE      0      100
SORT SECTIONS                  1      20
MAXIMUM STATEMENTS IN SECTION  2      600

```

\$\$\$END OF TRANSLATOR OUTPUT\$\$\$

MAIN. UPDATE.F10 FORTRAN V.5(515) /KI 4-JAN-78 17:04 PAGE 1

```

00001
00002
00003
00004
00005
00006
00007      SUBROUTINE UPDATE                                CSMPDECK
00008 C      *****                                CSMPDECK
00009 C      *****                                CSMPDECK
00010      COMMON      TIME                                CSMPDECK
00011      1,ZZ0000      ,DELT      ,ZZDELT      ,DELMIN      ,ZZDELN      ,FINTIM      CSMPDECK
00012      1,ZZFINT      ,PRDEL      ,ZZPRDE      ,OUTDEL      ,ZZOUTD      ,DELMAX      CSMPDECK
00013      1,ZZDELX      ,A      ,GR      ,IA      ,RGR      CSMPDECK
00014      COMMON/ZZHIST/KEEP,NALARM,IZ0000,IZ0001      CSMPDECK
00015      REAL      IA      CSMPDECK
00016      REAL*8      ZZTIME      CSMPDECK
00017      EQUIVALENCE(ZZTIME,TIME )      CSMPDECK
00018      GO TO(39995,39996,39997,39998),IZ0000      CSMPDECK
00019 C      SYSTEM SEGMENT OF MODEL                                CSMPDECK
00020 39995 CONTINUE                                CSMPDECK
00021      GO TO 39999                                CSMPDECK
00022 C      INITIAL SEGMENT OF MODEL                                CSMPDECK
00023 39996 CONTINUE                                CSMPDECK
00024      GO TO 39999                                CSMPDECK
00025 C      DYNAMIC SEGMENT OF MODEL                                CSMPDECK
00026 39997 CONTINUE                                CSMPDECK
00027      GR=RGR*A      CSMPDECK
00028 C      A      =INTCRL      (IA      ,GR      )      CSMPDECK
00029      GO TO 39999                                CSMPDECK
00030 C      TERMINAL SEGMENT OF MODEL                                CSMPDECK
00031 39998 CONTINUE                                CSMPDECK
00032 39999 CONTINUE                                CSMPDECK
00033      RETURN      CSMPDECK
00034      END      CSMPDECK

```




Fig. 1 | A simulation program for exponential growth written in CSMP. Page 1 contains the punched program and the other pages are generated by the system. Page 2 contains the translation table contents. Page 3 the FORTRAN subroutine 'UPDATE' and page 4 the output.

is a constant, numerical integration or simulation is not necessary because the solution may be found analytically. However, the analytical approach is frustrated by only slight variations in the system.

For instance, the relative growth rate of a bacterial population may depend on temperature, so that in a series of experiments with a bacteria species the following observations of the relative growth rate could have been made:

---

TEMP (°C)	0	5	10	20	30	40	50
RGR (h <sup>-1</sup> )	0.0	0.04	0.07	0.17	0.19	0.26	0.25

---

### Exercise 10

These are (faked) observational data. Represent the results on a graph and draw a smooth line through the data points.

What are your estimates for the best fit at ten-degree-intervals?

---

In many situations, bacteria populations are not exposed to a constant temperature; the temperature varies more or less in daily cycles and the question may be posed what the growth rate is under such conditions. Obviously, the relative growth is then not a constant which may be defined on a parameter card but a variable which is some function of temperature. To simulate this situation the PARAMETER card which defines the relative growth rate is removed from the program of Fig. 1 and replaced by the following function statement:

```
RGR = AFGEN(RGRTB,TEMP)
```

This Arbitrary Function GENERator states that the value of RGR depends on the temperature (TEMP), according to a tabulated function with the table name RGRTB. Here estimates given in the answer of Exercise 10 are used. This function is introduced into the simulation program in tabulated form on a FUNCTION card:

```
FUNCTION RGRTB=(0.,0.), (10.,0.08), ...  
(20.,0.16), (30.,0.21), (40.,0.24), ...  
(50.,0.25)
```

The first number between each pair of brackets presents a value of the independent variable (TEMP) and the second one the corresponding value of the dependent variable (RGR); the three dots at the end of the first line indicate that the table is continued on the next line. The AFGEN function finds the value of the RGR at the current temperature by linear interpolation between the tabulated values: i.e. if TEMP equals 25°,

then RGR equals  $.16 + (5/10)(.21 - .16) = 0.185$

---

### Exercise 11

Enter the tabulated data of FUNCTION RGRTB on the graph of Exercise 10 and join these data points by straight lines. This broken line now represents the relation between RGR and TEMP as introduced in the simulation model. Try to match your smoothed curve more satisfactorily by tabulating values for RGR at 2.5 °C intervals. This does not seem worthwhile. Why not?

---

The next step is to define how the forcing variable TEMPERATURE varies with TIME. This may also be done with a function generator:

```
TEMP = AFGEN (TMPTB, TIME)
FUNCTION TMPTB = .....
```

These tabulated functions of forcing variables tend to be very long because they have to cover the whole simulated time-span in sufficient detail. Often it suffices to present the experimental data by some mathematical function. For instance, if there is a daily temperature variation, a sinusoidal function may be used:

```
TEMP = AVTMP + AMPTMP*SIN(6.2832*TIME/24.)
```

The function SIN calculates the sine value of the variable in the argument: 6.2832 stands for  $2 \times \pi$ , TIME is the simulated time in hours since the start of the simulation and 24 stands for the hours in a day. The average temperature and the amplitude of the temperature are given by

```
PARAMETER AVTMP = 20., AMPTMP = 10.
```

---

### Exercise 12

Prepare a graph that shows the course of temperature during 24 hours. At what time is the temperature at its maximum?

---

The variable TIME is always needed in dynamic models and the simulation language automatically keeps track of it.

---

### Exercise 13

Reason that TIME could also be kept track of by the statement:

```
T = INTGRL(0., 1.)
```

---

A facsimile of the program and the output is given in Fig.2. Note that the variables RGR and GR are entered on the PRTPLT card between brackets. Only the printed output of these variables is requested, not a graphical display. Note also that the 'UPDATE' contains all equations in computational order, whereas in the program itself they are presented in some conceptional order. It is obvious that the readability of such simple simulation programs depends little on the sequence of the equations.

1 \$\$\$CONTINUOUS SYSTEM MODELING PROGRAM III VIM3 TRANSLATOR OUTPUT\$\$\$

TITLE RELATIVE GROWTH RATE TEMPERATURE DEPENDENT

A=INTGRL(IA,GR)

GR=RGR\*A

RGR=AFGEN(RGRTB,TEMP)

TEMP=AVTMP + AMPTMP\*SIN(6.2832\*TIME/24.)

INCON IA=1.

FUNCTION RGRTB=(0.,0.),(10.,0.08),(20.,0.16),(30.,0.21),...

(40.,0.24),(50.,0.25)

PARAMETER AVTMP=20.,AMPTMP=10.

TIMER FINTIM=48.,OUTDEL=1.,DELT=0.5

OUTPUT A,RGR,GR

METHOD RECT

END

STOP

\*\*\* OUTPUT VARIABLE SEQUENCE \*\*\*  
TEMP RGR GR A

MAIN. UPDATE.F10 FORTRAN V.5(515) /KI 4-JAN-78 17:25 PAGE 1

```

00001
00002
00003
00004
00005
00006
00007 SUBROUTINE UPDATE
00008 C *****
00009 C
00010 COMMON TIME
00011 1,ZZ0000 ,DELT ,ZZDELT ,DELMIN ,ZZDELN ,FINTIM
00012 1,ZZFINT ,PRDEL ,ZZPRDE ,OUTDEL ,ZZOUTD ,DELMAX
00013 1,ZZDELX ,A ,GR ,IA ,AMPTMP ,AVTMP
00014 1,RGRTB ,RGR ,TEMP
00015 COMMON/ZZHIST/KEEP,NALARM,IZ0000,IZ0001
00016 REAL IA
00017 REAL*8 ZZTIME
00018 EQUIVALENCE(ZZTIME,TIME )
00019 GO TO(39995,39996,39997,39998),IZ0000
00020 C SYSTEM SEGMENT OF MODEL
00021 39995 CONTINUE
00022 GO TO 39999
00023 C INITIAL SEGMENT OF MODEL
00024 39996 CONTINUE
00025 GO TO 39999
00026 C DYNAMIC SEGMENT OF MODEL
00027 39997 CONTINUE
00028 TEMP=AVTMP+AMPTMP*SIN(6.2832*TIME/24.)
00029 RGR=AFGEN( 1, RGRTB,TEMP)
00030 GR=RGR*A
00031 C A =INTGRL (IA ,GR )
00032 GO TO 39999
00033 C TERMINAL SEGMENT OF MODEL
00034 39998 CONTINUE
00035 39999 CONTINUE
00036 RETURN
00037 END

```




Fig. 2 | A simulation program for exponential growth with a temperature dependent relative growth rate.

### Exercise 14

Make reruns with a two times larger and two times smaller value of DELT than used in the program. Is DELT=0.5 hour a reasonable choice? Which values have to be entered on the parameter card to obtain results that are the same as those of the program in Fig. 1?

### 2.3 Time constant and time interval of integration

When the reaction time of a car driver is too long, accidents will certainly occur; The reason is that a system must be regulated with a reaction time shorter than the period in which the system can change to a certain extent. This period of time, which is characteristic for the behaviour of a system, is called

its time constant. For example, the time constant of a human population is about ten years, of the stock of a bakery it is a few hours. In a simulation model one does not speak about reaction time, but about time-interval of integration. However, because of the feedbacks that are always present in a model, time interval of integration is equivalent to reaction time. Therefore, to obtain reliable results, a system must be simulated with a time-interval of integration less than its time constant. How much less still depends on the type of integration method.

The program, given in Fig. 1, is used to give a more precise indication of the time constant of a system. The only system parameter containing the dimension time, is the relative growth rate RGR. Therefore the time constant must be related to this parameter. Since RGR has the unit  $\text{time}^{-1}$ , an obvious definition of the time constant is the inversion of the relative growth rate. The growth rate GR is thus defined by  $A/\text{TAU}$  where TAU is the time constant and equal to  $\text{RGR}^{-1}$ . In more complicated systems each integral has its own time constant, but the time-interval of integration is limited by the smallest time constant that occurs in the system. A basic method to find the time constant of each integral in a complicated system is to reduce the integral equations to the following expressions:

$$A_1 = \text{INTGRL}(IA_1, A_1/\text{TAU}_1 + \text{all other influences})$$

$$A_2 = \text{INTGRL}(IA_2, A_2/\text{TAU}_2 + \text{all other influences})$$

It is sometimes useful to have this scheme at hand, since even in complicated systems one feedback loop with a small time constant may be clearly distinct from the rest of the system. Negative and positive feedbacks can be distinguished by the sign of TAU. In exponential growth, TAU is positive and the feedback is positive. In exponential decay, TAU will be negative. The sign of TAU is immaterial for the time-interval of integration, so that only its absolute value need be used. There is still the ratio of the time-interval of integration and the time constant of the system to be considered. In Exercises 5 and 6 the time constant of the system was 10 hours, and time intervals of 2.1 and 0.5 h were tried. The exact solution at time 10 is also known.

---

#### Exercise 15

Construct a graph with A at time 10 versus DELT. Use the exact solution for DELT=0.

---

It is clear from the graph of the exercise that the extrapolated

numerical solution for DELT=0 coincides with the analytical solution. This graph provides a method to obtain an accurate solution, but instead an equation giving an estimate for the relative error as function of DELT would be more useful. As will be shown, an acceptable time interval can then be derived from the expression for the relative error.

#### 2.4 Integration method and time interval

According to the rectilinear integration method, the value of integral A after one time interval equals  $A_t + A_t * RGR * DELT$  (in exponential growth), so that the rate at time  $t + DELT$  is given by  $RGR * (A_t + A_t * RGR * DELT)$ .

A better estimate for A at time  $t + DELT$  can be obtained by averaging the rate at time t and at time  $t + DELT$ . The average rate during the time interval is then  $RGR * (A_t + A_t * RGR * DELT / 2)$ , so that the corrected estimate for  $A_{t+DELTA}$  becomes  $A_t * (1 + RGR * DELT + \frac{1}{2} * RGR^2 * DELT^2)$ . This integration procedure is known as the trapezoidal integration method and can be invoked in a CSMP program by inserting the card METHOD TRAPZ instead of METHOD RECT. The improvement of the result of this method with respect to the rectilinear method is given by the last term  $\frac{1}{2} * RGR^2 * DELT^2 * A_t$ . Of course, even this result is not exact, but it is a safe assumption that the deviation from the 'true' answer is of the same order as the correction of the last term. Therefore the relative error that is made when METHOD RECT is used, is about  $\frac{1}{2} * RGR^2 * DELT^2 * A_t / A_t = \frac{1}{2} * RGR^2 * DELT^2$  for each time-interval of integration. Every integration step this relative error is added to the present one, so that after n integration steps the total relative error amounts about  $n * \frac{1}{2} * RGR^2 * DELT^2$ , or  $\frac{1}{2} * RGR^2 * DELT * TIME$ , when n is equal to  $TIME / DELT$ .

---

#### Exercise 16

What is, according to this equation, the largest acceptable time-interval in Exercise 15 when a relative error of 5 percent at the end of the simulation run is acceptable? Use the graph, prepared in Exercise 15 and compare the results.

---

The expression for the relative error of METHOD RECT is found by a comparison with the result given by METHOD TRAPZ. The trapezoidal integration method gives a result that is one order more accurate than the rectilinear one. Similarly, the relative error in TRAPZ can be estimated by comparison with the next better integration method, and so on. The error of TRAPZ turns out as  $1/6 * RGR^3 * DELT^3 * A_t$ , so that the total relative

error after  $n$  intervals of integration is given by  $1/6 * RGR^3 * DELT^2 * TIME$  when  $n$  is equal to  $TIME/DELT$ . Hence the relative error is proportional to  $DELT^2$ . An instructive exercise is now to repeat Exercise 15 with METHOD TRAPZ and time-intervals of 1, 2, 5 and 10 hours.

The principle of comparison of the result of two integration methods is incorporated in METHOD RKS which chooses its own time interval of integration. When the results of two integration methods, those of Runge-Kutta and Simpson, deviate too much, the time-interval DELT is halved. If the error is much smaller than required, DELT is doubled for the next step. In this way the size of the time-interval is automatically adapted to the time constant of the system.

Sometimes the error criterion is not met by decreasing the time-interval. Then the error message 'DELT IS LESS THAN DELMIN' is produced, and the simulation is automatically terminated. Then it is advisable to look for a programming or conceptual error. Because of the feature of automatic adaption of the time-interval of integration, the method RKS is recommended as a standard method.

In the RKS method the statements of the computer programs are executed many times, only to obtain a preliminary estimation of the rates. How many times this execution is done can be checked by introduction of some counters into the program. To this end an initial segment is introduced by the card INITIAL. In the initial segment the counters COUNT1 and COUNT2 are set to zero. Then the part of the simulation program, that must be evaluated each time interval, is opened with the card DYNAMIC. In the last part of the DYNAMIC segment a section is opened with the card NOSORT to indicate that the statements after this card can not be sorted. It is closed with the card SORT.

```
NOSORT
COUNT1=COUNT1+1
COUNT2=COUNT2+KEEP
SORT
```

These statements cannot be sorted because the same variables occur to the left and to the right of the equal sign. Each time this statement is passed, COUNT1 is incremented by one, and COUNT2 by KEEP. The variable KEEP is an internal CSMP variable and has the value one if the integration step is actually executed, and a value zero if the statements are only executed for a preliminary evaluation. In this way both the number of time intervals and the number of calculations of the whole program can be kept track of. A NOSORT section should not be written in the middle of a program, as then the parts above

and below this section are sorted separately.

---

### Exercise 17

Add the discussed statements to the programs of Fig. 2. Try the methods RECT with DELT=0.5, TRAPZ with DELT=1, RKAFX with DELT=2 and the method RKS. Which combination do you prefer?

---

As has been said the RKS method is preferred for convenience and accuracy. There are however circumstances when RKS may not be used. Sometimes the content of an integral has to be changed instantaneously. For instance, when a crop is harvested, the standing biomass is suddenly decreased. Contents of integrals can only be changed by their rates. When an integral must be emptied suddenly, the rate of outflow must be defined as the content divided by the time interval. Thus application of the rectilinear method gives for the content A at time t+DELT:

$$A_{t+DELT} = A_t - DELT * (A_t / DELT)$$

so that the integral is emptied indeed. In this discontinuous situation sophisticated integration methods cannot be used. For instance, the average content during the last time-interval is A/2, so that the use of the trapezoidal method results in the removal of A/2 only. A Runge-Kutta method will give even odder results. The rule is that the rectilinear method has to be used, whenever a division by DELT occurs in one of the definitions of the rates of change of the integrals. The time-interval has then to be determined by trial and error without help of a RKS method.

To conclude this section a comparison of the methods RECT, TRAPZ and RKAFX (a Runge-Kutta/Simpson integration method with fixed time-interval) is given. The relative error, in method RECT, is given by  $\frac{1}{2} * RGR^2 * DELT * TIME$ . For normalization it is useful to take TIME equal to the time constant, that is equal to  $RGR^{-1}$ : After this period, the total amount equals the initial amount multiplied or divided by the number e. The relative error after this period becomes  $\frac{1}{2} * RGR * DELT$ .

It can be shown that the expression for the relative error of the method TRAPZ and RKAFX are  $1/6 * (RGR * DELT)^2$ , and  $1/120 * (RGR * DELT)^4$  respectively. The combination  $RGR * DELT = DELT / TAU$  gives the ratio of the time-interval and the time constant. In the following table the value is given for required relative accuracies of 10%, 1% and 0.1%.

	10%	1%	0.1%
RECT	0.2	0.02	0.002
TRAPZ	0.8	0.25	0.08
RKSFX	1.9	1.0	0.6

In method RECT the program is executed once per time-interval, in TRAPZ twice and in RKSFX 4 times. Therefore an accuracy as high as 0.1% can be obtained by 500 computations with RECT, by 25 computations with TRAPZ and by only seven computations with RKSFX. Thus RKSFX is by far preferable, if the time constant is known. It is also clear that with RKSFX not much computation time is saved by accepting an accuracy of 10% instead instead of 0.1%.

Of course much more can be said about the use and misuse that can be made of numerical integration methods, but this goes beyond the scope of this book. More information can be found in Milne (1960).

## 3 The growth of yeast


### 3.1 Description of the system

Growth is only exponential as long as the relative growth rate remains constant. This is usually so with yeast when it is grown under aerobic conditions with a sufficient supply of sugar and some other growth essentials. The sugar is then continuously consumed to provide the 'C skeletons' and the energy for the growth of new yeast cells and for maintenance of the yeast. The end-products,  $\text{CO}_2$  and  $\text{H}_2\text{O}$ , of the sugar broken down in the respiratory process do not pollute the environment of the yeast. However, if yeast grows under anaerobic conditions, one end-product of the respiratory processes is alcohol which may accumulate in the environment. This slows down and ultimately stops the development of yeast buds even when there is still enough sugar available for growth.

Growth curves for yeast that result under such conditions are given in Fig. 3. It should be noted that yeast once formed remains because only the bud formation is affected by the alcohol; the yeast itself is not killed. Two of the four growth curves are from an experiment of Gause (1934) with monocultures of the yeast species *Saccharomyces cerevisiae* and *Schizosaccharomyces 'Kephir'*. It is obvious that the initial relative growth rate and the maximum volume of yeast that is ultimately formed is highest for the first species.

Gause cultivated both yeast species not only in monoculture, but also in mixture. The results of this experiment are also presented in Fig. 3 by the other two curves. A comparison of the growth of both species in mixture with their growth in monoculture shows that both affected each other in the first situation. It was proposed by Gause that this was due to the formation of the same waste product, alcohol, that affected the bud formation of both species. In this chapter we shall analyse whether this explanation is acceptable by constructing a model that simulates the growth of two species independently and in mixture under the assumption that the production of the same harmful waste product is the only cause of interaction.

## Schizosaccharomyces



## Saccharomyces




Fig. 3 | The growth of *Saccharomyces cerevisiae* and *Schizosaccharomyces* 'Kephir' in monoculture and in mixture. The observational data were obtained by Gause (1934) and the curves are simulated, as explained in the text.

### 3.2 Relational diagrams

It is sometimes advantageous to summarize the main interrelations of a system in a relational diagram, and to formulate the quantitative aspects at a second stage of actual model building. Such relational diagrams may be presented in various ways, but the conventions introduced by Forrester (1961) prove to be the most convenient in ecology, although they were first developed for the presentation of industrial systems. Forrester assigned special symbols to the various types of variables that may be distinguished in state determined systems. The state variables or the contents of integrals are presented within rectangles, the rates of changes within valve symbols, auxiliary variables within circles and parameters are underlined. The flow of material is presented by solid arrows and the flow of information by dotted arrows.

The simple system of exponential growth is drawn according to Forrester's conventions in Fig. 4. The amount of organisms is a state variable; its value increases by a material flow,




Fig. 4 | A relational diagram of exponential growth, drawn according to the conventions of Forrester.

whose rate is the growth rate. The broken line between the state variable and the rate shows that the rate depends (in some way or another) on the state variable and the other broken line shows that the rate also depends on a parameter which is here considered to be a constant. This figure contains all the interrelations that play a role, but does not consider their details. For instance, in the relational diagram, it is still not stated whether the growth rate is proportional to the amount of organisms or to some power of this amount: this information is given later in the simulation program.

The relational diagram for the yeast system is presented in Fig. 5. There are three state variables; the amount of the first and second yeast species and the amount of alcohol. The lines of information flow show directly that the growth of yeast is supposed to depend on the amount of yeast, a relative growth rate and an auxiliary variable: a reduction factor. This reduction factor, in its turn, is given as a function of the amount of alcohol that is present. The relations are, of course, the same for both yeast species although numerical values of parameters and functions may be different. The amount of alcohol increases by the rate of alcohol production of both species. The alcohol production of each species is supposed to depend on the growth rate of the species and on an alcohol production factor.




Fig. 5 | A relational diagram for the growth and interference of two interfering yeast species

### Exercise 18

In Section 1.4 it is said that rates do not depend on each other in state determined systems. Why is the line of information flow between the rate of growth and the rate of alcohol production not in contradiction with this principle?

Relational models should contain as few details as possible, otherwise they are very difficult to grasp and so defeat their purpose. In studying them, much emphasis should be given to aspects that are not incorporated. For instance, in the present scheme there are no loops that relate the alcohol production directly to the amount of yeast, indicating that the cost of maintenance of yeast cells is not accounted for. The amount of sugar is also not considered because it is assumed to be always available in sufficient amounts.

### Exercise 19

Incorporate the aspect of limited food supply in the relational

diagram

---

### 3.3 Simulation

The growth of the first yeast species (*Saccharomyces*) is now simulated by stating that the amount of yeast equals

$$Y1 = \text{INTGRL}(IY1, RY1) \quad (3.1)$$

in which

$$\text{INCON } IY1 = 0.45$$

is the initial amount of yeast in the arbitrary units, used by Gause, and the rate of yeast growth is given by

$$RY1 = \text{RGR1} * Y1 * (1. - \text{RED1}) \quad (3.2)$$

The relative growth rate is defined with

$$\text{PARAMETER RGR1} = \dots\dots$$

It was observed by Gause that in both species the formation of buds was completely stopped at some maximum alcohol concentration which is given as a percentage by

$$\text{PARAMETER MALC} = 1.5$$

The dependence of the reduction factor on the alcohol concentration may now be obtained with an arbitrary function generator.

$$\text{RED1} = \text{AFGEN}(\text{RED1T}, \text{ALC}/\text{MALC})$$

The most elementary assumption is that bud formation decreases linearly with increasing alcohol concentration, which is introduced with

$$\text{FUNCTION RED1T} = (0., 0.), (1., 1.)$$

---

#### Exercise 20

Express RED1 directly in ALC and MALC without using the function generator.

---

The alcohol concentration itself is the integral of the alcohol

production rate which is zero at the initialization of growth:

$$\begin{aligned} \text{ALC} &= \text{INTGRL}(\text{IALC}, \text{ALCP1}) && (3.3) \\ \text{INCON IALC} &= 0. \end{aligned}$$

and the alcohol production rate is proportional to the growth rate of yeast:

$$\text{ALCP1} = \text{ALPF1} * \text{RY1} \quad (3.4)$$

Two values need to be determined now: the relative growth rate and the alcohol production factor. During the early stages of growth, RED is practically zero, so that the growth rate is equal to  $\text{RGR1} \times \text{Y1}$ . This allows a first estimate of RGR1 from the data in Fig. 3 for the monoculture. ALPF1 follows from the observation that growth was terminated when the alcohol concentration equalled 1.5 percent and the amount of yeast about 13 units.

---

#### Exercise 21

What is a first estimate of RGR1 in the correct units? What is the value of ALPF1 in the correct units? Is this value only physiologically determined or does it also depend on the volume of water in the vessels with yeast? What is the value of IALC when not only the initial amount of yeast is introduced at initialization, but also the corresponding amount of alcohol? Estimate the same values for *Schizosaccharomyces*, it being known that the alcohol concentration at which the formation of buds is completely inhibited is also 1.5 percent. Which species has the larger alcohol production factor?

---

The structural equations that describe the growth of the second species (*Schizosaccharomyces*) are, of course, the same as those for the first, so that in a model for concurrent growth it suffices to write them twice: once with a 1 at the end of the relevant symbols and once with a 2. The equation that describes the alcohol concentration becomes

$$\text{ALC} = \text{INTGRL}(\text{IALC}, \text{ALCP1} + \text{ALCP2}) \quad (3.5)$$

This equation holds on the condition that both species interfere only with each other through the production of the same alcohol.

Fig. 6 shows the resulting simulation program with MALC identical for both species and the proper data. In the main program IY1 and IY2 are both set to 0.45 units, so that the growth in

```

TITLE MIXED CULTURE OF YEAST
  Y1=INTGRL(IY1,RY1)
  Y2=INTGRL(IY2,RY2)
INCON IY1=0.45,IY2=0.45
  RY1=RGR1*Y1*(1.-RED1)
  RY2=RGR2*Y2*(1.-RED2)
PARAMETER RGR1=0.21,RGR2=0.06
  RED1=AFGEN(RED1T,ALC/MALC)
  RED2=AFGEN(RED2T,ALC/MALC)
FUNCTION RED1T=(0.,0.),(1.,1.)
FUNCTION RED2T=(0.,0.),(1.,1.)
PARAMETER MALC=1.5
  ALC=INTGRL(IALC,ALCP1+ALCP2)
  ALCP1=ALPF1*RY1
  ALCP2=ALPF2*RY2
PARAMETER ALPF1=0.12,ALPF2=0.26
INCON IALC=0.
FINISH ALC=LALC
  LALC=0.99*MALC
TIMER FINTIM=150.,OUTDEL=2.
OUTPUT Y1,Y2,ALC
END
STOP
ENDJOB

```

Fig. 6 | A simulation program for the growth of two yeast species that interfere through the production of the same waste product (alcohol).

the mixture is simulated. The two monocultures are simulated in reruns.

FINTIM is set at 150 hours, but the two lines

```

FINISH ALC = LALC
LALC = 0.99 * MALC

```

are inserted to avoid unnecessary 'number-grinding', when the alcohol concentration is close to its maximum. This condition FINISH indicates that the simulation is terminated as soon as the alcohol concentration reaches 99 percent of its maximum value.

The relative growth rates and the alcohol production factors are chosen such that the results of the two experimental monocultures are matched as well as possible. A comparison of the mixtures (Fig. 3) shows that the actual growth of *Schizosaccharomyces* is slightly more than the simulated growth. Barring statistical insignificance, we must conclude that both species do not interfere with each other's growth through the production of alcohol only, as assumed in the model. It may be that *Schizosaccharomyces* produces some other waste product that is harmful for the other or that *Saccharomyces* produces a waste product that stimulates the other. These possibilities cannot be distinguished from each other without additional information. And as long as this is not available it is a futile exercise to simulate such suppositions.

---

### Exercise 22

Try to reason whether a similar effect could result from the supposition that REDT for the species is not given by

```
FUNCTION RED1T= (0.,0.),(1.,1.)  
FUNCTION RED2T= (0.,0.),(1.,1.)
```

but by, for instance:

```
FUNCTION RED1T= (0.,0.),(0.5,0.75),(1.,1.) (Sacch.)  
FUNCTION RED2T= (0.,0.),(0.5,0.25),(1.,1.) (Schizos.)
```

If this is too difficult, you may find the answer by simulation.

---

These simulation programs are conveniently amended. For instance, the yeast cultures may be washed continuously with water that contains sufficient sugar. The integral of the alcohol concentration is then

$$ALC = \text{INTGRL}(IALC, ALPF1*RY1 + ALPF2*RY2 - ALC/WSC)$$

in which the washing constant (WSC) is expressed in hours and presents the average residence time of the water in the vials with yeast, as will be shown in Section 6.4.2.

---

### Exercise 23

What is in due course the alcohol concentration and the absolute growth rate of both yeast species for WSC equal to 10 hours?

---

### 3.4 Logistic growth

The simulation program in Section 2.2 was developed from the differential equation form. The differential equation form for the present problem will now be derived from the structural equations of the simulation program, but only for situations where the reduction factor is inversely proportional to the alcohol concentration so that (1-RED) may be replaced by (1-ALC/MALC). Since the alcohol concentration is equal to the amount of yeast times the alcohol production factor according to the Eqns (3.3) and (3.4), it is then possible to rewrite Eqn (3.2) in differential equation form as

$$dY/dT = RGR \times Y \times (1 - Y/YM) \quad (3.6)$$

in which Y is the amount of yeast, T is the time and YM stands

for the maximum amount of yeast. This equation may be integrated and then becomes

$$Y = \frac{YM}{1+Ke^{-RGRxT}} \quad (3.7)$$

---

#### Exercise 24

Express YM in MALC and ALCPF. What are the values of YM for both species of yeast? Show by differentiation that Eqn (3.7) is an integrated form of Eqn (3.6). Express the initial amount of yeast in the constant K and YM of Eqn (3.7). Calculate the time course of the growth of *Saccharomyces* and compare the result with the simulated course. Why does the differential equation only hold for situations where the initial amount of yeast is very small, whereas the simulation program is generally valid? (see also Exercise 21).

---

The growth curve that is described by the differential equation and also presented by the simulated growth curves for the monoculture yeast in Fig. 3 is called the logistic growth curve. This S-shaped curve is symmetrical, but this symmetry hinges on the assumption of inverse proportionality between the reduction factor of growth and the amount of growth that has been made. Especially Lotka (1925) and Volterra (1931) generalized the logistic differential equation for interfering species with the following set of differential equations:

$$dY1/dT = R1 \times Y1 \times (1 - A1 \times Y1 - B1 \times Y2)$$

$$dY2/dT = R2 \times Y2 \times (1 - A2 \times Y1 - B2 \times Y2)$$

In general this set of differential equations cannot be integrated into analytical expressions for Y1 and Y2 as functions of time and therefore it is wiser to leave such simplifying approaches alone and to formulate the problem directly in terms of a simulation model.

---

#### Exercise 25

Show to what extent the simulation model for mixed growth of yeast is covered by this set of differential equations. Express the constants R1, R2, A1, A2, B1 and B2 in the constants RGR1, RGR2, ALPF1, ALPF2 and MALC. Which constants of the differential equations are the same? Do they remain the same in situations where a species produces a waste product which is harmful for the other species only?

---

## 4 Interference of plants

### 4.1 Replacement series

The interference of plant species in the field is most conveniently studied by experiments based on the replacement principle.

Thus experimental plots are divided into small equal areas, usually squares. A seed of the first species is placed in each square of one plot and a seed of the second species in each square of another. In this way monocultures of the two species are obtained. On another plot, the seeds of both species are placed alternatively in the squares to create a mixture in which half of the space is allotted to one species and the other half to the other. Other mixtures may be obtained by allotting the individual squares to the species in other proportions. The relative seed density of a species in a mixture is now defined as the seed density of the species in the mixture divided by its seed density in the monoculture. Obviously, the sum of the relative seed densities  $z_1 + z_2$  then equals always 1. The yields of the species in monocultures are represented by the symbols  $M_1$  ( $z_1 = 1, z_2 = 0$ ) and  $M_2$  ( $z_1 = 0, z_2 = 1$ ) and the yield of each species in the mixture by  $O_1$  and  $O_2$ . Instead of a single seed also  $n$  seeds of the one species or  $m$  seeds of the other may be placed on each square, for instance when the individual plants of the species are of different sizes.

An experiment is now considered where the individual squares are so large that the two species do not interfere with each other. The seed densities are then low and the yields of both species consequently small. But  $M_1$  and  $M_2$  are of course not necessarily the same. Here the yield of each species in the mixture may be represented by

$$O_1 = \frac{z_1}{z_1+z_2} M_1 \quad \text{and} \quad O_2 = \frac{z_2}{z_1+z_2} M_2 \quad (4.1)$$

The linearity is due to the seeds being so far apart that the plants do not interfere with each other.

The yields may be expressed in dry weight per unit surface or

number of seeds per unit surface for seed-forming species. In the latter case the relative reproductive rate of the species may be defined as

$$\alpha_{12} = \frac{O_1/z_1}{O_2/z_2} = M_1/M_2 \quad (4.2)$$

and appears to be equal to the ratio of the yield of the species in monoculture. If  $\alpha_{12}$  is 1, the species match each other. If  $\alpha_{12}$  is greater than 1, species 1 gains on species 2: the latter eventually disappears from the mixture, if the harvested mixture is resown repeatedly at the original density. What happens if the individual squares on the experimental plots are made smaller and smaller? Then the seed rates of both species increase accordingly and so do the yields. But this is not the only effect. At a certain stage the space allotted to each seed is so small, that the plants interfere with each other. If the species have equal competitive ability one species will not infringe upon the space allotted to the other and Eqn (4.1), resulting in a linear relation between seed rate and yield of the species in the mixture, is still valid. However, one species usually has more competitive ability and will infringe upon the space allotted to the other. Then the yield of this species in the mixture is higher than expected and that of the other species is lower.

Many experiments of this type have been done and the result of one of them with barley and oats is given in Fig. 7. Here the squares were of two sizes: in one experiment 310 cm<sup>2</sup> was allotted to each seed and in the other 31 cm<sup>2</sup>. With the wide planting, barley infringed somewhat on the space of oats, but the yield curves were still practically straight. With the narrow planting, however, the yield of barley in the mixtures was relatively high and of oats relatively low, indicating that barley was by far the strongest competitor. The results of this and many other experiments with barley and oats (de Wit, 1960) show that the relative yield total of the mixtures, defined by

$$RYT = O_1/M_1 + O_2/M_2 \quad (4.3)$$

is practically equal to one. When RYT equals one, the species are mutually exclusive. This equality may be considered the operational definition of 'competing for the same niche', to use a term out of the field of animal ecology. The relative reproductive rate for seed producing species is now not equal to the ratio of the yields in the monoculture but may be expressed by




Fig. 7 | Seed yields in number of kernels per  $m^2$  in a replacement experiment of barley and oats at two densities of sowing (de Wit, 1960).

$$\alpha_{12} = \frac{O_1/z_1}{O_2/z_2} = k_{12}(M_1/M_2) \quad (4.4)$$

in which  $k_{12}$  is the relative crowding coefficient and characterizes to what extent one species infringes upon the space allotted to the other. Eqn (4.3) (with  $R_{YT} = 1$ ) and Eqn (4.4) may be combined and replaced by

$$O_1 = \frac{k_{12}z_1}{k_{12}z_1 + z_2} M_1 \text{ and } O_2 = \frac{z_2}{k_{12}z_1 + z_2} M_2 \quad (4.5)$$

These equations are similar to (4.1), except for the relative crowding coefficient which weights the relative frequency of sowing. Similar relations hold when biomass yields are considered, except that the relative reproductive rate loses its meaning.

---

### Exercise 26

Show that the equations (4.5) are consistent with the equations (4.3) and (4.4). In the following table the data, used to draw Fig. 7(left side), are given (kernels  $m^{-2}$ )

$z_b$	0	0.2	0.4	0.6	0.8	1.
Barley	0	4270	8050	9920	10670	12100
Oats	16200	11400	5800	3200	1330	0

Calculate RYT in the four mixtures as well as  $k_{bo}$  and  $\alpha_{bo}$ . Which species has the highest yield in monoculture and which species gains in competition?

---

The yield curves in Fig. 7 have been calculated by assuming that the relative crowding coefficient is independent of the relative seed frequencies and that  $RYT = 1$ ; the agreement between the curves and the experimental data over the whole range of frequencies show that this is a fair assumption. The constancy of the relative crowding coefficient has been confirmed by the analyses of many other experiments (de Wit, 1960; van den Berg, 1968), so that it is reasonable to state the following. If the relative yield total in replacement experiments equals about 1 over the whole range of seed frequencies, then the relative crowding coefficient may be considered independent of these seed frequencies.

Of course there are also situations where the species do not exclude each other, so that the relative yield total does not equal 1. The equations (4.5) cannot be applied in such situations. For instance, leguminous species and grass are not mutually exclusive when the first obtains its nitrogen from the air through nitrogen-fixing *Rhizobium* bacteria and the second from the soil and from the first species (de Wit et al., 1966). The relative yield total (RYT) may be also greater than one when one species has a longer growing period than the other. On the other hand, it has been shown that RYT is smaller than one when one species contains a virus which is harmful to the other (van den Bergh, 1968; Sandfaer, 1970).

#### 4.2 Density of sowing

Replacement experiments between two species and density of sowing experiments of single species have much in common, as is conveniently illustrated by considering the results in Fig. 8 of replacement experiments between barley and oats at different pH values of the soil. As far as the relative crowding coefficient is concerned, the two species matched each

other at a pH of 4. However, at a pH of 3.7, the relative crowding coefficient of oats with respect to barley was about 2, although the yields of the two species in monoculture were still the same as at the higher pH. Obviously a lower pH affects the competitive ability of barley. This effect was detrimental to yield when oats were around to claim the space, but not in monoculture. At a pH of 3.2 the situation was still worse: the relative crowding coefficient of oats with respect to barley increased to 3, whereas the yield of barley decreased to a low level. The physiological cause of the phenomenon is that the root development of barley is much more sensitive to low pH than of oats. The most extreme situation was reached at a still lower pH. Here the relative crowding coefficient of oats with respect to barley increased up to 20, whereas the barley did not grow at all, as reflected by its zero yield in monoculture. Such a replacement experiment of barley and oats in situations where barley does not grow at all is, in fact, an experiment on the density of sowing of oats. In other words density of sowing experiments are a limiting case of replacement experiments. Therefore the equations (4.5) not only describe the results of competition experiments, but those of density of sowing experiments as well. It is only necessary to give them a more suitable form.

As one species of the replacement series does not grow at all, the second equation may be omitted. The remaining equation is still in an unsuitable form because it is formulated in terms of relative seed frequencies and refers also to the species that is not sown or did not grow at all. A more suitable form is obtained when  $Z_1/Z_m$  is substituted for  $z_1$  and  $(Z_m - Z_1)/Z_m$  for  $z_2$  in which  $Z_m$  is the maximum seed rate used in the experi-




Fig. 8 | Replacement experiments of barley and oats at different pH-KCl values of the soil: 4, 3.7, 3.2, 3.1 for a, b, c, d, respectively (de Wit, 1960)

ments and expressed in absolute units, i.e. plants  $m^{-2}$ . When the subscripts 1 are omitted, the first equation of (4.5) becomes

$$O = \frac{B \times Z}{B \times Z + 1} O_m \quad (4.6)$$

---

*Exercise 27*

Derive this formula and express the constants B and  $O_m$  in the relative crowding coefficient k and seed rate  $Z_m$  and the yield M.

---

In Eqn (4.6)  $O_m$  and B are independent of the density of sowing Z. The dimension of Z is number of plants  $m^{-2}$  or a similar unit.  $O_m$  is the theoretical maximum yield, in  $g\ m^{-2}$  for instance, that is obtained when the seed density is very high and  $B \times O_m$  is the yield of a single plant growing alone. B itself has the dimension of  $m^2\ plant^{-1}$  and may be considered the amount of space that is occupied by a single plant growing alone. The value of  $O/O_m$  has a lower limit of 0 and an upper limit of 1.

---

*Exercise 28*

Construct a graph from Eqn (4.6) for  $O_m = 100$ ,  $B = 0.05$  and Z ranging from 0 to 100. Draw the asymptote  $O_m$  and the initial slope  $B \times O_m$  of the curve. Mark along the horizontal axis the position where the yield is half of the maximum yield  $O_m$ . Mark also the distance  $1/B$  along the horizontal axis. Give now expressions for:

$$\lim_{Z \rightarrow \infty} (O) = \dots$$

$Z \rightarrow \infty$

$$\lim_{Z \rightarrow 0} (O/Z) = \dots$$

$Z \rightarrow 0$

$$\lim_{Z \rightarrow 0} (O/O_m) = \dots$$

$Z \rightarrow 0$

$$\lim_{Z \rightarrow \infty} (O/O_m) = \dots$$

$Z \rightarrow \infty$

---

The result of a spacing experiment with subterranean clover harvested at various times after planting is given in Fig. 9.  $O_m$  increases monotonously with time. The rate of increase of this parameter presents in principle the growth rate of a closed crop surface from the beginning of growth onwards. Under favourable conditions it may be expected that  $O_m$  increases with about  $20\ g\ m^{-2}\ day^{-1}$ , this being the potential growth rate




Fig. 9 | A spacing experiment with subterranean clover (Donald, 1963), harvested at various times after planting.

of most agricultural crops in the Netherlands (de Wit et al., 1970). The value of  $B$  also increases monotonously with time; it represents the (calculated) ability of a single plant to occupy space during its growth and this ability is strongly affected by the stage of development and the distribution of photosynthesis products over the various plant organs.

Baeumer & de Wit (1968) did a spacing experiment with barley and oats on a soil well supplied with nutrients and water. Rows of plants, rather than single plants, were grown at distances of 25 and 100 cm, that means with seed densities of 4 and 1 row per metre. The dry matter yield was determined at four stages. The results of this experiment are summarized in Table 1.

---

#### Exercise 29

Calculate the values of  $B$  and  $O_m$  for barley and oats on the four harvesting dates by using Eqn (4.6). What are the units of  $B$  and  $O_m$ ? The calculated value for  $O_m$  on the first date is unreliable. Why is this so? Is there a combination of  $B$  and  $O_m$  that is much more accurate? Draw graphs of  $B$  and  $O_m$  against time. Linearize the curves for  $O_m$ , omitting the data points for the first harvesting date and recalculate  $B$  for the value of  $O_m$  estimated in this way.

---

Table 1 The dry biomass yield in  $\text{g m}^{-2}$  of barley and oats, sown at 25 and 100 cm on 2 May 1966. Emergence and seedling establishment was completed on 15 May. Field experiment IBS 975, 1966.

Date of harvest	Barley at		Oats at	
	25 cm	100 cm	25 cm	100 cm
7 June	117	36	81	22
21 June	426	223	319	142
5 July	588	341	503	263
19 July	858	496	789	516

There is also a graphical method to make a first estimate of  $B$  and  $O_m$  from the yields at different densities. Then  $O^{-1}$  (ordinate) must be plotted against  $Z^{-1}$  (abscissa). Since according to Eqn (4.6) the result should be a straight line, a fit by eye is easy. This method is convenient when experimental results of more than two densities are available.

### Exercise 30

How can  $O_m$  and  $B$  be found from such a graph? What is a disadvantage of the method?




Fig. 10 | Time functions of  $B$  and  $OM$  for barley and oats, calculated from the data in Table 1.

The calculated curves of B and  $O_m$  against time for barley and oats are given in Fig. 10. Barley grows somewhat better at low temperatures and its value of B increases during the early part of the growing season more rapidly than for oats. Hence, when both species are grown together, barley occupies relatively more space and by the time oats gets around to claim its share, all the space is already occupied. Therefore the competitive ability of barley in a mixture with oats is usually higher. It may be possible to calculate the mutual interference of both species in a mixture from the course of B and  $O_m$ , as determined from density experiments with one species.

### 4.3 Simulation of plant interference

To arrive at a simulation program for the interference of plants, it is necessary to distinguish the correct state variables and to find expressions for their rate of change. A convenient state variable is the relative space that is occupied by the species, defined as the yield (O) of the species, divided by the maximum yield ( $O_m$ ) obtained at very high seed density. This relative space is according to Eqn (4.6):

$$RS = \frac{B \times Z}{B \times Z + 1} \quad (4.7)$$

The term relative space is preferred because the term relative yield for this quotient would lead to confusion with the term relative yield used in the analyses of replacement series. The value of RS ranges from 0 to 1.

The rate of change of the state variable may be found by differentiating RS with time and rearranging the expression. The result is

$$\frac{d(RS)}{dT} = \frac{dB/dT}{B} \times RS \times (1 - RS) \quad (4.8)$$

This equation is very similar to the equation for logistic growth, derived in Section 3.4; the two main differences being the maximum value of 1 for the state variable and the non-constancy of the 'relative growth rate'.

#### Exercise 31

Derive the expression for  $d(RS)/dT$ . For this purpose, Eqn (4.7) is differentiated, taking into account that B is a variable

function and Z a constant function of time. Eqn (4.7) is then used again to eliminate Z. What is the dimension of  $(dB/dT)/B$ ? Does this relative rate of change increase or decrease with time? What is the expression for B against time when the relative rate of change is constant?

---

Eqn (4.8) holds for one species. The factor  $(1 - RS)$ , which may range from practically 1 in the beginning to practically 0 at the end of the growth period, characterizes the reduction of growth under influence of the space that is occupied. When two species are growing together, a situation may be visualized where plants do not distinguish between occupation of space by one species or the other. Then the relative spaces may be added as to their influence on the growth of each species so that the following set of equations characterize the situation:

$$\frac{d(RS1)}{dT} = \frac{dB1/dT}{B1} \times RS1 \times (1 - SRS)$$

$$\frac{d(RS2)}{dT} = \frac{dB2/dT}{B2} \times RS2 \times (1 - SRS)$$

$$SRS = RS1 + RS2$$


---

### Exercise 32

Construct a relational diagram of this type of plant interference.

---

```
TITLE COMPETITION BETWEEN BARLEY AND UATS
INCON DBI1=0.0047,DBI2=0.0033,RSI1=0.002,RSI2=0.002
      RS1=INTGRL(RSI1,(DB1/B1)*RS1*(1.-SRS))
      RS2=INTGRL(RSI2,(DB2/B2)*RS2*(1.-SRS))
      B1=AFGEN(BTB1,TIME)
      B2=AFGEN(BTB2,TIME)
      DB1=DERIV(DBI1,B1)
      DB2=DERIV(DBI2,B2)
      O1=RS1*AFGEN(OMTB1,TIME)
      O2=RS2*AFGEN(OMTB2,TIME)
      SRS=RS1+RS2
PRINT RS1,RS2,SRS,O1,O2
TIMER FINTIM=65.,PRDEL=5.
FUNCTION OMTB1=(0.,0.),(23.,377.),(37.,612.),(51.,780.),(65.,1132.)
FUNCTION OMTB2=(0.,0.),(23.,333.),(37.,552.),(51.,724.),(65.,956.)
FUNCTION BTB1=(0.,0.001),(23.,0.11),(37.,0.574),(51.,0.778),(65.,0.778)
FUNCTION BTB2=(0.,0.001),(23.,0.076),(37.,0.346),(51.,0.571),(65.,1.17)
END
STOP
ENDJOB
```

Fig. 11 | A simulation program for interference of two plant species that do not distinguish between the occupation of space by one species or the other.

The two differential equations are the basis for a simulation program of two species grown in a mixture, which is presented in Fig. 11.

The function tables for B and  $O_m$  are those for the barley(1) and oats(2) experiment of Table 1. DERIV is the only new function that is introduced. This function calculates the derivative of the second argument, here the value of  $dB/dT$  from the function of B against time. Like an integral, the derivative has to be initialized and this initial value is given as the first argument of the function.

---

### Exercise 33

Why is it necessary to set the value of B slightly above 0 at emergence? Initialize RSI1, RSI2, DBI1 and DBI2. Is it necessary to initialize the derivative functions accurately? Compare the results of the simulation graphically with those of the actual competition experiment in Table 2.

---

Table 2 The dry biomass yield in  $g\ m^{-2}$  of barley and oats, sown alternately in rows 25 cm apart. Field experiment IBS 975, 1966.

---

Date of harvest	Mixture	
	Barley	Oats
7 June	62	30
21 June	235	142
5 July	375	165
19 July	512	308

---

Inspection of the experimental data in Table 2 shows that barley occupied much more space than oats in the mixture although both species were planted alternately in rows. The simulated results given in the answer to Exercise 33 prove that this better performance may be explained by the more favourable course of the B curve for barley during early growth. The higher values of B for oats later are ineffective in the mixture because too much space is already occupied by the barley at the early stages of growth and at this sowing density.

Although this simple model of interference holds for mixtures of some species, it does not always hold in situations where species exclude each other. For instance, in mixtures of short and long peas, it makes a large difference to the short peas, whether the space is occupied by other short peas or by long

peas. In the latter case, practically all light is intercepted by the long neighbours so that the growth of the short peas is almost suppressed. Experimental and simulated results of a competition experiment with these species are given in Fig. 12. To obtain the simulated curves A, it was assumed that the simple model as used in this section for barley and oats was valid. The difference between actual and simulated results is so large that this supposition must be rejected. The curves B were obtained by assuming that the relative space of each species may be weighted according to their respective heights (H1 and H2) which differed at the end about threefold. This weighting was done according to the equations

$$\text{SRS1} = \text{RS1} + (\text{H2}/\text{H1}) \times \text{RS2}$$

$$\text{SRS2} = (\text{H1}/\text{H2}) \times \text{RS1} + \text{RS2}$$

for the sums of the relative spaces. In this way the different light interception is accounted for in a first approximation. The much better agreement shows that this explanation is reasonable for the bad performance of the short peas in the mixture. In situations where species interfere in other ways than by mutual exclusion, it is of course futile to construct a model of competitive interference on basis of data obtained in monoculture only.




Fig. 12 | Observed and simulated growth of long and short peas in a mixture.

A: without weighting according to height.

B: with weighting according to height.

## 4.4 Further modelling aspects

### 4.4.1 The INDEX and MACRO feature

The simulation programs for yeast growth in the previous chapter and for plant competition in this chapter are given for two species but may be extended to more species. For a mixture of  $n$  species, the relevant structural equations must be written  $n$  times. Many repeated statements, however, make the program less clear and often introduce errors, especially if it is necessary to change the structure. Hence it is better that statements are repeated by the computer. The INDEX feature is suitable for this purpose. If for instance, two plant species are competing, the equations for the relative space (RS) are written as

$$RS'1,2' = \text{INTGRL}(RSI'1,2', DB'1,2' / B'1,2' * RS'1,2' * (1. - SRS))$$

This statement is an order for the preprocessor (Appendix A) to write the equation twice: once with number 1 attached to the symbols, once with number 2. All variables that are different for each species obtain an appropriate number by order of the index '1,2' and all variables, that are the same for the two species, like SRS, remain the same. These variables have to be defined on their own, in this case by

$$SRS = RS1 + RS2$$

A summation of this type can be written more concisely by

$$SRS = \text{SUM1}(RS'1,2')$$

The other equations that describe the growth are given in Fig. 13, which contains a full program for competition between two species. The initial values for the relative space and the slopes of B are defined on parameter cards by

$$\begin{aligned} \text{INCON } RSI'1,2' &= \dots\dots\dots, \dots\dots\dots \\ \text{INCON } DBI'1,2' &= \dots\dots\dots, \dots\dots\dots \end{aligned}$$

The four functions must be given separately in FUNCTION tables. The corresponding CSMP program that is written by the preprocessor on the basis of this text, is also given in Fig. 13. Here it can be clearly seen that the INDEX feature is an order for repeated writing or defining of similar texts, parameters or output. For more than two species, i.e. four, the statement

```

TITLE COMPETITION BETWEEN BARLEY AND UATS
INCON DBI'1,2'=0.0047,0.0033
INCON RSI'1,2'=0.002,0.002
RS'1,2'=INTGRL(RSI'1,2',(DB'1,2'/B'1,2')*RS'1,2'*(1.-SRS))
B'1,2'=AFGEN(BTB'1,2',TIME)
DB'1,2'=DERIV(DBI'1,2',B'1,2')
O'1,2'=AFGEN(OMTB'1,2',TIME)
SRS=RS1 + RS2
TIMER FINTIM=65.,PRDEL=5.
FUNCTION BTB1=0.,0.001,23.,0.11,37.,0.574,51.,0.778,65.,0.778
FUNCTION BTB2=0.,0.001,23.,0.076,37.,0.346,51.,0.571,65.,1.17
FUNCTION OMTB1=0.,0.,23.,377.,37.,612.,51.,780.,65.,1132.
FUNCTION OMTB2=0.,0.,23.,333.,37.,552.,51.,724.,65.,956.
PRINT RS'1,2',SRS,O'1,2'
END
STOP
ENDJOB

```

```

TITLE COMPETITION BETWEEN BARLEY AND OATS
INCON DBI1=0.0047,DBI2=0.0033
INCON RSI1=0.002,RSI2=0.002
RS1=INTGRL(RSI1,(DB1/B1)*RS1*(1.-SRS))
RS2=INTGRL(RSI2,(DB2/B2)*RS2*(1.-SRS))
B1=AFGEN(BTB1,TIME)
B2=AFGEN(BTB2,TIME)
DB1=DERIV(DBI1,B1)
DB2=DERIV(DBI2,B2)
O1=AFGEN(OMTB1,TIME)
O2=AFGEN(OMTB2,TIME)
SRS=RS1 + RS2
TIMER FINTIM=65.,PRDEL=5.
FUNCTION BTB1=0.,0.001,23.,0.11,37.,0.574,51.,0.778,65.,0.778
FUNCTION BTB2=0.,0.001,23.,0.076,37.,0.346,51.,0.571,65.,1.17
FUNCTION OMTB1=0.,0.,23.,377.,37.,612.,51.,780.,65.,1132.
FUNCTION OMTB2=0.,0.,23.,333.,37.,552.,51.,724.,65.,956.
PRINT RS1,RS2,SRS,O1,O2
END
STOP
ENDJOB

```

Fig. 13 | A simulation program for interference of two plant species, written by using the INDEX feature and the CSMP program compiled from this by the preprocessor.

RS'1,4' = .....

generates equations for RS1, RS2, RS3 and RS4.

#### Exercise 34

Write a program for the growth of four yeast species in a mixture, using the INDEX feature.

A similar result may be obtained by using the MACRO feature. In a MACRO, a part of a process is described in general terms.

Every time a MACRO is called upon, CSMP writes its full text with the appropriate symbols. A MACRO is therefore not an order to execute a particular computation, but an order to write a part of a simulation program. Just as in a normal simulation program, it is not necessary to present the structural statements in computational order and it may well be that various parts of the MACRO are scattered throughout the computational program after the sorting process. The MACRO for the growth of a plant species may read as follows:

```
MACRO 0,RS = GROWTH(BTB,OMTB,DBI,RSI)
  RS = INTGRL(RSI,(DB/B)*RS*(1.-SRS))
  B = AFGEN(BTB,TIME)
  DB = DERIV(DBI,B)
  O = RS*AFGEN(OMTB,TIME)
ENDMAC
```

The first line indicates that there is a MACRO 'GROWTH', in which it is stated how the relative space and the yield (RS and O) depend on functions, variables and initial constants, given or calculated elsewhere in the CSMP program. The ENDMAC line indicates the end of the MACRO. Within the MACRO, the equations of the last section are given but with the numbers 1 and 2 omitted.

The MACRO is invoked by the sentence

```
O1,RS1 = GROWTH(BTB1,OMTB1,DBI1,RSI1)
```

for species 1 and

```
O2,RS2 = GROWTH(BTB2,OMTB2,DBI2,RSI2)
```

for species 2.

A program for competition between two species and the intermediate CSMP program that is generated are presented in Fig. 14. Detailed comparison of the text shows that three classes of names for variables, parameters and tables can be distinguished. First, those that are mentioned in the statement: these replace the dummy names at corresponding places in the MACRO definition. Secondly, those that are used within and outside the MACRO: these remain unchanged and are not necessarily mentioned in the invoking line. Thirdly, there are dummy names that are used only within the MACRO: these are replaced by unique names of the type ZZ... in order to avoid double definitions.

### Exercise 35

Make a detailed comparison of the 'intermediate' CSMP program written with the MACRO feature, the CSMP program written with the INDEX feature and the original CSMP program for competition between two species. It is only in this way that all logical aspects of the MACRO operations can be understood.

```
TITLE COMPETITION BETWEEN BARLEY AND OATS
INCON DBI1=0.0047,DBI2=0.033,RSI1=0.002,RSI2=0.002
MACRO O,RS=GROWTH(BTB,OMTB,DBI,RSI)
      RS=INTGRL(RSI,(DB/B)*RS*(1.-SRS))
      B=AFGEN(BTB,TIME)
      DB=DERIV(DBI,B)
      O=AFGEN(OMTB,TIME)
ENDMAC
      O1,RS1=GROWTH(BTB1,OMTB1,DBI1,RSI1)
      O2,RS2=GROWTH(BTB2,OMTB2,DBI2,RSI2)
      SRS=RS1 + RS2
TIMER FINTIM=65.,PRDEL=5.
FUNCTION BTB1=0.,0.001,23.,0.11,37.,0.574,51.,0.778,65.,0.778
FUNCTION BTB2=0.,0.001,23.,0.076,37.,0.346,51.,0.571,65.,1.17
FUNCTION OMTB1=0.,0.,23.,377.,37.,612.,51.,780.,65.,1132.
FUNCTION OMTB2=0.,0.,23.,333.,37.,552.,51.,724.,65.,956.
PRINT RS1,RS2,SRS,O1,O2
END
STOP
ENDJOB
```

```
TITLE COMPETITION BETWEEN BARLEY AND OATS
INCON DBI1=0.0047,DBI2=0.033,RSI1=0.002,RSI2=0.002
      ZZ1000=AFGEN(BTB1,TIME)
      ZZ1001=DERIV(DBI1,ZZ1000)
      SRS=RS1 + RS2
      ZZ1003=(ZZ1001/ZZ1000)*RS1*(1.-SRS)
      RS1=INTGRL(RSI1,ZZ1003)
      ZZ1004=AFGEN(BTB2,TIME)
      ZZ1005=DERIV(DBI2,ZZ1004)
      ZZ1007=(ZZ1005/ZZ1004)*RS2*(1.-SRS)
      RS2=INTGRL(RSI2,ZZ1007)
      O1=AFGEN(OMTB1,TIME)
      O2=AFGEN(OMTB2,TIME)
TIMER FINTIM=65.,PRDEL=5.
FUNCTION BTB1=0.,0.001,23.,0.11,37.,0.574,51.,0.778,65.,0.778
FUNCTION BTB2=0.,0.001,23.,0.076,37.,0.346,51.,0.571,65.,1.17
FUNCTION OMTB1=0.,0.,23.,377.,37.,612.,51.,780.,65.,1132.
FUNCTION OMTB2=0.,0.,23.,333.,37.,552.,51.,724.,65.,956.
PRINT RS1,RS2,SRS,O1,O2
END
STOP
ENDJOB
```

Fig. 14 | A simulation program for the interference of two plant species written by using the MACRO feature. The intermediate CSMP program, produced by the CSMP compiler is also given. The text of this intermediate program is not printed by the computer.

---

### Exercise 36

Only for readers that are familiar with FORTRAN.

What are the principal differences between a MACRO and a SUB-ROUTINE?

---

One may wonder why two methods are being developed to make simulation programs more lucid and to avoid repeated writing of structural statements, especially as these methods seem very much alike. However these methods are only similar in the context of the present small programs which are written for illustrative purposes. Later it will become evident that each method has its own field of use.

#### 4.4.2 The INITIAL and DYNAMIC section

The initial values for the relative space (RS) and the derivative of the space occupied by a single growing plant (DB) must be calculated before the simulation models discussed in the previous sections can be applied. To avoid errors and again to promote the clarity of the simulation, it is advantageous to incorporate this computation in the simulation program. Again it is convenient to use a MACRO, in which it is defined how RSI and DBI depend on the distance of sowing (DIST) and the function for B (BTB):

```
MACRO RSI,DBI = BEGIN(BTB,DIST)
  RSI = BI/(BI+DIST)
  BI = AFGEN(BTB,0.)
  DBI = (AFGEN(BTB,DELT)-BI)/DELT
ENDMAC
```

---

### Exercise 37

Derive the expression for RSI from Eqn (4.7). Can it be further simplified? Why is it possible to use the METHOD RKS, even when a division by DELT occurs?

---

The computational procedure, contained in this MACRO, has to be done only once for each species before the actual simulation is started. For this purpose an initial section of the simulation model is made, starting with an INITIAL card and ending with a DYNAMIC card.

The above MACRO is invoked twice within this INITIAL section. After this initial section, the normal dynamic structural statements of the simulation program are entered.

---

*Exercise 38*

Write a simulation program for the growth of three barley cultivars, assuming that B for the second variety and  $O_m$  for the third variety increase half as fast with time as for the first variety. Assume that the species are sown in rows 40 cm apart in a 1:1:1 ratio. Be careful about the value of DIST. Make use of the MACROS and the INITIAL section. Write the same program with the INDEX feature.

---

## 5 Growth and competition of Paramecium

### 5.1 Description of the system

Paramecia are protozoa: unicellular organisms that live in water and feed on bacteria. Gause (1934) did a series of experiments with the species *P.aurelia* and *P.caudatum* in monoculture and in mixture to study the principles of their mutual interference. The species were grown in test tubes with 5 cm<sup>3</sup> of Oosterhout's balanced physiological solution, buffered at pH 8.0. The medium was changed daily by centrifuging to separate the protozoa from the liquid with the waste products and the remaining food. A standardized amount of bacteria was added in the new solution as daily food. Just before centrifuging, the solution was carefully stirred and one tenth of the volume of liquid was taken out in which the number of protozoa were counted. Hence at the beginning of each day the number of protozoa was about nine-tenths of the number at the end of the day before.

---

#### Exercise 39

Why not exactly nine-tenths?

---

Two series of experiments were done, in the one loop experiment one standardized loop of bacteria was given each day and in the half-loop experiment a half of the standardized loop of bacteria was given. In both series, the species were grown in monoculture and in mixture. The monocultures were started with 20 protozoa of the species concerned and the mixed culture with 20 protozoa of each species.

The number of protozoa counted in the sample throughout a period of 16 days are given in Table 3.

---

#### Exercise 40

Plot the results on graphs and save these for a first estimation of parameters, later on.

---

The number of protozoa in the monoculture reached a maximum and stayed there, just as for yeast. The growth of yeast

Table 3 Numbers as sampled by Gause

Day of the experiment	Monoculture				Mixed culture			
	<i>P.aurelia</i>		<i>P.caudatum</i>		<i>P.aurelia</i>		<i>P.caudatum</i>	
	0.5 cm <sup>3</sup>		0.5 cm <sup>3</sup>		0.5 cm <sup>3</sup>		0.5 cm <sup>3</sup>	
	one half loop	half loop						
0	2	2	2	2	2	2	2	2
1	6	3	6	5	10	4	5	8
2	24	29	31	22	29	29	15	20
3	75	92	46	16	68	66	32	25
4	182	173	76	39	144	141	52	24
5	264	210	115	52	164	162	40	—
6	318	210	118	54	168	219	32	—
7	373	240	140	47	248	153	36	—
8	396	—	125	50	240	162	40	21
9	443	—	137	76	—	150	32	15
10	454	240	162	69	281	175	20	12
11	420	219	124	51	—	260	30	9
12	438	255	135	57	300	276	12	12
13	492	252	133	70	—	285	16	6
14	468	270	110	53	—	225	20	9
15	400	240	113	59	260	222	12	3
16	472	249	127	57	294	220	9	0

ceased because of the accumulation of waste products. But this cannot be the cause of stabilization here, since the waste products were removed every day by centrifuging. It stands to reason that here the ultimate size of the population was limited by the daily food supply. In the equilibrium situation this supply was then just sufficient to maintain the population and to replace the ten percent that was removed by sampling. In the mixed culture one of the species vanished, whereas the other survived at the same level as in monoculture. This competitive phenomenon has to be understood by a further analysis of the system.

To arrive at a quantitative description of the relevant growth and death processes, some assumptions have to be made. First it is assumed that a fixed ratio exists between the number of newly grown protozoa and the amount of food that is consumed. This ratio is called the conversion factor of food (CONVF) and has the dimension of number of protozoa per loop of bacteria. Second, it is assumed that there is a natural death rate which is proportional to the number of protozoa, so that it can be

characterized by a constant relative death rate (RDR), which is independent of the density. The rate of food consumption (CNRT) is assumed to be proportional to the number of protozoa (H), the density of food (FOOD) in the medium and the rate at which the protozoa search the water for food (RSW). The density of food is the amount of food (AFOOD) divided by the volume. However, the rate of food consumption per protozoa cannot exceed the maximum digestion rate of food (MRDIG).

---

#### *Exercise 41*

Determine the dimensions of the mentioned state, rate, and auxiliary variables and parameters and classify these according to type. Use as basic units: day, loop, protozoon, volume of test tube.

Construct a relational diagram for the growth of one protozoa species, taking into account that every day the population is sampled and the food is renewed. Show that the assumption of a constant relative death rate is mathematically equivalent to the assumption that food is needed to maintain the protozoa.

---

#### *5.2 A simulation program*

As done previously for the competition between plants, the dynamics of one species will be described in a MACRO, which is then invoked for each species with the appropriate names. The output variables of the MACRO are the number of protozoa (H), the rate of food consumption (CNRT) and the size of the sample (SPLE). The input variables are the rate of searching the water (RSW), the conversion factor of food (CONVF), the maximum digestion rate (MRDIG), the relative death rate (RDR) and the initial size of the population (HI). The moment of feeding and sampling (FDTIME) and the density of food (FOOD) are the same for both species, so that these are defined in structural statements outside the MACRO and do not appear in the MACRO definition.

The MACRO is as follows:

```
MACRO H,CNRT,SPLE=GROWTH(RSW,CONVF,MRDIG,RDR,HI)
```

The amount of protozoa is now defined by

```
H=INTGRL(HI,AGR)
```

The actual growth rate (AGR) is the difference between the net growth rate (NGR) and the rate of sampling (RSAM):

AGR=NGR-RSAM

and the net growth rate (NGR) is the difference between the gross growth rate minus the natural death rate (DR):

NGR=CNRT\*CONVF-DR

DR=RDR\*H

In calculating the consumption rate of food (CNRT), the maximum digestion rate must be accounted for. An AMIN1 function can be used, which takes the minimum of its arguments:

CNRT=H\*AMIN1(MRDIG,RSW\*FOOD)

---

### Exercise 42

Draw a graph of the consumption rate of food (CNRT) against the density of food (FOOD) for arbitrary values of MRDIG, RSW and H. How does this graph change with changing MRDIG, H or RSW. For which value of FOOD does CNRT equal zero and for which value of FOOD does NGR equal zero? Reason why this expression does not contain the amount of protozoa (H).

---

The calculation of the rate of sampling (RSAM) raises some problems because it is a discontinuous process. The sampling occurs only once a day and is zero for the rest of the time. The sample size is defined with

SPLE=FDTIME\*0.1\*H

in which FDTIME, as defined outside the MACRO, is one during one time-step at the end of the day and otherwise zero. To let the sampled quantity vanish during one time-step, the rate of sampling must be defined as the size of the sample divided by the time-step DELT:

RSAM=SPLE/DELT

as is seen from calculating (see also Section 2.4)

$$H_{t+\Delta t} = H_t - (0.1H_t/\Delta t)\Delta t$$

The MACRO is now terminated with

ENDMAC

In the main program, the MACRO is called for twice: once for

the species *P.aurelia*

```
HA,CNRTA,SPLEA=GROWTH(RSWA,CONVFA,MRDIGA,RDRA,HIA)
```

and once for the species *P.caudatum*

```
HC,CNRTC,SPLEC=GROWTH(RSWC,CONVFC,MRDIGC,RDRC,HIC)
```

In the main program FDTIME is defined by

```
FDTIME=IMPULS(1.,1.)
```

This function has the value 1 at the moment indicated by the first argument and subsequently at intervals defined by the second argument. The rest of the time, the function equals zero. The variable FDTIME is used within the MACRO to define the moments of sampling and outside the MACRO also to replenish the food at daily intervals, according to

```
PARAMETER VOLUME=1  
FEED=FDTIME*(L-AFOOD)/DELT  
AFOOD=INTGRL(L,FEED-CNRTA-CNRTC)  
FOOD = AFOOD/VOLUME
```

L is amount of food given daily after removal of the food that is left over from the previous day and either equal to 1 or 0.5 loop of bacteria. The amount of food during the day is continuously diminished by consumption by the *P.aurelia* and *P.caudatum* species, but only once a day replenished to the original level.

---

#### Exercise 43

Why is VOLUME equal to 1 rather than 5? Why is it advisable to distinguish between AFOOD and FOOD?

---

Due to the discontinuity in the food supply and in the sampling it is necessary to integrate according to the

METHOD RECT

and to specify DELT also on the TIMER card:

```
TIMER FINTIM=16, DELT=0.01, OUTDEL=1
```

---

#### Exercise 44

Why is METHOD RECT used here, and not METHOD RKS?

---

For comparison with Gause's data it suffices to print the size of the samples SPLEA and SPLEC each day, but a more frequent printing of population numbers is necessary to study the behaviour of the simulated populations within a day. To complete the program all initial values and parameters must be defined on parameter cards.

There are eight parameters: CONVF, RSW, RDR and MRDIG that have to be derived from the experimental data and must be substituted in the simulation program. In principle, these can be found by trial and error, using some goodness of fit criterion to the observational data. But such a procedure can be started only in practice when the order of magnitude of all the variables concerned are known from a preliminary analysis of the data.

---

#### Exercise 45

Why?

---

### 5.3 Preliminary estimation of parameters

Gause observed that at first the medium remained opaque during the whole day, but that later the medium became transparent within a few hours after the addition of new food. From this he concluded that all food was consumed rapidly, once the size of the population was not far from its maximum. Hence there is sufficient time for digestion and searching so that the maximum size of the population does not depend on the rate of digestion of the food or on the rate of searching water. Instead it depends only on the amount of food given, the conversion factor for food, the relative death rate and the rate of sampling. About  $H \times (RDR + 0.1)$  number of protozoa die or are sampled and  $CONVF \times L$  number grow in the monoculture in a day when the daily food is consumed completely.

In equilibrium situation, these quantities are equal so that  $CONVF \times L = H_{eq} \times (RDR + 0.1)$ .

This equation contains two unknowns; CONVF and RDR, so that another equation is necessary to estimate their values. This second equation can be obtained by considering the growth rate (GR) at the moment that three-quarters of the maximum population size is reached, because Gause observed that the food was exhausted well within a day at this density. The growth rate may be estimated by

$$GR = CONV F \times L - 0.75 \times H_{eq} \times (RDR + 0.1)$$

Combining both equations allows a first estimate of CONV F and RDR.

The rate of searching the water (RSW) and the maximum rate of digestion (MRDIG) are estimated from the dynamics of the populations at the beginning of the experiment. During the early stages, the number of protozoa is so small that the concentration of bacteria stays practically the same during the whole day. The data show that the initial growth rates of the 0.5 and 1 loop series with *P.aurelia* are about the same so that the maximum digestion rate is at least reached at the 0.5 loop concentration. In other words, at this level

$$MRDIG = 0.5 \times RSW$$

but instead of 0.5 a lower value could be more appropriate. This is again an equation with two unknowns, so that another equation is necessary to make a first estimate of both parameters. This second equation can be obtained by considering the initial relative growth rate of the 1 loop series. This relative growth may be estimated from the experimental data and is equal to

$$RGR = MRDIG \times CONV F - RDR - 0.1$$

at least as long as the bacterial concentration is so high that the maximum digestion rate is maintained during the day. For *P.caudatum* the relative growth rate of the 1 loop series is higher than of the 0.5 loop series, so that there is no certainty that the maximum rate of digestion is reached at a bacterial concentration of 1 loop per volume. A first estimate of the parameters may be obtained here by assuming that

$$MRDIG = 1 \times RSW$$

but instead of 1, a higher value could be more appropriate.

Apart from the uncertainty about the exact value of the constant in the equation for the maximum digestion rate, the estimation procedure is also unfavourably affected by the large scattering of the data. This makes it difficult to arrive at a value for the initial relative growth rate. It is therefore still worthwhile to inspect the system for other interrelations between the constants. These are obtained from the observation that the maximum number of *P.aurelia* in both the 1 loop and 0.5 loop series is about 4 times higher than the number of *P.caudatum*

so that probably the *P. aurelia* individuals are about 4 times smaller. Thus, it is logical to assume at first that the conversion factor of food with the unit protozoon loop<sup>-1</sup> is 4 times larger for *P. aurelia* and that the maximum rate of digestion in the unit loop protozoon<sup>-1</sup>day<sup>-1</sup> is 4 times smaller.

---

#### *Exercise 46*

Take the graphs that were drawn for the monocultures in Exercise 40 and estimate for both species and both series, the maximum population size ( $H_{eq}$ ), the growth rate (GR) at the moment that the population equals 0.75 of the maximum and the initial relative growth rate (RGR). Calculate the parameter CONVF, RDR, MRDIG and RSW for both species and both series independently with the 2 x 2 equations given. Make first estimates of these parameters for both species, taking the size of the individual protozoon into account. Try to find as many reasons why these first estimates may be considerably in error.

---

#### *5.4 Final determination of parameters*

There are many reasons why the first estimates, especially of the rate of searching and the maximum rate of digestion are very rough indeed. It is therefore necessary to improve on these by comparing the results of simulation runs with the actual results. In principle the results of the monocultures should only be used for this purpose, but the scattering of the observational data is so large that it is very difficult to estimate the parameters with sufficient accuracy. Fortunately, the results of the competition experiments are also available to improve the estimates. When these results are used, it is implied that the interference between both species as proposed in the model is correct. Thus a comparison between simulated and actual results of the competition series cannot be used to validate this assumption. However, the large scattering of the observational data necessitates this way of working.

Further simulations show that the course of *P. caudatum* in the mixture as characterized by the time at which the maximum population size is reached and the rate of its decline in later stages, is especially governed by the ratio between the searching rates of the water by both species and by the ratio of the relative death rate and the conversion factors. In other words, the differences between both species in this respect are especially manifest in the competitive situation.

**Exercise 47**


Explain why this is so.

Finalize also the simulation program and try to find better estimates of the parameters by trial and error.

After a considerable amount of experimentation with the simulation program the best agreement between simulated and actual results, as judged visually on graphs was obtained with the parameter values listed in Table 4.

**Table 4** Parameter values for *P.aurelia*(A) and *P.caudatum*(C)

	A	C	
Relative death rate (RDR)	0.45	0.45	day <sup>-1</sup>
Conversion factor (CONVF)	3000	750	prot.loop <sup>-1</sup>
Saturation level (MRDIG)	.56x10 <sup>-3</sup>	2.25x10 <sup>-3</sup>	loop.prot <sup>-1</sup> day <sup>-1</sup>
Rate of searching water (RSW)	.006	.006	volume.prot. <sup>-1</sup> day <sup>-1</sup>



**Fig. 15** | Simulated and observational results for the half loop experiment with *P.aurelia* and *P.caudatum*.

The simulated and observational data for the half loop series are summarized in Fig. 15, to show that within the limits of accuracy governed by the scattering of the data there is a good agreement and that the results can at least be understood by assuming that the species only affect each other by competing

for the same food. The ratio area to weight of protozoa and with this their mobility and the chance of meeting bacteria, influences their competitive ability. Therefore the largest species, *P. caudatum*, loses in competition, less bacteria being available for the larger animal per unit biomass of protozoa. Being small is obviously a competitive advantage when the concentration of bacteria is small. At higher concentrations, the consumption is governed by the maximum rate of digestion which is four times larger for the four times larger species. Thus the species match each other in this respect.

In the analysis of the original experiments of Gause, we took into account that during the early stages the concentration of protozoa was so small that the food level hardly decreased during the day and that during later stages the food was rapidly depleted. Although not observed by Gause, we are now in a position to consider in more detail the daily course of food concentration and number of protozoa because these have been simulated. Some of these simulated results are presented in Fig. 16 for further inspection.




Fig. 16 | Simulated course of growth and food supply of a *Paramecium* species in monoculture during a day at the beginning and the end of the experiment.

---

#### Exercise 48

Why is the growth of protozoa during the beginning of the experiments nearly exponential? Why, at the end of the experiment, is the maximum size of the population at some time during the day, larger than the population size, observed at the end of the day? Which is higher, death through natural causes or through sampling?

---

## 5.5 Stochastic aspects

The simulation program presented in the previous sections is fully deterministic and does not explain at all the large scatter of the observational data. There are, however, two stochastic phenomena that are accessible for further analysis. These are the sampling process and the death process. As far as the sampling process is concerned, it was assumed that exactly 1/10 of the population is taken away when 1/10 of the solution is removed. However this assumption is not true. The protozoa are, after stirring, randomly distributed throughout the solution so that either more or fewer protozoa than the average may actually be found. To simulate the actual number that are in the sample, this number must be drawn out of a probability function around the average. Since the number of protozoa may be small, the probability function of Poisson is used.

This function can be introduced into the simulation program by replacing the statement for the sample size in the MACRO GROWTH:

```
SPLE=FDTIME*0.1*H
```

by the statements

```
AVSMP=0.1*H  
SPLE=POISS(P,AVSMP,1.)
```

The first statement calculates the size of the average sample at every time-interval and the second statement invokes a MACRO called POISS that assigns an appropriate random number to the sample size. The value of the first variable in this argument is an odd number, to be specified on a parameter card (outside the MACRO) and is necessary to start the process of generating random numbers. The second variable in the argument is the average number of protozoa in the sample and the number 1 indicates that the sample is taken with an interval of one day. As far as the death process is concerned, the amount of protozoa that die during one time-step (AD) is on the average:

```
AD=H*RDR*DELT
```

and the random number that dies is accordingly

```
RD=POISS(P,AD,DELT)
```

The third variable in the argument is DELT because death occurs every time-step. The rate of dying is now calculated from the

amount that dies by dividing again by DELT with

$$DR=RD/DELT$$

These three statements replace the statement

$$DR=RDR*H$$

in the original MACRO GROWTH.

Results of some simulations are presented in Fig. 17, which shows the growth of *P.caudatum* in the half loop series. The




Fig. 17 | Observational and simulated results of *P.caudatum* in the half loop experiment under various assumptions regarding the operation of random processes

solid line is the growth curve obtained by deterministic simulation. The Roman crosses are the simulated results with a deterministic sampling process and a random death process and the Greek crosses present the simulated results that are obtained with a stochastic sampling and stochastic death process. The open dots are the observed data. It must be concluded from these results that the main contribution to the variability is due to the method of sampling and that the quality of the experiment would have been very much improved if some method of measuring the whole population had been introduced. The scattering due to the stochastic sampling is much larger

than due to the stochastic death process, although about 45 percent of the population dies during one day and only 10 percent is sampled. A simple calculation can explain this. Let the equilibrium population be 1000 individuals. The average sample size is then 100 and the standard deviation is according to the binomial probability function  $\sqrt{0.9 \times 0.1 \times 1000} = 9.5$ . Each day an average of 450 individuals dies out of 1000 and the standard deviation of this number is  $\sqrt{0.45 \times 0.55 \times 1000} = 16$ . Because one tenth is sampled, the standard deviation is reduced to 1.6, which is only one sixth of the standard deviation caused by the sampling process itself. Moreover, the death process is distributed over the day so that some deviation may be even levelled by negative feedback throughout the day.

---

#### Exercise 49

Explain now why the scattering of the observational data for *P.aurelia* is much smaller than for *P.caudatum*.

---

### 5.6 The programming of probability functions

To simulate stochastic processes, CSMP contains a so-called random generator that generates numbers between 0 and 1 out of a standard uniform probability function and a Gaussian generator that generates numbers out of a normal probability function with a specified average and standard deviation. The language does, however, not contain a Poisson generator. Such a generator can be most conveniently introduced by the user in the form of a MACRO. In this section the content of this MACRO is described. Unfortunately it is only understandable for the reader who has some knowledge of FORTRAN and probability calculus.

The heading is:

```
MACRO N=POISS(P,MEAN,PERIOD)
```

DO loops and IF statements as such cannot be sorted by CSMP, so that the statements are given in computational order. This is indicated by the card:

```
PROCEDURAL
```

The invoked MACRO is then sorted within the main program with the names for P, MEAN and PERIOD as input and N as output. If the time is not equal to n times PERIOD, the sampling need not be executed and N equals zero:

```
N=0
```

```
IF(IMPULS(0.,PERIOD).LT.0.5) GO TO 1
```

whereby 1 is a CONTINUE statement at the end of the MACRO and LT means 'less than'. If the expectation value is larger than 25, the Poisson distribution is sufficiently approximated by a Gauss distribution with a standard deviation equal to the square root of the average.

```
IF(MEAN.LT.25) GO TO 2
N=GAUSS(P,MEAN, SORT(MEAN))
GO TO 1
2 CONTINUE
```

The Gauss function is a CSMP function that executes the random choice out of a normal distribution. P can be any odd integer. The second and the third argument represent the average and the standard deviation, respectively.

Below a number of 25, the deviation between the Poisson distribution and the Gauss distribution becomes too large. To execute the selection from the Poisson distribution a number is first drawn between 0 and 1 according to a standard uniform probability function. This is done by a CSMP function:

```
LOT=RNDGEN(P)
```

P is again the odd integer.

Then this number is used to read the output from a cumulative Poisson distribution function. The cumulative Poisson distribution is obtained by a series development. The probability of a number to be smaller than or equal to 0, 1, 2, 3... is given by  $e^{-z}(1+z/1!+z^2/2!+z^3/3!...)$  where z is its average. The program is as follows:

```
SUM=1
PROD=1
EMINZ=EXP(-MEAN)
DO 4 J=1,100
IF(LOT.GT.SUM*EMINZ) GO TO 3
N=J-1
GO TO 1
3 CONTINUE
PROD=PROD*MEAN/J
SUM=SUM+PROD
4 CONTINUE
```

Then the MACRO is concluded by

1 CONTINUE  
ENDMAC

The PROCEDURAL card also ensures that all statements of the MACRO are sorted as one block at a place where P, MEAN and PERIOD are available and N is needed, as indicated by the MACRO definition card.

By using a Poisson probability distribution function which is for higher numbers replaced by the Gaussian function, numbers higher than the total number of individuals in the population may be drawn. This problem does not exist when the sampling process is formulated on basic principles.

For this purpose, the protozoa in the solution are considered analogous to the black balls and the volumes of water equal to the volume of protozoa analogous to the white balls in the traditional jar with coloured balls. The following symbols can now be defined:

N: the total number of volume elements and n: the number drawn,  
B: the total number of protozoa and b: the number drawn (black balls),  
W: the total number of volume elements water and w: the number drawn (white balls).

Hence,  $N = B + W$  and  $n = b + w$ .

According to basis theory, the number of combinations of drawing a number of n balls out of a total of N is:

$$\frac{N!}{(N-n)!n!} \quad (5.1)$$

Similar expressions hold for the white and black balls, so that the number of combinations of drawing b black balls and w white balls equals the product

$$\frac{B!}{(B-b)!b!} \times \frac{W!}{(W-w)!w!} \quad (5.2)$$

To obtain the probability of obtaining b black and w white balls in the sample, this expression must be divided by the total number of combinations. This gives

$$\frac{B!}{(B-b)!b!} \times \frac{W!}{(W-w)!w!} \times \frac{(N-n)!n!}{N!} \quad (5.3)$$

In the present situation, the volume of water is infinite with respect to the volume of paramecia, so that W and w are infinite with respect to B and b.

Hence when a fraction f of the volume is sampled the total number of volume elements (water and protozoa) is fixed according to

$$n = f \cdot N$$

Since also  $W = N - B$  and  $w = n - b$ , expression (5.3) for the probability can be transformed into

$$\frac{B!(N-B)!(f \cdot N)!((1-f) \cdot N)!}{(B-b)!b!((1-f) \cdot N - B + b)!(f \cdot N - b)!N!} \quad (5.4)$$

which approaches to

$$\frac{B!f^b(1-f)^{B-b}}{(B-b)!b!} \quad (5.5)$$

with increasing  $N$ .

This is a binomial probability distribution function.

The chance to find 0, 1, 2, ... paramecia in the sampled volume is now

Number	Chance
0	$(1-f)^B$
1	$Bf(1-f)^{B-1}$
2	$B(B-1)f^2(1-f)^{B-2}/2$

The sampling may now be programmed as follows:

```
MACRO N=BINOM(B,F,P,PERIOD)
```

$N$  is the number which is actually drawn,  $B$  is the total number of paramecia in the vessel,  $F$  is the fraction of the liquid which is taken out,  $P$  is some odd integer and PERIOD is the interval of sampling.

PROCEDURAL

N=0.

IF(IMPULS(0.,PERIOD).LT.0.5) GO TO 100

LOT=RNDGEN(P)

PROD=(1.-F)\*\*B

SUM=0.

DO 400 J=1, 100

SUM=SUM+PROD

PROD=PROD\*(B-J+1)\*F/(J\*(1.-F))

IF(LOT.GT.SUM) GO TO 400

N=J-1

GO TO 100

400 CONTINUE

100 CONTINUE

ENDMAC

Still one remark should be made. If the expectation value of the sample is small and  $f$  is small, the expression for the probability distribution may be simplified even more. The expectation value is then  $f \times B$ . If this product stays at a constant low value, then  $f$  decreases with increasing  $B$ . The ratio  $B!/(B-b)!$  approaches  $B^b$  and the power  $(1-f)^{B-b}$  approaches  $(1-f)^B$  which can be replaced by  $e^{-fxB}$ . Substitution in the expression for the binomial distribution Eqn (5.5) gives

$$\frac{(f \times B)^b \times e^{-fxB}}{b!} \tag{5.6}$$

Replacing the expectation value  $f \times B$  by  $z$  gives

$$\frac{z^b e^{-z}}{b!} \tag{5.7}$$

which is the Poisson probability distribution function.

## 6 Modelling of development, dispersion and diffusion

### 6.1 Introduction

In Chapter 1 it was stated that systems ecology is based on the assumption that the state of an ecosystem at any particular time can be expressed quantitatively and that changes in the system can be described in mathematical terms. Various models of ecosystems were given and in all examples it was possible to use a very limited number of state variables and associated rate equations. This is not surprising. Yeast and paramecium are simple organisms and the responses as a population are hardly dependent on such attributes as size and stage of development. The small grain example concerns more complicated organisms that are synchronized in time and whose responses strongly depend on size, stage of development and physiological conditions and on the continuously changing physical environment. However, here the problem was simplified by a model with a limited number of state variables by only treating the interference of similar plant species. No attempt was made to construct a model of the growth and development of form and function of the single species.

Although we may accept that the ultimate purpose of biology in general and ecosystems analysis in particular, is the construction of models that predict growth and development of single and interfering species in natural conditions, we must admit that at present this goal is unrealistic. The knowledge of the relevant processes is quantitatively, but also qualitatively, far too fragmentary and even if this were not so, there would be serious modelling problems, because the number of state variables involved would be very large.

Obviously, it is necessary to limit the goals of systems analyses drastically to proceed at all. Rather than analysing all aspects, a distinction is often made between growth and morphogenesis: growth is the main subject of study and morphogenesis is taken more or less for granted. For instance, it is assumed that maize plants develop out of maize seeds, wheat plants out of wheat seeds and spiders out of spider's eggs. In models, such broad assumptions are made operative by introducing preconceived information on the development of the species.

For instance, in a model of a wheat plant, a germination, a vegetative and generative stage are distinguished a priori, and it is assumed that nine to eleven leaves develop in the vegetative stage, and that the main growing point develops into the reproductive organ. Likewise, a distinction is made a priori between the successive development stages of insects, such as eggs, instars, pupae and imagos. What is left to be simulated is the growth within various stages and the rate of their development to subsequent stages. Biologists that are interested in understanding the development of form and function may have another view on the matter and may argue that this approach is too simplified, but appreciation of simplification is more a matter of goal than of principle.

What are the consequences of such an approach for the technique of modelling? Rather than modelling a system fully in terms of measurable state-variables, it is also characterized by historical information which in its most elementary form becomes a record of age only. This is an external record, because age can be known only when the moment of birth is recorded and cannot be determined as such by means of analyses. On the other hand, when age is recorded, relevant properties may be derived from it by correlation.

For instance, in demographic studies the chances of marriage, childbirth, and death may be arrived at in this way. Individuals are lumped at their birth in age-cohorts. Then the ages of the cohorts are kept track of and from them the number of offspring and deaths in a year are calculated. Such a crude technique may be adequate for warm-blooded animals, but not for plants, insects and many other organisms, as their development rate depends largely on environmental conditions. It is then often attempted to conserve past experience in another variable of state: some physiological age. This may be a simple external integral of the temperature: the temperature sum, but it may also be a numerical characterization of the development stage. As long as such cohorts are characterized by age only, no dispersion occurs. Human individuals that are classified at their birth in the cohort 1970 remain there for their whole lifespan and if nobody is classified in the cohort 1971, this cohort will remain empty. But, as soon as a physiological age criterion is introduced, some individuals that are born early may age slowly and may be overtaken by individuals that are born later. In other words, individuals that belong to the same age-cohort may become dispersed over a range of physiological ages and it is necessary to develop programming techniques that account for such dispersion phenomena.

Such techniques may seem sophisticated, but are in fact still

very primitive, because they are based on correlations between relevant variables and an external record of past experience. They avoid the problem of modelling the main aspects of development of form and function on basic principles.

## 6.2 Physiological age and development stage

The development stage of warm-blooded animals may be often characterized by a record of the chronological age only. This situation is completely different for many other organisms, such as insects and plants.

Temperature is then often the main determinant of physiological age, so that the development stage is often accounted for by the temperature sum:

$$TS = \text{INTGRL}(0., \text{AMAX1}(0., T - TT))$$

in which  $T$  is the current temperature and  $TT$  a threshold temperature below which the development processes proceed at a negligible rate. Based on experimental results, it is then assumed that certain development stages are reached at certain values of the temperature sum. For instance, it may be found that the threshold value for maize is  $12^\circ\text{C}$  and that tasselling occurs at a temperature sum of 400 degree-days and the plant ripens at a temperature sum of 700 degree-days.

If this approach is taken, it is implicitly assumed that the development rate of the species is proportional to the temperature above the threshold value. However, in general, there is also a non-linear response of development rate to temperature in the higher ranges, as is illustrated in Fig. 18 for two plant species. Here, a constant temperature during growth is given along the horizontal axis and the development rate along the vertical axis, the latter being defined as the inverse of the number of days from emergence to flowering or tasselling. A more sensible approach seems therefore to consider the development stage of the plant as defined by

$$DVS = \text{INTGRL}(0., DVR)$$

in which the development rate in  $\text{day}^{-1}$  is a function of the current temperature according to

$$\begin{aligned} DVR &= \text{AFGEN}(DVRTB, TEMP) \\ \text{FUNCTION } DVRTB &= (0., 0.), (12., 0.), (26., 0.035), (28., 0.038), \dots \\ &(30., 0.039), (40., 0.041) \end{aligned}$$

flowering or tasselling being reached when  $DVS$  passes the




Fig. 18 | The development rate of the plant species maize and oats in relation to temperature at a daylength of 14 hours.

value of one development unit.

It is assumed that the influence of temperature on the development rate is the same during the whole period of growth, and this assumption is confirmed by the well-known fact that at constant temperature the time between appearance of successive leaves is constant (de Wit et al., 1970) and that accordingly, a certain calculated development stage fully characterizes the number of leaves and other morphological properties of the plant. If the temperature sum or the development stage approach is used, the question remains whether the response in rate of development is immediate or not. It makes a considerable difference whether the temperature with its fluctuations throughout the day and from day to day should be used or some average temperature over one day or more.

---

#### Exercise 50

Calculate manually the development stage of maize after 20 days when

- a the temperature is 14 °C all the time,
- b the temperature is 7 °C for 12 hours of the day and 21 °C for the other 12 hours,
- c the temperature is 30 °C all the time,
- d the temperature is 40 °C for 12 hours of the day and 20 °C for the other 12 hours.

Explain the differences.

---

At least for plants (de Wit et al., 1970) there are many indications that the response to temperature is instantaneous, so that use of average daily temperatures may lead to considerable errors.

Of course, there are other problems. The development rate may be influenced by daylength or even rate of biomass growth. Like temperature, these factors may also be accounted for on an experimental basis. However, interactions are often so complicated that the development of the plant cannot be accounted for by a simple physiological age. Then more than one characteristic for the development stage may be considered. But problems can then multiply at such a disastrous rate that it is better to take the hard road: modelling of the morphogenesis processes.

What has been said in this section holds in principle for other plant growth stages and other organisms as will be shown later by means of various examples.

### 6.3 Demographic models

#### 6.3.1 Age-classes

Decay of radioactive material occurs with a constant relative rate, apart from random effects that become manifest at low rates. Similar decay processes were assumed to exist for protozoa. However, they are more the exception than the rule with living organisms. In general these organisms develop and age accordingly, and their chances of dying appear to increase with increasing age.

To simulate such situations it is necessary to have the age distribution of the population at hand. Now it is practically impossible and for most applications unnecessary to memorize the age of each individual. Instead it suffices to memorize the number of individuals in age-classes. For instance, in demographic studies it is customary to classify human beings according to their age in years. Of course this classification in years is an arbitrary choice, depending on purpose. For some applications it would be better to classify according to age in months and for others it would suffice to classify in units of five or ten years.

Such age distributions are memorized conveniently by using the INDEX feature to create a series of age-classes. For instance, human populations may range in age from 0 to about 100 years and if an age distribution has to be stored in age-classes of

five years, it suffices to write:

```
H'1,20'=INTGRL(HI'1,20',FLOW'1,20'-FLOW'2,21')
FLOW1=TBR
FLOW'2,21'=PUSHD*H'1,20'
PUSHD=PUSH/DELT
PUSH=IMPULS(5.,5.)
INCON HI'1,20'= (20 data)
```

Printed output of the variables H1, H2...,H20 may be requested by

```
PRINT H'1,20'
```

Here PUSH has the value zero, except once every five years, when its value is set to 1. Only at that moment are the contents of all age-classes shifted to the next one. As in other examples (Section 5) this shift is achieved by introducing a rate of change which is equal to the shifted amount divided by the time interval of integration. The whole age distribution of the population is stored in this way with a resolution of five years.

So far the simulation program does not contain death rates. These can be accounted for by subtraction of an additional rate DR'1,20' in the integral statement. The death rates are calculated from the relative death rates according to

```
DR'1,20'=RDR'1,20'*H'1,20'
```

The total birth rate TBR is the sum of the birth rates produced by each age-class.

These are given by

```
BR'1,20'=H'1,20'*RBR'1,20'
```

The relative birth rates are given on a parameter card:

```
PARAMETER RBR '1,20' = (20 data)
```

The twenty values are summed with

```
TBR=SUM1(BR'1,20')
```

The total birth rate TBR can also be calculated more directly as

```
TBR=SUMX(H'1,20',RBR'1,20')
```

by using the preprocessor function SUMX, which takes the sum of the products of pairs. The total population is obtained by  $TH=SUM1(H'1,20')$ .

In demographic studies it is often customary to report relative birth and death rates as an annual total rather than as an instantaneous rate and then it is best to integrate with time intervals of one year.

---

### Exercise 51

What is the difference between relative death and birth rates reported as an annual total rather than as an instantaneous relative rate? Why is it necessary to integrate according to the METHOD RECT?

---

### 6.3.2 Errors of approximation

The lumping of populations into age-classes introduces errors of approximation. These are small and negligible when many classes are used, but may be worth considering if a limited number of classes are distinguished. For instance, in a demographic model of a human population, age-classes of 0-5, 5-10, 10-15 years may be distinguished. Every five years the contents of the classes are shifted one place, so that generally the residence time in each class is five years. However for the first class the residence time is shorter, because it has a continuous inflow from the birth rate. Only the individuals born just after a shift will stay here five years. As time proceeds the residence time of individuals born later will become progressively shorter. On the average the residence time in the first class will be the half of the 'interval of pushing'. In other words, just after each shift the first age-class contains only individuals close to zero years, and just before the next shift the individuals are 0-5 years. The next age-class contains individuals of 0-5 years just after the shift, and those of 5-10 years just before the next shift. With a constant birth rate, the average age of the individuals in the age-classes is therefore not 2.5, 7.5, 12.5 years and so on, but 1.25, 5, 10 years and so on. Therefore the age-classes lie between 2.5-7.5, 7.5-12.5 years and so on. The first class covers then the period between -2.5 and +2.5 years. Since birth occurs at zero years, the average age in this class is 1.25 years.

There is still a pitfall in initialization. At time zero, each age-class will be initialized with the number of individuals that are between the above given boundaries. Then it takes only 2.5 years before the centre passes to the next class. Therefore the first push should not occur after five years, but after

2.5 years, which can be achieved with

PUSH=IMPULS(2.5,5.)

Another error is best illustrated by considering the integral for the first age-class, under the assumption that the total birth rate (TBR) is zero for some time. At the time when PUSH = 1, this integral is diminished by its own content and by the number of deaths during that time interval so that at the next moment the content of the integral is  $-H \times RDR \times DELT$  rather than zero. The reason is that too many individuals were shifted. The number that die during this time interval, should not be removed another time by shifting. Therefore it is necessary to shift not the whole content of the integral but its content minus the number that is lost by death during that time-interval. Such a procedure is realized by

FLOW'2,21'=PUSHD\*H'1,20'\*(1.-RDR'1,20'\*DELT)

---

### Exercise 52

The following tables contain demographic data of the population of the Netherlands on 31 December 1968. The data are grouped in classes with their centres at 1.25, 5, 10, ... years (Set 1) and 2.5, 7.5, 12.5, ... years (Set 2).

Write a simulation program for the growth of the population in the Netherlands, using age-cohorts of five years. Which set of data must be used, Set 1 or Set 2?

Why is the time interval of integration a half year?

Simulate over a period of 50 years and ask for the total men and woman and the relative composition of the population as to sex and age every five years.

Determine also the number of graves after 50 years, if these are maintained for a period of 50, 25 and 10 years.

Death rates during the first year of life are much higher than during the next years. Is there a simple way of taking this into account?

Population size

Class centre in years		Number of men		Number of women	
Set 1	Set 2	Set 1	Set 2	Set 1	Set 2
1.25		305 000		291 000	
	2.5		611 000		582 000
5		612 000		584 000	
	7.5		613 000		587 000
10		597 000		570 000	
	12.5		580 000		553 000
15		575 000		548 000	
	17.5		569 000		543 000
20		576 000		548 000	
	22.5		583 000		554 000
25		517 000		487 000	
	27.5		452 000		420 000
30		429 000		400 000	
	32.5		405 000		380 000
35		399 000		380 000	
	37.5		393 000		381 000
40		382 000		379 000	
	42.5		371 000		378 000
45		367 000		377 000	
	47.5		362 000		376 000
50		338 000		353 000	
	52.5		314 000		330 000
55		306 000		327 000	
	57.5		297 000		323 000
60		280 000		310 000	
	62.5		262 000		298 000
65		223 000		262 000	
	67.5		184 000		226 000
70		184 000		226 000	
	72.5		184 000		226 000
75		150 000		180 000	
	77.5		120 000		150 000
80		90 000		110 000	
	82.5		60 000		70 000
85		40 000		60 000	
	87.5		20 000		25 000
90		13 000		23 000	
	92.5		3 000		13 000
and more					
Total			6 383 000		6 415 000

Death rates per thousand men and woman per year.

Class centr years		Men		Women	
Set 1	Set 2	Set 1	Set 2	Set 1	Set 2
1.25		15.6		11.4	
	2.5		3.9		2.8
5		1.8		1.2	
	7.5		0.7		0.8
10		0.5		0.3	
	12.5		0.5		0.3
15		0.5		0.3	
	17.5		0.6		0.3
20		0.7		0.4	
	22.5		0.9		0.4
25		1.0		0.4	
	27.5		1.0		0.5
30		1.2		0.6	
	32.5		1.4		0.8
35		1.5		1.0	
	37.5		1.8		1.2
40		2.2		1.5	
	42.5		3.1		2.0
45		4.0		2.5	
	47.5		5.2		3.2
50		6.5		4.0	
	52.5		7.8		4.7
55		9.0		5.5	
	57.5		10.7		6.7
60		11.5		8.0	
	62.5		13.7		10.5
65		16.0		13.0	
	67.5		25.5		16.5
70		35.0		20.0	
	72.5		52.0		35.0
75		70.0		50.0	
	77.5		110.0		85.0
80		150.0		120.0	
	82.5		200.		180.
85		300.		250.	
	87.5		400.		380.
90		600.		500.	
	92.5		900.		760.

The relative number of births per year per age group of the mother

Class centre  
years

Class centre years		Class centre years	
Set 1	Set 2	Set 1	Set 2
15		0	
	17.5		0.022
20		0.091	
	22.5		0.137
25		0.159	
	27.5		0.188
30		0.152	
	32.5		0.113
35		0.084	
	37.5		0.055
40		0.036	
	42.5		0.016
45		0.010	
	47.5		0.002
50		0	

The ratio between the number of boys and number of girls that are born, is 1.048

### 6.3.3 The matrix method

If DELT equals the length of the class in a program with age-classes, the contents are shifted every time-step one place and are diminished at the same time by the amount died. If the relative death and birth rates do not change with time a matrix method, introduced by Leslie (1945), may be applied to predict the relative composition and the relative growth rate of the population in the stationary state. This is not a simulation method, but will be discussed here because it shows some advantages and disadvantages of matrix algebra versus simulation in demographic studies.

Let the contents of the age-classes be the elements of a vector. If there are ten age-classes, the vector is ten dimensional. The number of individuals in each age-class one time-interval later is now found by multiplying this vector by a matrix as in Fig. 19.




Fig. 19 | The matrix method,  $H_1$ - $H_{10}$  are the number of individuals in the age-classes,  $F_1$ - $F_{10}$  the relative number of births in and  $S_1$ - $S_9$  the fraction of each class that survive the time-span  $\text{DELTA}$ .

At the right side, the vector at time  $T$  and at the left side the vector at time  $T + \text{DELTA}$  is given. The latter is found by multiplication of the vector at time  $T$  by the matrix. In the matrix,  $F_i$  is the relative number of births per time step in class  $i$  and  $S_i$  is the fraction of class  $i$  that passes to  $i + 1$ ; in other words 1 minus the fraction that dies during a time interval.

It has been proven that repeated multiplication of a vector by a matrix results, in due course, in a vector that has a constant relative composition and whose length increases by a constant factor each time. This vector is called the dominant eigenvector and the constant factor is its corresponding eigenvalue.

Hence the population will approach a stable age-distribution with a constant relative growth rate, provided that birth and death rates are constant. The standard method to find the dominant eigenvector and its corresponding eigenvalue is the power method (Faddeev & Faddeeva, 1964) in which the multiplication is repeated until a stable relative composition is reached.

This method is therefore very similar to the simulation method and has no computational advantage.

However, there are shorter methods to achieve the eigenvector for matrixes that contain zeros except in the top row and one diagonal. One of these methods is presented in Fig. 20, which is de Jonge's modification of the method of Gauss-Seidel (pers. commun.). This iterative method is very cheap in terms of computing time. It is not explained here because it requires

```

TITLE MATRIX METHOD APPLIED TO GROWTH OF THE NETHERLANDS POPULATION
PARAMETER C1=1.,R=7.,P=1.
INITIAL
PARAM F'1,17'=3*0.,.055,.343,.47,.282,.137,.04,.005,7*0.
PARAM S'1,16'=.965,.996,.998,.998,.997,.996,.995,.993,.988,.98,.97, ...
      .96,.94,.92,.8,.5
      N=0.
*   ITERATION
NOSORT
  4 CONTINUE
    N=N+1.
    IF(N.GT.20.) GO TO 6
    C'2,17'=C'1,16'*S'1,16'/P
    Q1=F1+C1
    Q'2,17'=Q'1,16'+F'2,17'+C'2,17'
    Q=Q17/C1
    WRITE(6,800)Q,C1,C2,C3,C4,C5,C6,C7,C8,C9,C10,C11,C12,C13,C14, ...
    C15,C16,C17
  800 FORMAT(1H ,F8.5/,9F8.5/,8F8.5///)
    IF(ABS(P-Q).LT.1.E-6) GO TO 6
    P=(R*P+Q)/(R+1.)
    GO TO 4
  6 CONTINUE
DYNAMIC
TIMER FINTIM=1.,DELT=1.
END
STOP
ENDJOB

```

Fig. 20 | An iterative determination of the eigenvector and its eigenvalue of a matrix as in Fig. 19, written as an INITIAL section in CSMP.

some knowledge of matrix algebra.

The method gives directly the eventual stable age-distribution and the corresponding relative growth rate, which is the eigenvalue minus one. The method does not give the total population after  $n$  years. To achieve this important value, the power method or straightforward simulation must be applied.

## 6.4 Germination models

### 6.4.1 Boxcar train without dispersion

Like the development of plants, the germination of seeds or the hatching of eggs may take some time, which depends on environmental conditions, especially temperature. If a certain amount of seeds is placed suddenly in a position where the germination process may start, its germination stage at any moment may be defined by

$$GS = \text{INTGRL}(0., VDV)$$

in which the velocity of development in  $\text{day}^{-1}$  is defined as a function of temperature by

VDV=AFGEN(VDVTB,TEMP)

FUNCTION VDVTB=(10.,0.065),(15.,0.143),(20.,0.143)

The data hold for seeds of the winter annual *Veronica arvensis*, that have been stored for 15 weeks (Janssen, 1973).

---

#### Exercise 53

Write a simulation program for the germination stage, in which the temperature varies sinusoidally with the time of day with an amplitude of five degrees, and an average of 15 degrees (see also Fig. 2). The computation may be terminated as soon as the germination stage passes the value 1. What does this mean? How is this achieved?

---

The above procedure may be used to follow the development of one batch of seeds. However, it is easy to visualize a situation with seeds in different stages of germination and then their age-distribution has to be taken into account. For this purpose, classes have to be distinguished and because development is very much a function of temperature, these must be development classes rather than age-classes. Hence, the contents must not be shifted at preset time-intervals, but at the moments that the development stage is increased by the inverse of the number of classes (N).

---

#### Exercise 54

Why  $1/N$ ?

---

This shift is achieved by defining a 'PUSH' according to

PUSH=INSW(GS- $1/N$ ,0.,1.)

GS=INTGRL(0.,VDV-PUSH\* $1/(N*DELT)$ )

Here PUSH is set to one by the INSWitch, at the moment GS is larger than  $1/N$ . This moves the contents of the classes and decreases at the same time the integral GS by the amount  $1/N$ , resetting this integral at the correct value close to 0. GS is increased again at the proper rate by the velocity of development.

---

#### Exercise 55

Write now a simulation program for the germination of *Veronica arvensis* seeds at 20 °C with ten development classes. Execute the program introducing an initial amount of 1000 seeds at time

zero. At which moment do these seeds germinate?

---

#### 6.4.2 Boxcar train with constant relative dispersion

Usually germination does not take the same number of days for different seeds, because neither the seeds nor their micro-environments are exactly the same. The overall effect is illustrated in Table 5, where the percentage germination of a batch of *Veronica arvensis* is given.

Table 5 Germination percentages of *Veronica arvensis* seeds at 10 °C, stored for 15 weeks

---

Day	10	13	14	15	16	20	22	27
percent germ.	1	12	28	46	56	87	91	100

---

#### Exercise 56

Make graphs of the percentage germination and the rate of germination against time. Calculate the average time of germination and its standard deviation from the data of Table 5.

---

The time curve for the rate of germination has the bell-shaped form of the Gaussian distribution function. It will be shown that such distribution functions are obtained also by simulation, if the contents of the classes are not pushed at certain moments but moved continuously from one development class to the next with a rate that is proportional to the rate of development. In the most simple situation, only one development class is considered - ungerminated seeds - and germination is described as an exponential decay process of ungerminated seeds according to

$$H = \text{INTGRL}(H_1, -RTG)$$

in which the rate of germination is given by

$$RTG = H * RDV$$

and RDV is the relative rate of development, or germination. The total amount of seed that are germinated equals then

$$TG = \text{INTGRL}(0., RTG)$$

H, RTG and TG are presented in Fig. 21. For obvious reasons (see also Chapter 2) H and RTG decrease exponentially with time




Fig. 21 | Amount of ungerminated seeds (H), germination rate (RTG) and amount of germinated seeds (TG), when germination is described as an exponential decay of ungerminated seeds with a relative germination rate of  $1 \text{ day}^{-1}$ .

and TG approaches HI accordingly. The average germination period is the integral of the rate of germination at any moment multiplied by the time that has elapsed since the start of the process, standardized at a unit amount of seed ( $H = 1$ ). This is the standardized surface under the curve of H versus time in Fig. 21 and in CSMP notation defined by

$$\text{AGP} = \text{INTGRL}(0., \text{TIME} * \text{RTG} / \text{HI})$$

This integral approaches the average germination period by the time that H approaches zero.

### Exercise 57

Calculate manually the average germination period when out of a batch of 100 seeds:

100 germinate on day 5,

100 germinate on day 10,

50 germinate on day 5, and 50 on day 10

75 germinate on day 5, and 25 on day 10

Finish the simulation program to calculate the average germination period. Use the method RKS for integration and terminate simulation as soon as the content of H is  $1/100$  of its original content. Execute the program for a relative rate of development of 0.01, 0.05, 0.1 and  $0.5 \text{ day}^{-1}$  and take FINTIM equal to 500 days. Multiply the average germination period by the relative germination rate. What is the dimension of this product  $\text{AGP} \times \text{RDV}$  and what is its numerical value?

Now prove the equality of the inverse of the relative germination rate and the average germination period by making use of the analytical expression:

$$H = HI \times e^{-RVD \times T}$$

and of the equality:

$$\frac{1}{HI} \int_0^{HI} T \times dH = - \frac{1}{HI} \int_0^{\infty} T \times \frac{dH}{dT} \times dT$$


---

If the above exercise is done correctly it will be clear that the product of the average germination period and the relative rate of development (APG x RDV) is always 1. Hence the relative rate of development as defined in the above program may be replaced by the inverse of the average germination period. The results that are obtained by considering only one development class of ungerminated seeds describe much more a decay process of seeds than a germination process. This is different when more development classes are considered, as is again most conveniently done by means of the INDEX feature. For instance, a germination process that is described by means of ten development classes may be programmed as follows:

```

H1=INTGRL(HI,-FLOW2)
H'2,10'=INTGRL(0., FLOW '2,10'-FLOW'3,11')
FLOW'2,11'=H'1,10'/REST
TG=INTGRL(0.,FLOW11)

```

Obviously, when the average germination period is AGP and the number of classes ten, then the residence time (REST) in each class is

$$REST=AGP/10$$


---

#### *Exercise 58*

What is the time constant of this system?

---

The average germination period may be again a function of the environmental conditions and the time interval of integration should be a tiny fraction of REST.

The rate of germination and the cumulative amount of germinated seeds - the breakthrough curve - are given in Fig. 22 by the curves marked 10, it being assumed that the average germination period is 20 days. The curves marked 5 and 20 hold when 5 and

20 development classes are distinguished. The form of the curve suggests that the simulation procedure leads to a Gaussian distribution function of germination, at least when a sufficient number of classes are used. With low class numbers, the results suggest a Poisson distribution. A mathematical analysis (Goudriaan, 1973) showed that this suggestion is indeed correct and that the relation between residence time (REST) or the average germination period (AGP = REST x N), the number of classes (N) and the standard deviation of germination (S) is given by

$$N = S^2 / \text{REST}^2 = \text{AGP}^2 / S^2 \quad (6.1)$$

provided that the time interval of integration (DELTA) is small enough. This relation gives the number of development classes, that are necessary to achieve a certain relative dispersion (S/AGP), independent of the average germination period. The method is rather flexible. It is not necessary to start with a given batch of seeds, and death rates depending on conditions may be introduced at any development stage. Moreover, the transfer of contents is continuous, so that the method RKS with a self-adapting time interval of integration may be used.




Fig. 22 | The rate of germination and the breakthrough curve, when 5, 10 and 20 development classes are considered and the average germination period is 20 days.

### 6.4.3 Boxcar train with controlled dispersion

Two methods to simulate germination have been discussed. The first method does not introduce any dispersion and the second method gives a constant relative dispersion, once the number of development classes is fixed. There are, however, a few remaining problems. In the first place, the number of classes is as large as 100 when a relative dispersion of 10 percent is to be simulated. In the second place, it is impossible to change the relative dispersion according to conditions, because the numbers of classes cannot be varied during simulation. Both problems may be overcome by following an intermediate course, in which a variable fraction  $F$  (between 0 and 1) of the contents of each class is shifted once every fraction  $F$  of the residence time in a class. For ten development classes the program is as follows:

```
H1=INTGRL(HI,-FLOW2)
H'2,10'=INTGRL(0.,FLOW'2,10'-FLOW'3,11')
FLOW'2,11'=H'1,10'*PUSHDF
PUSHDF=F*PUSHD
PUSHD=INSW(GS-1.,0.,1./DELT)
GS=INTGRL(0.,1./(F*REST)-PUSHD)
```

Inspection of the statements shows that no dispersion is obtained when  $F$  equals 1 and a constant relative dispersion, as defined by Eqn (6.1) when  $F$  is set equal  $DELT/REST$ . It can be shown that for intermediate situations  $F$  must be chosen as

$$F = 1. - N \times \left(\frac{S}{AGP}\right)^2$$

to achieve a standard deviation, equal to  $S$ .

With  $F$  equal to  $DELT/REST$  and  $DELT$  sufficiently small, this equation transforms, of course, into Eqn (6.1).

Fig. 23 gives an example of the result. The continuous curve is obtained by means of 100 development classes and  $F$  equal to  $DELT/REST$ . The dots are the result of using 25 classes and  $F$  equal to 0.75. In both cases the relative dispersion is 0.1, but in the second case the curve is not smooth. The given dots have  $OUTDEL$  as time interval. The discontinuity and the use of METHOD RECT is the penalty that has to be paid for reducing the number of classes and retaining a small dispersion. As has been said, the advantage of the procedure is that  $F$ , and with this dispersion, can now be varied independently of the average germination period and the number of classes that has been




Fig. 23 | Breakthrough curves for 100 development classes with  $F$  equal to zero and for 25 development classes with  $F$  equal to 0.75. The average germination period is 20 days.

chosen.

---

**Exercise 59**

What is the value of  $F$  when the relative dispersion is 0.25 and  $N$  equals 25? What should be done in this situation?

---

This method with controlled dispersion has been used by Janssen (1973) to simulate the germination of *Veronica arvensis* and *Myosotis ramossima* seeds. However useful this method is, it should be realized that by applying this type of simulation, results of experiments are 'mimicked' rather than simulated. The term mimicked is used here to emphasize that the main aim is the summarizing of the experimental results in a program that simulates germination, but that no serious attempts are made at this stage to base the equations and parameters that are used on more detailed physiological knowledge of the processes involved.

---

**Exercise 60**

Complete the following table for  $F = 0.5$ :

TIME	H <sub>1</sub>	H <sub>2</sub>	H <sub>3</sub>	H <sub>4</sub>
0	1	0	0	0
0.5 x REST				
1.0 x REST				
1.5 x REST				

in which REST is the residence time in each class. What is the name of the resulting probability distribution function?

This function was discussed in Section 5.6. Readers with some knowledge of probability calculations should read this section again and answer the following questions.

Express B and f of Section 5.6 in TIME, F and REST.

What kind of probability distribution is obtained when F equals DELT/REST and DELT approaches zero?

---

#### 6.4.4 Refinements

A discussion on the demographic models showed that the age-class that was intended to cover for instance the years 10-15, appeared to cover the years 7.5-12.5; the average age of the class was less by half its range. For continuous flow, this error of lumping does not occur, but the simulation process results in a constant, relative dispersion.

The simulation method with controlled dispersion ranges between two situations: when  $F = 1$ , the error of lumping is fully present and when  $F = \text{DELT}/\text{REST}$  (and DELT small) the error is absent. It can be derived that for any value of F the shift in development of each class equals  $F \times \text{REST}/2$ . Therefore, in front of the first class a 'preclass' is constructed with an average residence time of  $F \times \text{REST}/2$ , so that the centres of the following classes are independent of the value of F. Then

```

H0=INTGRL(0.,FLOW0-FLOW1)
FLOW0=TBR
FLOW1=H0*2./(F*REST)
H'1,20'=INTGRL(HI'1,20',FLOW'1,20'-FLOW'2,21')
FLOW'2,21'=H'1,20'*PUSHDF
PUSHDF=PUSHD*F
PUSHD=INSW(GS-1.,0.,1./DELT)
GS=INTGRL(0.5.,1./(F*REST)-PUSHD)

```

The continuous inflow FLOW0 enters H0 rather than H1. The initial value of H0 is always zero. H1 comprizes the population with an age ranging from zero to REST, H2 from REST till 2\*REST etc, and must be initialized accordingly. The initial value of GS is set at 0.5, so that the first PUSH occurs after  $0.5 \times F \times \text{REST}$ . In this way the initial average age or development within each

class is correctly accounted for.

---

### Exercise 61

Apply this method to simulate the growth of the Netherlands population. Which set of data from Exercise 52 should be used now?

---

### 6.5 The flow of heat in soils

There are considerable similarities between the simulation of ageing and dispersion in populations and of physical diffusion and dispersion processes in time and space. The similarities are illustrated here by developing a simulation program for the flow of heat and temperature variations in the soil with the temperature at the surface as a forcing function.

For this purpose a uniform soil column from an infinite slab is considered which is placed on an insulating layer. To calculate the temperature as a function of time and depth, this column is divided into 25 equal compartments. Heat flow into and out of each compartment is calculated at any instant of time from the temperature difference between compartments and the conductivity between compartments. These heat flows are integrated to follow the heat content of each compartment and thus the temperature.

Simulation is done most conveniently by creating integrals of the heat content via:

$$HC'1,25' = \text{INTGRL}(HCI, NFL'1,25')$$

If the soil is uniform, the compartments are of the same size (TCOM) and the initial temperature (TI) does not vary with depth, then the initial heat content is given by

$$HCI = \text{TCOM} * \text{VHCAP} * \text{TI}$$

in which VHCAP is the volumetric heat capacity of the soil. The net flow into each layer is the difference between the flows over the boundaries:

$$NFL'1,25' = \text{FLW}'1,25' - \text{FLW}'2,26'$$

---

### Exercise 62

Which direction of flow is assumed to be positive?

---

The flow is proportional to the temperature differences between the layers and the conductivity of the soil (COND) and inversely proportional to the distance between the centres of the layers (here also TCOM):

$$FLW'2,25'=(TMP'1,24'-TMP'2,25')*COND/TCOM$$

The flow out of the 25th layer is zero, because the column is placed on an insulating layer. It would also be zero if the column was so long that temperature changes in the last compartment were negligible. Hence:

$$FLW26=0$$

The flow into the first layer is

$$FLW1=(TMPS-TMP1)*COND/(0.5*TCOM)$$

in which the temperature at the surface has to be defined as a forcing function, for instance:

$$TMPS=TAV + TAMPL*SIN(6.2832*TIME/86400.)$$

if a cyclic daily fluctuation is assumed.

---

### Exercise 63

Why is the thickness of the compartment multiplied by 0.5?

What is the unit of time?

What are TAV and TAMPL?

---

The temperature of the compartments is obtained by:

$$TMP'1,25'=HC'1,25'/(TCOM*VHCAP)$$

The integration is best done with

METHOD RKS

and a stationary state of the cyclic variations is obtained in about four days, so that

$$TIMER FINTIM=345600., PRDEL=3600.$$

suffices.

The output of all 25 temperatures and of other relevant parameters are requested with

PRINT TMPS, FLW1, TMP'1,25'

As an example, the parameters are defined with:

PARAMETER TCOM=0.02,COND=0.42,VHCAP=1.05E6, TI=20

with time in seconds, distance in m, heat in joule and temperature in °C.

---

*Exercise 64*

What are the units of all variables and parameters used in the simulation program?

---

With a uniform soil and with a sinusoidal forcing function, the variation of temperature may be also calculated by an analytical solution. This calculation has been done for comparison, the result being presented in Fig. 24. The analytical




Fig. 24 | Analytical and simulated solution for the temperature course in a uniform soil, with a sinusoidal temperature variation at the surface.

and simulated solutions agree within 0.01 percent so that it is not necessary to use very thin compartments for accurate results.

---

*Exercise 65*

Rewrite the program, so that conductivities and heat capacities that vary with depth can be introduced. Is it necessary to use compartments of the same size?

---

In another monograph of this series (de Wit & van Keulen, 1972), simulation programs of this type have been developed to study the transport and diffusion of water, salts and ions in soils, the only difference being that instead of the INDEX feature, the more cumbersome DO-loop feature from FORTRAN is used.

In another monograph of this series (Goudriaan, 1977), similar techniques are also used to simulate micro-meteorological phenomena, but it goes beyond the scope of this monograph to discuss the principles of these.

Finally it is remarked that simulation programs with compartmentalization of space may be used to study dispersion of animals, seeds and spores. However these programs become large when dispersion in two and certainly in three dimensions is considered, so that other techniques must be developed for these purposes.

# 7 Growth and development of *Helminthosporium maydis*

## 7.1 Introduction

*Helminthosporium maydis* is a fungus of corn (*Zea mays*). Especially the leaves may be covered with lesions, which develop microscopic sporophores. These produce spores that are dispersed by wind and rain and so reach new healthy leaf tissue. There they germinate and penetrate the plant tissue; new lesions appear after incubation. Under suitable conditions, the life cycle is completed within a week.

The fungus is responsible for Southern corn leaf blight, a disease that ravaged the corn fields of the USA, especially in 1970. The yield was 15 percent less than that estimated before the disease struck, and losses of half or more were common in the Gulf region. The disease suddenly appeared because the T (Texas) type of cytoplasmic male sterility was applied on a large scale in the hybrid system. This type appeared vulnerable for *H. maydis*, which had existed for a long time in a non-virulent form.

To anticipate the growth of the disease in the field, Waggoner et al. (1972) analysed this new disease and made a simulation program for its growth and development. A comparison of important aspects of simulated results with field observations (Shaner et al., 1972) showed the merit of this approach.

This simulation program for the growth and development of epidemics of *H. maydis* (EPIMAY) is written in FORTRAN and keeps track of the development of the lesions formed on each day after infection. The program is difficult to read. A further analysis of Waggoner & de Wit showed that a simulation program that is more lucid and easier to handle could be developed by using the state variable approach as developed in this monograph.

The meteorological factors that effect the fungus are temperature, light, wetness, wind and rain. The influence of these factors on growth and development of the disease in various life cycles was analysed for the Illinois isolate of race T. of *H. maydis* growing on the corn cultivar PA 602A, F1 hybrid in the laboratory and the greenhouse. Undoubtedly the condition of the host affects the growth responses, but the study was restricted to well fertilized and good growing specimens of the

host. These observations and general knowledge about growth and morphogenesis of the fungus form the basis for the construction of the simulation program. This program is presented in the form of relational diagrams, together with sufficient quantitative information to leave the writing of the actual program to the reader of this monograph.

## 7.2 The weather

Even, if a corn crop is uniform, the micro-meteorological conditions for developing fungus are not the same, but vary with height. The radiation during the day is higher, the wind more turbulent and the leaves are dry longer near the top of the crop than near the soil surface. Programs to simulate the micro-meteorological conditions in the crop were developed and published in a book of this series (Goudriaan, 1977) but these are likely to be of use only after the simulation program for the pathogen is refined. At present the microclimate in the crop is not simulated, but instead the macro-weather factors are employed as forcing functions as some 'average' for the whole crop.

---

### Exercise 66

Why is this a dangerous approach?

---

The parameters are temperature, wind speed, light, rain and the presence of water on the leaves. These can be introduced in the form of function data throughout a season, but for the present it suffices to define a particular daily course of the weather which is repeated every day. The following weather data are assumed for some simulations in this chapter.

```
FUNCTION TEMPT = (0.,14.), (12.,35.), (24.,14.)
FUNCTION WINDT = (0.,1.), (6.,1.), (14.,4.), (19.,2.), (24.,1.)
FUNCTION WETT = (0.,1.), (7.99,1.), (8.,0.), (19.99,0.), ...
(20.,1.), (24.,1.)
FUNCTION LITET = (0.,-1.), (5.99,-1.), (6.,1.), (20.,1.), ...
(20.01,-1.), (24.,-1.)
FUNCTION RAIN = (0.,0.), (24.,0.)
```

The units for temperature, wind and rain are °C, m s<sup>-1</sup> and mm h<sup>-1</sup>, respectively. Especially the temperature course is simplified, to facilitate later analysis of the results. For light and wetness only two conditions are distinguished: light (LITE = 1) and dark (LITE = -1) and wet (WET = 1) and dry leaves (WET = 0).

To read the graphs, time in hours during the day has to be known.

As TIME is expressed in days, the hour of the day may be calculated with

```
HOUR = 24.*(TIME-AINT(TIME))
```

in which the function AINT(TIME) conserves the integer part of time, and assumes, for instance, the value 6 when time is between six and seven days. HOUR can also be defined by

```
HOUR = 24.*AMOD(TIME,1.)
```

where AMOD generates a sawtooth function with period 1. Although it does not belong to the weather section, the growth of the crop must be considered. Simulation of a disease is especially important when crop growth is not seriously affected, because that is the time to control the disease. Thus we can assume that crop growth is independent of the growth of the disease, so that it can be introduced in the program as another forcing function. It suffices to use for this purpose the course of the leaf area index, that is the ratio between the surface of the leaves and the surface of the soil, which varies from 0 at emergence to about 5 at flowering. In the present simulation it is simply assumed that

```
FUNCTION LAIT = (0.,3.),(140.,3.)  
LAI = AFGEN(LAIT,TIME)
```

---

### Exercise 67

Write the section WEATHER of the simulation program, complete with AFGEN functions and FUNCTION tables and the temperature (TEMP), the wind speed (WIND), the wetness of leaves (WET), the dryness of the leaves (DRY), the light condition (LITE) and LAI as outputs.

---

Mistakes in input data may result in a situation where it rains and WET is nevertheless zero. Inconsistencies may be avoided by reading from the tables an auxiliary variable WETX and then computing

```
WET = FCNSW(WETX + RAIN, 0.,0.,1.)
```

which means that WET = 1. for WETX + RAIN greater than 0, and otherwise 0.

### 7.3 Appearance and growth of lesions

If spores of *H.maydis* are present on healthy leaves and conditions for germination are suitable, some spores will form germ tubes which penetrate through the stomata and so infect the leaves. This penetration rate is calculated at the end of the program. Here it is used as an input.

Fig. 25a shows the resulting growth of the lesion area at 30 °C as a function of the number of days after incubation. Lesions appear after about two days showing that the first stages of development occur inside the leaves. Thereafter the lesions grow to their final size with a speed that is dependent on temperature only.




Fig. 25 | Some experimental data and the curves mimicked by the relevant parts of the simulation programs.

The relational diagram of development and growth of the lesions is given in Fig. 26. The development period inside the leaves is accounted for by seven development classes with dispersion, and a residence time of 0.5 day in the first six classes. This residence time is assumed to be independent of temperature. The content of the last integral gives the number of visible lesions and of the other the number of invisible lesions. As long as it is assumed that defoliation and decay of leaves are




Fig. 26 | Relational diagram for the growth of lesions.

negligible, this number does not decrease. Since, a soil surface of one hectare is used as reference, the number of lesions is expressed in  $\text{ha}^{-1}$ . All lesions grow to a final size (MAL) of about  $100 \text{ mm}^2$  or  $10^{-8} \text{ ha}$  and the growth rate of the individual lesions can be conveniently described by assuming that this rate is proportional to the difference of the maximum area of a lesion minus its actual area ALS. The proportionality factor (PAL) is, according to experiments, a function of temperature only, and is sufficiently defined by

FUNCTION PALT = (0.,0.), (10.,14.), (18.,33.), (23.,80.), ...  
 (30.,80.), (35.,14.), (40.,0.)  $\text{day}^{-1}$

The formula for the growth rate of the area of a single lesion is then:

$$\text{RALS} = \text{PAL} * (\text{MALS} - \text{ALS})$$

$$\text{ALS} = \text{INTGRL}(0., \text{RALS})$$

The initial value of this integral is zero, because the lesions entering class NVL have an area that is practically zero. To obtain the growth rate of the total area, RALS must be summed over all the visible lesions present, a number equal to NVL

$$\text{RAL} = \sum_{n=1}^{\text{NVL}} \text{PAL} \times (\text{MALS} - \text{ALS})$$

or

$$RAL = PAL \times (NVL \times MALS - \sum_{n=1}^{NVL} ALS)$$

or

$$RAL = PAL \times (NVL \times MALS - AL)$$

if AL is the total area of the lesions, given by

$$AL = \text{INTGRL}(0., RAL)$$

### *Exercise 68*

Why is the expression for RAL so similar to the one for RALS? Write the section GROWTH OF LESIONS, with the number of visible lesions NVL, the rate of growth of the total area, and the total area as outputs. What is the total residence time of the lesions in the invisible stages? Explain why some lesions enter NVL already at the 1.5 day. Calculate the standard deviation of lesion appearance. Why has the simulated curve for AL in Fig. 25a a sigmoid form?

The points in Fig. 25a are observations and the curve is the mimicked result. A similar analogy between observation and simulation is obtained at other temperatures.

### *7.4 Sporophore or stalk formation*

The technical term for the microscopic stalk that holds the spore in the air above the leaf is sporophore, but here the more popular term 'stalk' is used. The growth of the stalk occurs only when the leaves are wet and otherwise depends on temperature and light. The maximum number of stalks on a hectare of lesions is  $300 \times 10^{10}$ , but the experimental data in Fig. 25b for a few temperature and light conditions during formation show that this value is not reached under all conditions. Moreover there is some delay in the formation of stalks.

To mimic these results it is assumed that there is a potential number of stalks per surface unit of lesions - a number of opportunities for stalk formation - which materialize through some classes and that during the actual stalk formation a part of this potential number develop into stalks and the rest become extinct, depending on conditions. These assumptions are presented in the relational diagram of Fig. 27. The growth of the number of opportunities is the product of the maximum number per area ( $MOA = 300 \times 10^{10}$  per ha) and the growth of the




Fig. 27 | Relational diagram for the formation of green stalks.

area of the lesions. This potential number enters into a series of four classes with a residence time of 0.0625 day in three classes. The realization of the opportunities is arrested by drought. There are three possibilities: opportunities are destroyed, set back to their initial stage or their advance is arrested. Not much is known about these processes, so that at present the middle course is taken: it is assumed that the opportunities are returned to the first class in case of drought. The following rate is introduced from the integral of the Xth class

$$EMPT = DRY * OPX / DELT$$

To avoid manipulation of very small numbers in some computers

('underflows'), it may be advisable to program this rate as

```
EMPT = INSW(OPX - 1.E-50, 0., DRY*OPX/DELT)
```

Hence the classes are not emptied when their contents are below the very small value of  $10^{-50}$ .

The opportunities end in the last or fourth class and are from there removed either by stalk formation or extinction. The relative rate of extinction (POX) and stalk formation (POG) depend on temperature and light, whereas the process only occurs when the leaves are wet. An analysis of the experimental data showed that the process is sufficiently mimicked when the following functions of temperature are used for the relative rates in  $\text{day}^{-1}$ :

During light:

```
FUNCTION POGL = (0.,0.), (14.,.04), (18.,.12), (23.,1.4), ...  
(30.,1.2), (35.,0.)
```

```
FUNCTION POXL = (0.,0.), (14.,.04), (18.,.12), (23.,1.4), ...  
(30.,0.)
```

and during darkness:

```
FUNCTION POGD = (0.,0.), (14.,.10), (18.,.27), (23.,.27), ...  
(30.,1.33), (35.,.67), (40.,0.)
```

```
FUNCTION POXD = (0.,0.), (14.,.02), (18.,.03), (23.,.18), ...  
(30.,.88), (35.,1.54), (40.,0.)
```

The proper functions can be selected again by an inswitch which is operated by the variable LITE. For instance:

```
POG = INSW(LITE,AFGEN(POGD,TEMP),AFGEN(POGL,TEMP))
```

The points in Fig. 25b are again observational data for a few conditions and the corresponding curves are obtained by mimicking stalk formation and opportunity extinction. The process of opportunity formation is described by the equations and functions, but is not explained on a physiological basis. The stalks that are formed are virgin or green stalks. Because these maintain another rate of spore formation than stalks that have sporulated once or have been subjected to drought, they must be accounted for separately in an integral that maintains the number of green stalks.

---

### Exercise 69

Write now the section FORMATION OF GREEN STALKS, with the rate of green stalks formation (ROG) as output. What is the dimension

of ROG? This rate as a fraction of the potential rate (sum of actual formation and extinction) depends on light and temperature.

---

### 7.5 Sporulation of green stalks

The name 'green' stalks has been used explicitly because there are also 'dried' stalks. Dried stalks are stalks that have sporulated at least once or have been subjected at least once to drought. The distinction is made because the influence of temperature and light on sporulation is different for both categories: green stalks sporulate more rapidly than dried stalks.

Fig. 25c shows how this growth of spores on green stalks may depend on temperature and light. Here the scale of 0 to 1 represents the number of green stalks with a spore. A stalk cannot carry more than one spore at the same time. Only 50 percent of the stalks produced spores after two days in the light and at 23 °C, but there is sufficient information to assume that in due course all stalks will sporulate under these conditions. The relational diagram for sporulation of green stalks is given in Fig. 28. Three classes with a residence time of 0.0625 day in the first two are again introduced to mimic the observed delay between the formation of green stalks and the first appearance of spores. The first class is loaded according to the rate of green stalk formation; thus this class contains stalks ready for sporulation.

There are two circumstances that arrest spore formation. One when the leaves become dry; then the growing spores are aborted and the green stalks are reclassified as dried stalks. The other when the green stalks are completely destroyed, usually by rain beating against the fragile stalks. The relative destruction rate in  $\text{day}^{-1}$  is assumed to be a function of the rainfall rate, in  $\text{mm hour}^{-1}$  according to

```
RBETR=AFGEN(BEATT,RAIN)
FUNCTION BEATT=(0.,0.), (0.25,.08), (0.75,.32), (6.25,2.), ...
(18.8,5.6), (25.,6.7)
```

This function summarizes some factual information, but is largely based on a qualified opinion of the process.

---

#### Exercise 70

What rate of rainfall is needed to destroy 63 percent of the stalks in 5 hours?

---




Fig. 28 | Relational diagram for the formation of spores on green stalks.

When the stalks have been passing through the classes and have not been dried up or beaten by rain, they form spores at a rate which is dependent on light and temperature, provided, of course, that the leaves stay wet. The experimental data are mimicked with sufficient accuracy by introducing the temperature dependence

FUNCTION PGSL = (0.,0.), (14.,.15), (18.,1.44), (23.,.32), ...  
 (30.,0.), (40.,0.)

in the light and

FUNCTION PGSD = (0.,0.), (14.,.06), (18.,14.), (23.,14.), ...  
 (30.,.44), (35.,0.), (40.,0.)

in the dark for the proportionality factor of spore formation in  $\text{day}^{-1}$ . The points in Fig. 25c are again observations and the curves mimic results of sporulation of green stalks.

---

### Exercise 71

Write now the section FORMATION OF SPORES ON GREEN STALKS with the rate of spore formation on green stalks (RGS) as output. What is the dimension of RGS?

---

Further spore formation is arrested, as long as the spore remains on the stalk. Once removed, the stalk is no longer green, but classified as a dry stalk, which may also form spores but at a different rate.

### 7.6 Sporulation of dried stalks

As has been said, dry stalks are distinct from green stalks because their rate of spore formation is slower. Dried stalks are generated in various ways. When spores are removed from either a green or a dried stalk, the stalk is ready to produce a new spore at a rate characteristic for dried stalks. During spore formation when the growing spore is aborted by drought green stalks also become dried stalks. Fig. 25d shows some experimental results which are mimicked according to the relational diagram in Fig. 29. The mean residence time in each class is again 0.0625 day, the slowness of the process as compared to green stalks being accounted for by an extra class and another proportionality factor in  $\text{day}^{-1}$  for spore formation according to the temperature dependence

FUNCTION PDSL = (0.,0.), (14.,.17), (18.,1.75), (23.,.25), ...  
(30.,0.), (40.,0.)

in the light and

FUNCTION PSDS = (0.,0.), (14.,.07), (18.,2.95), (23.,2.2), ...  
(30.,.53), (35.,0.), (40.,0.)

in the dark.

When dried, the growing spores are aborted and the dried stalks are again reset into the first class. When the green stalks become dry they also enter this class. At last, the stalks that are denuded of spores, either by wind or rain and not destroyed in the process are again ready to form new spores. Dried stalks are also beaten and destroyed by rain at the same rate as for green stalks.




Fig. 29 | Relational diagram for the formation of spores on dried stalks

### Exercise 72

Write now the section FORMATION OF SPORES ON DRIED STALKS with the rate of spore formation on dried stalks (RDS) as output. Inputs are the rate of transfer of green stalks and dried stalks to the first class during drought and the rate of spore removal from stalks (RSR).

### 7.7 Dispersal of spores

When it is dry, the spores are removed by the turbulent action of the air. Some of the spores are carried away to other fields, and others settle on the soil, on lesions or on healthy foliage segments. The stalks are also denuded by rain. Especially at the onset of heavy showers, part of the spores are dispersed through the air, but with gentler rain the spores are washed

from the stalks and end up again on the soil, on lesions or on healthy foliage. The processes that are involved are very little understood particularly because the quantitative aspects are complicated: fields may be of limited size and infections are not uniformly distributed.

At such stages, the model builder has to make a difficult decision: either to abandon the whole problem or to advance for better or for worse. The latter course is usually chosen for various reasons. First, sensitivity analyses may show that the dynamics of the system are hardly determined by the processes that are difficult to handle both conceptionally and practically. Thus, it would be a waste of time to pay much attention to these processes. Unfortunately, spore dispersal is one of the important processes that governs fungal epidemics. Secondly, life goes on and operational decisions have to be made whether the system is completely understood or not: even models with unsatisfactory parts may be better than no model. Of course, this supposition must be made probable. Thirdly, it is possible to view a model not so much as a representation of the real system but as a representation of our knowledge of the system and our opinion about it. Then the weak sections should not be ignored but exposed as will be done here.

The most simple supposition is that spores are removed from stalks at a rate proportional to the number of spores present. The relative rate of removal is assumed to be zero when the leaves are wet and it does not rain. However, when it rains the spore removal rate in spores per day is

$$\text{SPRR} = \text{RWASH} * \text{STSP}$$

and when it is dry

$$\text{SPRD} = \text{RBLow} * \text{STSP}$$

in which STSP is the integral 'stalks with spore'.

The relative rates of spore removal (RWASH and RBLow) are assumed to be independent of the number of stalks with spores, although it is not unlikely that these relative rates decrease because at first the most exposed spores are removed. The main problem is to obtain a reasonable estimate of these relative rates.

Waggoner et al. (1972) estimated that with a sprinkling rate of about 6 mm/hour for 3 hours, 86 percent of the spores were removed from exposed leaves, so that the value of  $\text{RWASH} = -8. \ln(0.14) = 15.7 \text{ day}^{-1}$ .

---

### Exercise 73

Check this calculation.

---

It seems reasonable to assume that RWASH is proportional to the rainfall intensity and that below an LAI of 2 the leaves do not protect each other, so that RWASH is independent of LAI. Above this value mutual protection may exist, but since water may drop from one leaf on to the other, probably the mistakes are not large when this protective effect is neglected. The relative rate of spore removal under dry conditions depends primarily on the wind velocity. It was assumed by Waggoner et al., that at a wind speed of 2 metres per second, and a leaf area index of 3, about 5 percent of the spores are removed in 3 hours, so that SPRD can be estimated under this circumstance. Since the force of the wind is proportional to the square of its velocity, it could be assumed that the relative rate of spore removal is proportional to the second power of the wind speed. Then spore removal is zero when the wind speed is zero. However turbulence is also generated by the temperature difference within and outside the crop. This effect may be approximated by assuming that the wind speed is never less than 1 m/s. The relative rate of spore removal is also influenced by the leaf area index, because the wind speed decreases more or less exponentially with increasing depth of the crop. This effect is so uncertain, that it is not considered further.

---

### Exercise 74

Calculate RBLOW for WIND equal to  $2 \text{ m s}^{-1}$ .

---

The calculation of RBLOW completes the estimates of the rate of spore removal. The next step is the estimation of the fraction of removed spores that may become effective by settling on healthy foliage.

With strong winds and a small field most spores may be blown away and become ineffective. However they may be compensated for by spores blown in from neighbouring fields. Another question is how many spores in the air are caught by leaves and how many end up on the soil surface where they can do no harm. Again Waggoner et al. assumed that with a leaf area index of 3 and a wind speed of  $2 \text{ m s}^{-1}$ , 3 percent of the spores are caught by the leaves. This percentage is likely to depend linearly on the leaf area and is programmed as such. The percentage is also likely to decrease with increasing wind speed,




Fig. 30 | The relational diagram for germination and penetration of spores.

especially on small fields. This effect is too complicated to consider here.

The greater the intensity of the rain, the more spores are washed to the ground. According to Waggoner et al. only 0.3 percent of the spores are caught by the leaves at a rainfall intensity of 2.5 mm/hour. The maximum fraction is caught at a negligible rainfall rate, but does not exceed 20 percent. These are very rough estimates indeed.

**Exercise 75**

Write the section SPORE DISPERSAL with as output: SPRR, SPRD and their sum RSR and the rate of spore arrival at the foliage (RASP). How would it be possible to take into account the influence of host exhaustion when the area of the lesions is not negligible?

## 7.8 Germination of spores and penetration of germ tubes

The spores on healthy tissue are now considered. These germinate eventually on wet foliage. The fraction of spores that complete germination and the rate of germination depend on temperature according to the observations in Fig. 29e. The process of germination is complicated because germinating spores may be washed down from the leaves or killed upon desiccation.

The relational diagram that describes germination and penetration of germ tubes into the leaves is presented in Fig. 30.

Two integrals are distinguished: the spores on the (healthy) foliage (SF) and the germ tubes on (healthy) foliage (GT).

The number of spores increases with the rate of arrival of new spores (RASP) and decreases when spores are washed, killed or germinate. The relative rate of spore removal by rain from the leaves is set equal to the relative rate of spore removal by rain from the stalks (RWASH). The killing of spores upon desiccation is more difficult to handle. Spores can only desiccate when they have been wet. Since all spores are killed, the killing rate is

$$RKSP = KILL * SF / DELT$$

Obviously, the killing is governed by the variable KILL which may be 0 or 1. If the leaves during the previous time-interval were wet and are dry during the current time-interval, KILL is set at 1. Such a condition may be programmed by using a 'PROCEDURE' that contains a series of statements that have to be executed in the order in which they are presented. The whole sequence of statements is then sorted at a place where the inputs are available and the outputs not yet used.

The procedure that is used here is called 'DESS' from desiccation and has as input the variable WET and as output the variable KILL:

```
PROCEDURE KILL = DESS(WET)
```

The statements within the procedure are:

```
KILL=0.  
IF((WETP-WET).GT.0.)KILL=1.  
WETP=WET
```

The first statement sets KILL equal to 0 and the second statement reads: if the difference between WETP and WET is greater than 0, then reset KILL to the value 1. The next statement sets the previous value of wet (WETP) equal to the current value

and this reset value is used in the 'IF' statement during the next updating. The end of the series of statements that have to be sorted as one block is now defined with the line

ENDPROCEDURE

The spores germinate or become extinct at relative rates in  $\text{day}^{-1}$  that depend on temperature, according to the function tables

FUNCTION PFTT=(0.,0.), (10.,.4), (15.,1.8), (20.,4.6), (23.,7.0), ...  
(35.,3.7), (40.,0.)

for completion and

FUNCTION PFXT=(0.,0.), (10.,0.), (15.,1.8), (20.,4.2), (23.,2.6), ...  
(35.,3.7), (40.,0.)

for extinction.

The simulated germination is again presented by the curves in Fig. 25e. Note that at 15 °C some observations deviate considerably from the simulated line because the function tables PFTT and PFXT are assumed to be smooth and were adapted also to observational data at other temperatures.

The spores with germ tubes are also killed upon desiccation according to the rate

$RKGT = KILL*GT/DELT$

and also washed away by rain at the same relative rate RWASH as spores are washed from the stalk. Depending on temperature, only a fraction of the germ tubes ever penetrate the leaves; this observation is again mimicked by introducing relative rates in  $\text{day}^{-1}$  of penetration and extinction according to

FUNCTION PTNT = (0.,0.), (18.,.48), (23.,.65), (30.,.25), (35.,0.), ...  
(40.,0.)

for completion and

FUNCTION PTXT = (0.,0.), (18.,1.3), (23.,2.6), (30.,2.2), ...  
(35.,0.), (40.,0.)

for extinction.

These functions are found by comparing the number of lesions with the number of germ tubes formed upon incubation of spores.

---

### Exercise 76

Write the section GERMINATION AND PENETRATION with the rate of penetration (RTN) as output.

---

The cycle is completed by calculating the rate of penetration of the germ tubes, RTN being the rate needed to start the growth of the number of lesions.

### 7.9 Timing, initialization and output organization

Since there are discontinuous processes involved, it is necessary to execute the simulation according to the METHOD RECT. The time interval of integration has to be chosen small compared with the relative rates of change. An analysis of the data and parameters shows that these rates are fastest in the classes for the growth of stalks, which are governed by a residence time of 0.0625 day. When DELT equals this value, the contents of the classes are pushed without any dispersion. Here this is completely acceptable. For practical reasons of organizing input and output it is, however, convenient to set DELT to 0.04 day. Then the program is updated 25 times during one day and computing costs are acceptable.

The initialization of every integral in the program could be achieved by observing at one moment the number and area of lesions, the number of green and dry stalks, the number of spores and so on in a particular field. These observations are not worthwhile at this stage of knowledge. Usually initialization is achieved by assuming a certain number of spores or a certain number of lesions, the contents of the other integrals being set at zero. Because it is often the purpose to study the dynamics of the disease without complications due to exhaustion of the host, it is good practice to start with a small number of lesions, which may be taken as 100 per hectare.

However, in other situations it may be necessary to program a certain invasion rate of spores from the outside during some period.

---

### Exercise 77

Program an invasion rate of  $10^6$  spores per hectare per hour during the first week, but only when it is light and the leaves are dry.

---

The output of every variable may be requested of course, but it is good practice to limit the number to the most essential ones. These are in general the contents of the main integrals:




Fig. 31 | The number of visible lesions in dependence of time, when initialized with 100 spores that completed penetration (NIL1 = 100).

the number and area of the lesions, the number of green and dry stalks and of stalks with spores and the number of spores and germ tubes on healthy leaves. To study the state of the system at certain moments it may be convenient to have all outputs available by introducing the statement

OUT1 = DEBUG (N,T)

in which T is the moment at which this output procedure starts to operate and N the number (without decimal point) of successive updates for which output is requested. As many debugs as needed may be introduced.

---

### Exercise 78

Ask for a debug of ten rounds at time zero and of debugs of two rounds at time 5, 5.5, 10 and 10.5.

---

### 7.10 Results and sensitivity analyses

A simulated epidemic, as characterized by visible lesion number (NVL), is presented in Fig. 31 on a logarithmic scale, starting with 100 lesions per hectare (NIL1), the growth being simulated for the defined stationary weather pattern. During the first periods of growth, the effect of initialization can still be distinguished. At a later stage it is possible to characterize growth by a relative growth rate of number of fungi lesions, which is in the present example  $0.34 \text{ day}^{-1}$ . Other important characteristics are the rates of spore production (RSP) and dispersal by wind (SPRD). The simulated results of these rates for days 8 and 9 are given in Fig. 32, together with relevant weather data. The rate of spore removal by wind may be verified in a relative sense by studying the density of spores above the crop. Verification of simulated data on epidemics is difficult for two reasons. In the first place, a good meteorological network that provides not only the course of the standard meteorological parameters throughout the day, but also detailed information of the wetness of the leaves must be available. In the second place, field observations must be organized. Sometimes a rating of severity in a wide range of localities may do, but preferably the relative growth rate of the disease over a certain time span should be observed. Some comparisons of simulated results, obtained with the original 'EPIMAY' of Waggoner et al. and actual results throughout the USA are given in Fig. 33. The left graph is a comparison of the simulated multiplication rate of lesions with a net increase of blight ratings in various places in the Mid-Eastern USA in 1971 and the right graph compares simulated and actual multiplication rates in Western Indiana in 1971. Only the latter gives a comparison in absolute terms, but it should be taken into account that some 'fudging' of parameters has been done to achieve correspondence of level. Whether such fudging is acceptable or not is not a matter of principle, but of purpose. If it is the purpose to develop a forecasting technique as soon as possible, one may incorporate experience of previous years into the program. However it is then very difficult to judge which of the numerous parameters should be left alone and which should be adapted. If extensive adaptation is necessary, it is doubtful whether much is gained at all by simulation compared with the application of one of the standard multiple correlation




Fig. 32 | The simulated rate of spore production (RSP) and rate of spore removal by wind (SPRD) during two days.

techniques.

If it is the purpose to understand the dynamics and the quantitative aspects of the disease, fudging of parameters to achieve better agreement should be avoided. Instead, a sensitivity analysis under the prevailing conditions should be executed, to evaluate which parameters mainly control the disease. The result of this analysis, should then be a guide to further experiments and study.

Such a sensitivity analysis consists of varying inputs and parameters over a certain range and a comparison of their relative influence on the end result. If the influence of a certain parameter or input is relatively small, further analyses may be left for some time, but if the influence is large, more work should be invested in a further analysis of the section of the program where this parameter plays a role. The problems of evaluation of simulation programs are treated more thoroughly in another book of this series (Arnold & de Wit, 1976).




Fig. 33 | A comparison of simulated and actual results in the Mid-Eastern USA(a) and in Western Indiana(b) in 1971.

The weather parameters that are most likely to affect the severity of the epidemic are the duration of the wetness of the leaves, the presence of showers and the temperature. The simulated influence of duration of wetness on the relative growth rate of the lesions for the standard weather conditions, but in the absence of rain, is given in Fig. 34. The propagation of the disease is practically zero when the duration of wetness is less than . hours, because the fungus needs wetness periods of finite length to complete its development in various stages. The relative multiplication rate increases to a maximum at 18 hours of wetness, but then it decreases again to zero at 24 hours of wetness, because spores are assumed not to disperse by wind when the leaves are wet, and because rain is supposed to be absent.

The picture changes completely when at 24 hours of wetness rain is assumed to occur at a rate of 6 mm per hour for 3 hours per day; then the relative growth rate equals 0.19 day<sup>-1</sup>. Here the rain causes the dispersal of the spores. The influence of decreasing the intensity of the shower and increasing its duration is considerable. A rainfall of 1 mm per hour for 18 hours per day causes a relative growth rate of 0.84 day<sup>-1</sup>. At a lower rate the relative beating rate RBETR decreases so that fewer stalks are destroyed, but the spore dispersal by rain increases, so that many more spores are caught by leaves. As long as the total amount of rain is the same, the change of RWASH has little




Fig. 34 | The relative growth rate of the number of lesions in relation to the duration of wetness, without rain.

influence because it is proportional to RAIN. Thus the knowledge of the daily total rainfall is not sufficient; the rainfall distribution must be known as well. The influence of temperature is analysed under the assumption that the other weather conditions are standard. Two situations are distinguished: in one series, the temperature amplitude is fixed at 5 °C and the average temperature is varied from 15 °C to 35 °C and in the other the average temperature is fixed at 25 °C and the amplitude is varied from 0 ° to 15 °C.

---

#### Exercise 79

Program this situation by assuming a sinusoidal temperature course throughout the day with a maximum at 14 hours.

---

The results are given in Table 6 and show that one temperature value, such as an average temperature, does not give detailed enough information. The influence of daylength under otherwise standard conditions is found to be small.

A sensitivity analysis of the parameters and function tables that are included in the program may be made also. For instance the influence of the residence time in the various classes may be evaluated and especially because the residence times are assumed to be independent of temperature. Another aspect that may be of importance is the assumption on the development of green stalks. Does it make much difference whether developing stalks are destroyed by drought during development, whether

Table 6 The influence of temperature average and amplitude (°C) on the relative growth rate (day<sup>-1</sup>)

AVTMP AMPL	15	20	25	30	35
0	—	—	0.452	—	—
5	0.079	0.321	0.459	0.407	0.192
10	—	—	0.389	—	—
15	—	—	0.254	—	—

their growth is only arrested or whether they are reset in their first class upon drought? Is it worthwhile to make a distinction between green and dry stalks? Another important and largely unknown set of parameters concerns dispersal and recapture.

## 8 Population models for fruit-tree red spider mite and predatory mites

Techniques for simulating growth and development were presented in Chapter 6 and used throughout Chapter 7 to simulate the population dynamics of the fungus *Helminthosporium maydis*. This present chapter describes a basic model for the population dynamics of the harmful fruit-tree red spider mite, *Panonychus ulmi* Koch, and one of its natural enemies, the predatory mite *Amblyseius potentillae*. The model, its results and evaluation are briefly presented, a more detailed description of the model, its construction and parameterizing, verification and evaluation being given in another monograph of this series (Rabbinge, 1976).

The fruit-tree red spider mite belongs to the family Tetranychidae, a subgroup of the class of Acarina. The members of this vast family are found almost all over the world and may cause damage in several types of plants. In deciduous fruit orchards, the fruit-tree red spider mite is one of the most harmful organisms and its control is the main task of the fruit grower. During the last decennia, the mites have become increasingly resistant to different biocides, so that the development of other control techniques has been stimulated. Biological control is an attractive alternative, especially as field experiments with release of predatory mites have shown promising results.

### 8.1 Fruit-tree red spider mite

In Fig. 35 the life cycle of the fruit-tree red spider mite is given. The mites develop from eggs through different juvenile stages into adult females and males. The females deposit their eggs on leaves and a new generation starts; up to six generations per year may occur. Induced by a combination of daylength, average daily temperature and food quality, some juveniles may develop in the 'winter form' which oviposit in sheltered places on branches and twigs, but are otherwise indistinguishable from the summer form. The transition to the 'winter form' which is partly reversable is a complicated process and is not treated here.

Their eggs, winter eggs, possess a thicker scale and are more reddish than the summer eggs. The winter eggs hatch only after




Fig. 35 | Life cycle of fruit-tree red spider mite (*Panonychus ulmi* Koch)

the winter period.

To mimic the dispersion in hatching, development classes of the eggs are distinguished, the number of classes and the method of simulation depending on the relative dispersion at different temperatures (see Chapter 6). The start of the hatching process is induced by a combination of external variables: length of cold period, daylength and temperature. During the hatching process some eggs die. Their relative mortality rate depends on temperature.

Eggs hatch into juveniles which moult several times during development into females or males. Within each juvenile stage, different development classes can be distinguished, the number of classes depending on the relative dispersion. Fig. 36 is a relational diagram for the development of *P. ulmi* during the juvenile phase from larvae into deutonymph. This part of the juvenile phase is distinguished and treated separately because the 'winter form' is only induced in the older juvenile stages.




Fig. 36 | Growth and development in the juvenile stages of the fruit-tree red spider mite (larva, protochrysalis, protonymph, deutochrysalis).

Table 7 gives the length of the development period and its dispersion for the juvenile stages and for winter eggs in relation to temperature.

Table 7 Duration ( $\bar{x}$ ) of different developmental stages of fruit-tree red spider mite and its standard deviation ( $s(x)$ )

	Temperature							
	15°C		18°C		25°C		30°C	
	$\bar{x}$	$s(x)$	$\bar{x}$	$s(x)$	$\bar{x}$	$s(x)$	$\bar{x}$	$s(x)$
Egg	18.0	3.0	10.0	1.6	6.3	1.06	5.0	1.25
J	10.1	2.0	6.5	1.3	4.0	0.8	2.5	0.62
JS	5.5	1.1	4.3	0.85	2.4	0.6	2.0	0.5

J = juveniles insensitive to diapause-inducing conditions  
(larva, protochrysales, protonymph, deutochrysales)

JS = juveniles sensitive to diapause-inducing conditions  
(deutonymph, teleiochrysalis)

#### Exercise 80

Calculate the number of classes that is required to mimic the dispersion of the stages EGG, J and JS, when a boxcar train with constant relative dispersion (see Section 6.4.2) is used. Why can this method only be used when the temperature is constant? Which method must be used when the temperature varies? What is then a good choice for the number of classes? Write a simulation program for the hatching process of winter eggs, assuming an initial quantity of 1,000 winter eggs and a daily temperature fluctuation between 15 and 30 °C. The mortality during this development process may be neglected. Calculate the relative rate of mortality in the juvenile stage for a constant temperature of 30 °C, if the experimental results on mortality show that at this temperature, 50% of the juveniles die during their development from egg to deutonymph.

Development of both eggs and juveniles is simulated with boxcar trains to mimic the dispersion during these processes. A subroutine for this mimicking procedure, which has general applicability, was developed by de Jonge and Rijdsdijk. This subroutine is added at the end of a program and then called upon in the program by inserting the statement:

```
OUT1,TOT1 = BOXCAR (TOTI1,REST1,SD1,RM1,IN1,N1)
```




Fig. 37 | Relative rate of mortality in day<sup>-1</sup> in different stages of fruit-tree red spider mite against temperature in °C.

In this statement the variables OUT1 and TOT1 express the output of the subroutine, OUT1 is the rate of outflow of the last integral of the train and TOT1 the sum of the contents of the integrals in the train with initial value TOTI1. The variable REST1 is the total residence time in the train (a parameter or auxiliary variable derived from literature or determined by experiments), SD1 is the standard deviation of this residence time (development period, diffusion time, etc.) and RM1 is the relative rate of mortality during this development period. IN1 is the rate of inflow in the first integral of the train and N1 the number of classes in the train.

The subroutine BOXCAR is described in Appendix B, as a listing of the computer program with comprehensive comments.

---

### Exercise 81

Write a simulation program for the development of the fruit-tree red spider mite into adults using the data of Table 7 and Fig. 37. Use the subroutine BOXCAR. Temperature is the only changing external variable.

---

Females and males emerge from the last juvenile stage in equal

Table 8 The oviposition rate in eggs per day for each age-class and total oviposition period in days, both as a function of temperature.

	Temperature									
	10°C		15°C		20°C		25°C		30°C	
	$\bar{x}$	s(x)	$\bar{x}$	s(x)	$\bar{x}$	s(x)	$\bar{x}$	s(x)	$\bar{x}$	s(x)
Age class 1	0.6	0.15	1.2	0.2	1.9	0.7	3.1	0.8	4.2	1.1
Age class 2	0.5	0.1	1.2	0.2	2.2	0.7	3.7	0.8	5.5	1.2
Age class 3	0.5	0.1	1.1	0.2	1.8	0.6	3.1	0.6	3.8	1.1
Age class 4	0.4	0.08	1.	0.2	1.4	0.4	2.0	0.4	1.4	0.4
Age class 5	0.2	0.05	0.8	0.15	0.8	0.2	1.0	0.2	0.8	0.1
Oviposition period	24	14.5	25	12	13	5.0	9.0	3.7	7.5	2.4

proportion. After copulation the fertile females mature during a temperature-dependent pre-ovipositing period and then start laying eggs. The oviposition rate and the rate of ageing of the females depend strongly on temperature (Table 8). Moreover, the oviposition rate is also dependent on the physiological age of the females.

This age dependence means that normally the oviposition rate of young females is higher than average whereas that of old females is much less than average. When calculations are executed with the average oviposition rate during the whole life period of a female, the simulated total number of produced eggs may be correct but the course of the cumulative egg production curve is wrong. At the start, egg production per day is underestimated and at the end oviposition rate is overestimated.

### Exercise 82

Explain why the subroutine BOXCAR may not be used to mimic the ageing process when oviposition rate is dependent on physiological age.

Ageing of the reproducing female is simulated with the basic method of Section 6.3.1, taking into account a temperature and age dependent relative mortality rate. These relative mortality rates are calculated as follows:

The maximum period of living is defined as the mean life span plus 3 x standard deviation (s). The residence time in a single class is the maximum period, as just defined, divided by the number of classes N. The percentages of animals alive at the end of each age class are now read from the cumulative frequency distribution drawn on probability paper and with the formula used in exercise 80 the relative mortality rate per

age class is calculated. This leads to the results in Table 9.

Table 9 Relative mortality rate per age class in dependence of temperature in day<sup>-1</sup>.

Age class	Temperature				
	10°C	15°C	20°C	25°C	30°C
1	0.011	0.011	0.005	0.005	0.003
2	0.033	0.031	0.049	0.050	0.027
3	0.088	0.089	0.161	0.245	0.273
4	0.15	0.138	0.360	0.504	0.709
5	0.2	0.15	0.4	0.6	0.8

The only function of the males is copulation, for which the number of males is not limiting: more than one copulation being possible. So from this aspect, they can be disregarded. However, they act as a source of food for predators, so that they cannot be omitted from the simulation. The ageing process of the males is described by inserting the statement:

```
OUTM,MALES=BOXCAR(MALEI,RESTM,SDM,RMM,INM,NBM)
```

The females lay winter eggs that overwinter on twigs or branches or summer eggs that give rise to a new generation of mites during that summer. The hatching process of summer eggs is again mimicked by a boxcar train.

### Exercise 83

Extend the simulation program of Exercise 81 to include oviposition and ageing of females. Use for this purpose the data of Tables 8 and 9.

## 8.2 Predatory mite, *Amblyseius potentillae*

The life cycle of the predatory mite is very similar to that of the fruit-tree red spider mite (Fig. 35). The only difference is that the predatory mite overwinters as an adult female and not as an egg. The sensitive period for induction of either summer or winter adults is the same: the older juvenile

stage. Therefore the structure of the program is almost the same and not given here.

### 8.3 Relations between predator and prey

For the fruit-tree red spider mite, the food source is unlimited as long as the damage threshold of the trees is not exceeded. For the predatory mite the food supply is restricted. The availability of prey affects the development rate, the reproduction rate and fecundity (total number of eggs produced during the life of a female) of the predator. When prey are scarce, the relative mortality rates of the predators in the different stages may even increase.

The hunger of the predatory mite is characterized by its gut content (Fig. 38) and determines the relative predation rate and the prey utilization by the predator. The relative predation rate is the absolute predation rate per area divided by the prey density; this variable has the dimension  $\text{time}^{-1}$ . Hungry predators possess a high relative predation rate. Moreover the activity of the predators increases with decreasing gut content. The encounters of hungry predators with prey are nearly always successful (fatal for the prey) and the dead prey is sucked out completely. Well-fed predators, on the other hand, are less active and have a low success ratio (successful encounters divided by total number of encounters) and the prey that are killed are only partly consumed. A predator kept at a constant prey density reaches in general a steady state (Fransz, 1974) in which a unique relation exists between gut content and temperature on the one hand and the relative predation rate and prey utilization on the other.




Fig. 38 | State (gut content) of the predatory mite *Amblyseius potentillae*.

Old adult female predators can eat all stages of prey except eggs and show a clear preference for the younger stages, especially at higher gut contents. Young juveniles of the predator can only kill larvae and young juveniles of the prey, because the older prey stages are too strong an opponent. The gut content is digested at a rate proportional to the content so that a relative rate of digestion exists which is only dependent on temperature.

---

#### *Exercise 84*

Write a simulation model for the state (gut content) of a predator based on the relations presented above. Assume that temperature and density of prey and predator are constant. You can fill in the required data yourself. Give the dimensions of all parameters and variables.

---

#### *8.4 Preference as a competitive process*

So far preference of the predator for some type of prey is treated by a difference in relative predation rate. At higher gut contents the relative predation rate of an unattractive prey type drops to zero by dividing the predation rate by the corresponding prey density. This relative predation rate is introduced in the simulation model as a variable dependent on temperature and gut content. The relative predation rate is easily determined from the functional response curve (predation rate as a function of prey density (Fig. 39)). Each prey density corresponds to a well determined level of gut content of the predator and thus relative predation rates of different prey types can be related through the gut content of the predator. This approach is straightforward and its validity can be tested in experiments with replacement series of two prey types (Rabbinge, 1976). The results support the assumption that the density of one prey type can effect the predation rate of another one only through the gut content of the predator.

Another approach in which simulation is not needed and which allows us to bypass the gut content of the predator, is to consider predation and preference as competitive processes. Prey compete for space in the gut of the predator, so that it should be possible to derive the predation rates in mixed prey populations from the predation rates in pure populations (monocultures, see Chapter 4). The predation rate as a function of the prey density (Fig. 39) is described by




Fig. 39 Functional response curve of fruit-tree red spider mite.

$$PR = \frac{B \times D}{B \times D + 1} PRM \tag{8.1}$$

where D is prey density, PRM the maximum predation rate and B an apparent area per prey. PRM and B are determined for each combination of prey type and predator type from the functional response curve. Inversion of Eqn (8.1) gives:

$$\frac{1}{PR} = \frac{B \times D + 1}{B \times D} \times \frac{1}{PRM} = \frac{1}{B \times D \times PRM} + \frac{1}{PRM}$$

The corresponding graphical representation is a straight line that crosses the y-axes at  $1/PRM$  and has a slope of  $1/(PRM \times B)$ . Maximum predation rate and apparent area per prey can be derived directly from the given lines. The analogy to Eqn (4.6) is obvious. It is therefore tempting to describe predation in a mixed prey composition with the same equation:

$$PR_1 = \frac{B_1 \times D_1}{B_1 \times D_1 + B_2 \times D_2 + 1} PRM_1 \quad (8.2)$$

and similarly for the predation rate of prey type number 2. Comparison with both the experimental results and the simulation results with the more fundamental method described earlier show that this approach gives satisfactory results (Fig. 40). Of course its validity was only confirmed under the restriction that the gut content of the predator must be in equilibrium with the available density and composition of prey.




Fig. 40 | Simulated and experimental results of a replacement series of larvae and adult females of *Panonychus ulmi* with one adult female of *Amblyscius potentillae* at 15°C, in terms of simulated values and measured confidence intervals.

### Exercise 85

Why in Chapter 4 could the yield in a mixed culture not be described by such a simple extension of Eqn (4.6) and why was simulation of the relative space RS necessary?

In a replacement series the sum  $D_1 + D_2$  is constant and equal to  $D_m$ . The monoculture predation rate ( $D_1 = D_m, D_2 = 0$ ) is then given by

$$PR_{mon,1} = \frac{B_1 \times D_m}{B_1 \times D_m + 1} PRM_1 \quad (8.3)$$

Elimination of  $PRM_1$  from the Eqns (8.2) and (8.3) results in the following expression for  $PR_1$  in the mixed culture

$$PR_1 = \frac{(B_1 \times D_m + 1) \times d_1}{B_1 \times D_1 + B_2 \times D_2 + 1} PR_{mon,1} \quad (8.4)$$

in which  $d_1$  stands for  $D_1/D_m$ .

After division of numerator and denominator by  $D_m$  we obtain

$$PR_1 = \frac{(B_1 + 1/D_m) \times d_1}{(B_1 + 1/D_m) \times d_1 + (B_2 + 1/D_m) \times d_2} PR_{mon,1} \quad (8.5)$$

This expression is analogous to Eqn (4.5), with a relative crowding coefficient  $k_{12}$  equal to

$$k_{12} = (B_1 + 1/D_m) / (B_2 + 1/D_m) \quad (8.6)$$

The coefficient  $B_1 + 1/D_m$  can be obtained graphically from the functional response curve for a single prey type, as in a similar way maximum predation rate and apparent area per prey were found from Eqn (3.1). Thus  $B_1 + 1/D_m$  is simply the distance from this intersect along the  $1/D$  axis to the total density.

### Exercise 86

If prey and predator properties are constant, what are the extreme values of the relative crowding coefficient  $k_{12}$  when the total density  $D_m$  is varied?

### 8.5 Verification

Model building is mostly a futile exercise if model output and results of independent experiments are not compared. Such verifications are preferably done at different levels. One such level is mixed prey predation as derived from predation of single prey types. This verifies only the







Fig. 41 | Simulated and experimental results of a population experiment of *P. ulmi* and *A. potentillae* in a greenhouse; simulated results and experimental results in terms of confidence intervals.







Fig. 42 | Simulated (---) and experimental (—) results of *P. ulmi* and *A. potentillae* in an orchard. The experimental results are the average of several orchards.

assumption about the interaction between different prey types as was discussed in the previous section. A second level of verification concerns population experiments of prey and predator in a larger but still small system with well defined boundaries under controlled conditions, e.g. a small apple tree in a greenhouse. The comparison of such measurements with simulated results indicates a reasonable correspondence (Fig. 41) so that a third phase of verification is justified: the apple orchard. Some results of simulations for orchards and experimental results are given in Fig. 42. The reasonable agreement increases the confidence in the model. Verification on different levels facilitates pinpointing errors in model structure or parameter values. Another important tool in error spotting is sensitivity analysis, because it shows which parameters have great influence and could thus be res-

possible for deviations between measurements and results. The results of a sensitivity analysis also help in deciding research priorities. For instance sensitivity analysis shows that in this model the abiotic mortality of prey and predator are of minor importance. The most important factors are the oviposition rate, the predation rate of the adult female predator and the length of the juvenile period of the prey. Also the gut content, especially when it is low, has a quite important effect on the predator's rates of development and reproduction. Experimental data on these relations are scarce and inaccurate, so that more research in this direction is required.

The model described may also be used for the development of a practical pest management method. Present-day crop growth models can be combined with population models on phytophagous and predacious arthropods. After quantification of the relation between host plant and phytophagous animal, these combined models may be used for calculating reduction in yield. With a preset limit for acceptability of yield reduction, threshold levels for the density of the harmful animal may be found.

## 9 Solutions of the exercises

① The differential equations for the falling apple are:

$$\frac{dv}{dt} = g, \quad \frac{ds}{dt} = v$$

The rate of change of the amount of electric charge on a capacitor is equal to its charging current and the potential across the capacitor is equal to its amount of charge divided by its capacitance, so that

$$\frac{de}{dt} = i/c$$

By substituting  $i_1 = g \cdot c_1$  and  $i_2 = e_1 \cdot c_2$  for the charging currents of a first and second capacitor, the differential equations for the falling apple are obtained.

② Our results are as follows:




Fig. 43

In a first model of this system it may be sufficient to assume that there are two variables of state, the distances, which increase by the rate of walking of each child. There are two independent rate determining processes: one in the mind of each child who wants to stay side by side with the other. However, this model would not explain why the step sizes in the beginning of the process are more variable than at the end. It has to be realized that each child has a memory which is

able to conserve the sizes of the step of the other. Each child thus determines the step size to be taken not only on the difference in position but also on a 'normal step size for the other child', figured from data conserved in memory. These memorized data characterize also a state of the system. Hence a more sophisticated model requires more than two state variables. Digital computers are much more suitable to memorize such historical data than analogue computers and this is one of the reasons why they are preferred to simulate complicated state determined systems.

③ A is expressed in grams and GR in grams hour<sup>-1</sup>. Eqn (2.1) is only dimensionally consistent when RGR is expressed in hour<sup>-1</sup>. For the relative growth rate to remain constant there should be a constant amount of food and also of harmful waste product. This situation is achieved most simply by an abundant food supply beyond saturation, and an entire removal of waste products. Moreover environmental conditions such as temperature must be kept on a constant level.

④ The result is

TIME	0	2	4	6	8	10
A	1.000	1.221	1.492	1.822	2.226	2.718

The relation between the logarithm of the amount and time is linear, since taking the logarithm is per definition the inverse of taking the exponent.

If

$$A = e^{RGR \times T}$$

then

$$\ln(a) = RGR \times T$$

in which ln stands for the logarithm with base e.

It is recalled that

$${}^{10}\log(e) = 0.43429$$

or that

$$e^{\log(10)} = \ln(10) = 2.3026,$$

so that

$$\ln(A) = 2.3026 \times {}^{10}\log(A)$$

⑤ The results with DELT = 2 hours are:

TIME	A	RGR x A	RGR x A x DELT
0	1.000	0.1	0.2
2	1.200	0.12	0.24
4	1.440	0.144	0.288
6	1.7280	0.1728	0.3456
8	2.0736	0.20736	0.41472
10	2.4883		

⑥ Some results with DELT equal to 1 and 0.5 hours are

TIME	0	2	4	6	8	10
A(DELTA=1)	1.000	1.210	1.464	1.772	2.144	2.594
A(DELTA=0.5)	1.000	1.216	1.477	1.796	2.183	2.653

$A_0$  (=IA) being the initial amount,  $A_1$  equals  $A_0 + A_0 \times \text{DELTA} \times \text{RGR}$  after one time-interval and after two time-intervals

$$A_2 = A_1 + A_1 \times \text{DELTA} \times \text{RGR}.$$

In general the relation

$$A_n = A_{n-1} (1 + \text{DELTA} \times \text{RGR})$$

holds.

Since  $A_n$  can be written as the product of  $A_{n-1}$  and  $(1 + \text{DELTA} \times \text{RGR})$  and RGR is constant, the expression can be transformed into

$$A_n = A_0 (1 + \text{DELTA} \times \text{RGR})^n$$

This is the value of  $A_n$  at time  $n \times \text{DELTA}$ , so that

$$A_n = A_0 (1 + \text{DELTA} \times \text{RGR})^{\text{TIME}/\text{DELTA}}$$

or

$$A_n = A_0 \left(1 + \frac{1}{X}\right)^{X \times \text{RGR} \times \text{TIME}}$$

with  $X = 1/(\text{DELTA} \times \text{RGR})$

When TIME stays constant and DELTA approaches zero, X approaches infinity and the expression for A approaches to

$$A = A_0 \times e^{\text{RGR} \times \text{TIME}}$$

in which the number e is standing for

$$e = \lim_{X \rightarrow \infty} \left(1 + \frac{1}{X}\right)^X = 2.7182$$

This is the so-called analytical solution for the differential equation of exponential growth, which is just a standardized way to write the procedure for a numerical solution.

⑦ RGR is the only variable containing the dimension of time. If RGR is expressed in  $\text{hour}^{-1}$ , TIME assumes the dimension hour. If programs contain more variables with the dimension time, care must be taken to use the same unit of time. Of course it is always necessary to express variables in consistent dimensional units: a pitfall that is easily overlooked.

⑧ See Fig. 1.

⑨ The rectilinear method of integration always gives an underestimate when the integral shows an upward curvature as a function of time. The accuracy of the estimate can be improved by choosing a smaller value of DELT, but an underestimate cannot be turned into an overestimate in this way.

⑩ Our function looks as follows:




Fig. 44

⑪ Compared with the scatter of the observational data, the deviations between the smoothed curve and the straight segments is small, so that it is unnecessary to use smaller temperature-intervals in the tabulated function.

- (12)  $\text{sine } (0^\circ) = 0$   
 $\text{sine } (15^\circ) = 0.259$   
 $\text{sine } (30^\circ) = 0.5$   
 $\text{sine } (60^\circ) = 0.865$   
 $\text{sine } (90^\circ) = 1.$

The maximum temperature is reached just 6 hours from the beginning of the day.

- (13) The rate of change of  $T$  is 1, so that  $T_{T+\text{DELTA}} = T_1 + \text{DELTA}$  and because the initial value of  $T$  is zero,  $T = \text{TIME}$ .

- (14) The answers do not differ very much for different choices of DELTA, so that 0.5 hour seems to be a reasonable choice. The relative growth rate is 0.1 at a temperature of  $12.5^\circ\text{C}$  and the temperature is maintained on this level by introducing

PARAMETER AVTMP = 12.5 , AMPTMP = 0.




Fig. 45

- (16) An acceptable relative error of 5 percent means that  $\frac{1}{4} * \text{RGR}^2 * \text{DELTA} * \text{TIME}$  must be 0.05 at most. With  $\text{RGR} = 0.1$ ,  $\text{TIME} = 1$  the maximum value for DELTA is 1. According to the graph that was plotted in Exercise 15, the maximum value of DELTA is

1.2 hour. The agreement between both methods is reasonable. The difference is due to the neglect of the higher order terms in the derivation of the equation for the relative error.

(17) The listing of the program is

```

TITLE RELATIVE GROWTH RATE TEMPERATURE DEPENDENT
INITIAL
    COUNT1=0.
    COUNT2=0.
DYNAMIC
    A=INTGRL(IA,GR)
    GR=RGR*A
INCON IA=1.
    RGR=AFGEN(RGRTB,TEMP)
FUNCTION RGRTB=(0.,0.), (10.,0.08), (20.,0.16), (30.,0.21), ...
    (40.,0.24), (50.,0.25)
PARAMETER AVTMP=20., AMPTMP=10.
    TEMP=AVTMP+AMPTMP*SIN(6.2832*TIME/24.)
NOSORT
    COUNT1=COUNT1+1.
    COUNT2=COUNT2+KEEP
TIMER FINTIM=48., OUTDEL=2., PRDEL=2., DELT=0.5
PRINT COUNT1,COUNT2
OUTPUT A,RGR,GR
METHOD RECT
END
METHOD TRAPZ
TIMER DELT=1.
END
METHOD RKSFX
TIMER DELT=2.
END
METHOD RKS
END
STOP
ENDJOB

```

The results at time 48 are

	RECT DELT=0.5	TRAPZ DELT=1.	RKSFX DELT=2.	RKS aut.
A	1034.1	1330.3	1367.7	1368.4
COUNT1	98	98	98	180
COUNT2	97	49	25	19

We must assume that the results obtained with the automatically

adapting method RKS are the most accurate. To get this result the program was executed ('updated') 180 times, and 19 time intervals of integration were required to cover the total simulation period of 48 hours. On the average almost ten calculations of the program are needed for each time step. This number drops to four with RKSFX, to two with TRAPZ and to one with RECT. Therefore the number of times the program was calculated was the same in these three runs with DELT = 2, DELT = 1 and DELT = 0.5, respectively. With the same computational effort method RKSFX, a Runge-Kutta/Simpson method with fixed interval of integration, gives by far the most accurate results, the next is TRAPZ and RECT scores the worst. The deviation with RECT is of the order to be expected. On the average the relative growth rate was  $\ln(1368)/48$  or 0.15. With DELT = 0.5 and TIME = 48 the relative error is 0.27, so that the result should be about 1000. RGR was varying, but still this estimate for the relative error is quite reasonable.

⑱ It was said that two rates do not depend on each other, but not that one rate cannot depend on the other. Here, the rate of growth and the rate of alcohol production are consequences of the same process: the biosynthesis of yeast material out of sugar. Therefore, there is a fixed ratio between rate of growth and rate of alcohol production. The rate of sugar consumption is stoichiometrically related to the above two rates: laws of conservation of matter, energy etc. can be formulated in such a way that some rate of appearance always equals some rate of disappearance.

⑲ The rate of sugar consumption is equal to a sugar consumption factor times the rate of yeast growth for each species. The amount of sugar is an integral which is emptied by both rates. The amount or concentration of the sugar in the medium should feed back on the growth rate of the yeasts. The quantitative aspects of this feedback are not presented in the diagram.

⑳  $RED1 = ALC/MALC$ , as here  $0 < ALC < MALC$ .  
Otherwise, RED1 should be given by

$$RED1 = \text{LIMIT}(0., 1., ALC/MALC)$$

㉑ The best estimate of RGR1 is obtained by presenting the amount of yeast during early growth on a logarithmic scale against time and drawing a straight line through the data. The value is about  $0.2 \text{ hour}^{-1}$ . The value of ALPF1 is the alcohol concentration at the end, divided by the amount of yeast newly grown or  $1.5/(13.-0.45) = 0.12$  percentage of alcohol per unit

of yeast. The alcohol production factor depends on the size of the vessel. In a larger vessel, the same amount of alcohol would cause a smaller percentage. It would be more elegant not to mix up the influence of physiological aspects (alcohol production rates) with experimental aspects (vessel size), but Gause did not give the latter. The alcohol percentage corresponding with the initial amount of yeast is  $ALPF1 \times Y1$ , but Gause did not add this alcohol with the yeast at time zero. Relevant figures for *Schizosaccharomyces* are:

$$RGR2 = 0.05 \text{ hour}^{-1}, ALPF2 = 0.26 (\% \text{ alc.}) (\text{unit yeast})^{-1}$$

*Schizosaccharomyces* has the largest alcohol production factor.

② Here *Saccharomyces* would grow more slowly and *Schizosaccharomyces* faster than suggested by a linear dependency of the reduction factor on the alcohol concentration. In the monocultures, this deviation from the linear dependency would not affect the ultimate amount of yeast that is formed, but in the mixture it would lead to less *Saccharomyces* and more *Schizosaccharomyces*. However, the growth curves for the two species in the monoculture would also be of different form. The scatter in the data is too large to detect a difference of this kind. The alcohol concentration in the mixture may be calculated by multiplying the final yields with the respective alcohol production factors. A concentration of 1.43%, rather than 1.5% is then calculated. Hence, there is less yeast in the mixed culture than would be expected.

③ The yeast will grow and increase its amount, and thereby its growth rate and alcohol production rate, until the alcohol concentration approaches 1.5%.

In this situation an infinite amount of yeast will maintain a growth rate which is just sufficient to produce the alcohol that is continuously removed by washing. The removal rate of alcohol is 1.5/10, and the absolute growth rates are obtained by dividing it by the alcohol production factor. This rather ridiculous result is obtained because of the assumption that the maintenance of yeast cells does not need energy and thus does not result in some alcohol production. Obviously, a simulation program which is satisfactory in some situation is not satisfactory in others because simplifications that apply in one situation do not necessarily apply in another.

④  $YM$  equals  $MALC/ALPF$ . Since  $ALPF$  was calculated from  $YM$ , it is not surprising that the  $YM$  equals 13 and 5.8 for the species. The first derivative of  $c/v$  equals  $(-c/v^2) \times (dv/dT)$ , when  $c$

is a constant, so that the first derivative of Eqn (3.7) is:

$$\frac{dY}{dT} = \frac{-YM}{(1 + K \times e^{-RGR \times T})^2} \times (-K \times RGR \times e^{-RGR \times T})$$

The two minus signs cancel, and part of the expression can be replaced by Y itself:

$$\frac{dY}{dT} = Y \times \frac{K \times RGR \times e^{-RGR \times T}}{(1 + K \times e^{-RGR \times T})}$$

The fraction  $\frac{K \times e^{-RGR \times T}}{(1 + K \times e^{-RGR \times T})}$  can also be written as

$$1 - \frac{1}{(1 + K \times e^{-RGR \times T})}$$

Substituting Y for a second time gives

$$\frac{dY}{dT} = RGR \times Y \times (1 - Y/YM)$$

In this way the differential equation (3.6) is again arrived at. The initial amount of yeast can be found by substituting for time the value zero into the integrated equation (3.7). This gives:

$$IY = \frac{YM}{1 + K}$$

The differential equation for the rate of alcohol production can only be replaced by the integral equation for the amount of alcohol if the initial amounts of yeast are small. Otherwise, the appropriate amount of alcohol, ALPF x IY, has to be added together with the initial amount of yeast if the analytical solution is to be used. Such restrictions do not hold in the simulation program because no equations are eliminated there.

②5 If we again neglect the initial amounts of yeast, the amount of alcohol in the mixed culture is given by

$$ALC = ALPF1 \times Y1 + ALPF2 \times Y2$$

Assume that RED = ALC/MALC for both species, then the growth

rates may be formulated as

$$\frac{dY1}{dT} = Y1 \times RGR1 \times \left(1 - \frac{ALPF1 \times Y1}{MALC} - \frac{ALPF2 \times Y2}{MALC}\right)$$

$$\frac{dY2}{dT} = Y2 \times RGR2 \times \left(1 - \frac{ALPF1 \times Y1}{MALC} - \frac{ALPF2 \times Y2}{MALC}\right)$$

R1 and R2 correspond to RGR1 and RGR2, A1 and A2 are equal to ALPF1/MALC and B1 and B2 to ALPF2/MALC. A1 and A2 are equal because it is assumed that Y1 and Y2 are equally sensitive to the alcohol produced by Y1. If Y1 produces some product that is more harmful for Y2 than for Y1, A2 is larger than A1.

(26) The denominator  $k_{12}z_1+z_2$  cancels by division, so that Eqn (4.4) is obtained. Addition of  $O_1/M_1$  and  $O_2/M_2$  gives  $(k_{12}z_1 + z_2)/(k_{12}z_1 + z_2) = 1$  so that Eqn (4.3) is obtained. The results of the calculations are:

$z_b$	0.2	0.4	0.6	0.8
$RY_b$	0.353	0.665	0.820	0.882
$RY_o$	0.704	0.358	0.198	0.082
$RYT$	1.06	1.02	1.02	0.96
$k_{bo} = \frac{z_o}{z_b} \times \frac{RY_b}{RY_o}$	2.01	2.79	2.76	2.69
$\alpha_{bo} = k_{bo} \frac{M_b}{M_o}$	1.50	2.08	2.06	2.01

Obviously some smoothing is necessary to obtain a RYT equal to 1 and a constant relative crowding coefficient.

Oats has the highest yield in monoculture, but barley gains in competition.

(27) Replacing  $z_1$  and  $z_2$  by  $Z_1/Z_m$  and  $Z_2/Z_m$ , taking into account that  $Z_1 + Z_2 = Z_m$  and omitting the subscript 1 transforms Eqn (4.5) into

$$0 = \frac{k \times Z/Z_m}{k \times Z/Z_m + (Z_m - Z)/Z_m} M = \frac{k \times Z}{(k - 1) \times Z + Z_m} M$$

This equation must be equal to

$$O = \frac{B \times Z}{B \times Z + 1} O_m$$

Hence B equal to  $(k-1)/Z_m$  and  $O_m$  to  $\frac{k}{k-1}M$ .




Fig. 46

$$\lim_{Z \rightarrow \infty} (O) = O_m$$

$$\lim_{Z \rightarrow 0} (O/Z) = \lim_{Z \rightarrow 0} \frac{B}{B \times Z + 1} O_m = B \times O_m$$

$$\lim_{Z \rightarrow 0} (O/O_m) = \lim_{Z \rightarrow 0} \frac{B \times Z}{B \times Z + 1} = B \times Z$$

$$\lim_{Z \rightarrow \infty} (O/O_m) = \lim_{Z \rightarrow \infty} \frac{B \times Z}{B \times Z + 1} = 1$$

29

	Barley		Oats	
	$O_m$	B	$O_m$	B
7 June	470.	0.083	761.	0.0297
21 June	612.	0.574	552.	0.346
5 July	780.	0.778	724.	0.571
19 July	1132.	0.778	956.	1.17

B has the dimension  $m \text{ row}^{-1}$ , because the seed density has the dimension  $\text{row } m^{-1}$ . The calculated values of  $O_m$  ( $g \text{ m}^{-2}$ ) are very inaccurate for the first harvest because the yields are very far from their maximum at both densities.

The product  $B \cdot O_m$  is then quite accurate, because the denominator in Eqn (4.6) is still close to 1. As pointed out before,  $B \cdot O_m$  is the yield of a single growing plant.

When linearized the growth rate of  $O_m$  for barley and oats are 16.4 and 14.5  $g \text{ m}^{-2} \text{ day}^{-1}$ , so that the estimated values for  $O_m$  at the first harvest are 377 and 333  $g \text{ m}^{-2}$ . The values of B calculated on this basis are 0.11 and 0.076  $m \text{ row}^{-1}$ .

30 The intercept with the ordinate gives  $O_m^{-1}$ , and with the abscissa -B. A disadvantage is that the standard error  $\sigma$  is distorted because  $\sigma$  of  $1/O$  equals  $\sigma$  of  $O$  divided by  $O^2$ .

It is therefore advisable to present the experimental results in two ways:  $1/O$  versus  $1/Z$  and  $O$  versus  $Z$  and to arrive at acceptable parameters by an iterative procedure, going from one graph to the other.

31

$$\begin{aligned} \frac{d(RS)}{dt} &= \frac{d(BxZ)/dt \times (BxZ+1) - d(BxZ+1)/dt \times B \times Z}{(BxZ+1)^2} = \\ &= \frac{Z \times \frac{dB}{dt} \times B \times Z + Z \times \frac{dB}{dt} - Z \times \frac{dB}{dt} \times B \times Z}{(BxZ+1)^2} = \\ &= \frac{B \times Z}{(BxZ+1)} \times \frac{1}{(BxZ+1)} \times \frac{1}{B} \times \frac{dB}{dt} = RS \times (1.-RS) \times \frac{1}{B} \times \frac{dB}{dt} \end{aligned}$$

The dimension of  $(dB/dT)/B$  is  $\text{time}^{-1}$ , the same as a relative growth rate. Usually its value decreases with time. There is exponential growth when this ratio is constant.




Fig. 47

③③ The initial relative growth rate becomes infinite, if B is zero at emergence. The derivative of a variable with respect to time is calculated as the difference between the present value and the value either one sufficiently small time-interval earlier or one ahead, divided by this time-interval, taking care for the sign. In simulation only the first method can be used, as future values are not known. At time zero, however, the past is not known either so that the initial value must be given by the programmer on an INCON card.

The use of the derivative function DERIV is only allowed, if the derivative is taken of an externally given variable, such as an AFGEN function of TIME. In this situation the simulation program is used to convert some given variable time to its derivative with respect to time. As soon as some rate of an integral depends on a derivative of a variable which depends also on some integral, the DERIV function must not be used. Its results are nonsense, because an internal, algebraic loop is introduced.

When the self-adapting integration method of Runge-Kutta is used, the time-interval will be chosen so small that the choice of the initial value has hardly any influence. With METHOD RECT it is better to initialize properly. We calculated the following initial values of the derivative of B:

Barley:  $0.0047 \text{ m row}^{-1} \text{ day}^{-1}$   
 Oats :  $0.0033 \text{ m row}^{-1} \text{ day}^{-1}$

The initial values of RS are calculated with

$$RS = \frac{B \times Z}{B \times Z + 1}$$

with  $Z = 2$  rows  $m^{-1}$  and  $B = 0.001$  m row $^{-1}$ .  
The simulated results in  $g\ m^{-2}$  are:

Date of harvest	Barley	Oats
7 June	60.5	37.1
21 June	242	140
5 July	338	214
19 July	490	338

These are in good agreement with the experimental results of

③④

```

TITLE FOUR COMPETING YEAST SPECIES
INITIAL
INCON YI'1,4'=0.1,0.1,0.1,0.1
DYNAMIC
Y'1,4'=INTGRL(YI'1,4',RY'1,4')
RY'1,4'=RGR'1,4'*Y'1,4'*(1.-RED'1,4')
PARAMETER RGR'1,4'=0.1,0.3,0.4,0.5
RED'1,4'=AFGEN(RDTB'1,4',ALC)
FUNCTION RDTB1=0.,0., 1.5,1.
FUNCTION RDTB2=0.,0., 1.,0.8,2.,1.
FUNCTION RDTB3=0.,0., 0.9,1.
FUNCTION RDTB4=0.,0.,1.,0.5,1.5,1.
ALC=INTGRL(0.,ALCPT)
ALCPT=SUM1(ALCP'1,4')
ALCP'1,4'=ALPF'1,4'*RY'1,4'
PARAMETER ALPF'1,4'=0.5,0.4,0.3,0.2
TIMER FINTIM=50.,PRDEL=1.,OUTDEL=1.
PRINT Y'1,4',ALC
END
STOP
ENDJOB

```

③⑤ The main differences between the MACRO and the INDEX features are simply of a practical nature. The index feature distinguishes the variables by a number at the end. Since a variable name may consist at the most of six alpha-numerical symbols, ABCDE'1,10' creates also the variable ABCDE10 and this

is an unacceptably long name. The INDEX feature can be used to any single expression that is normally used in the program. The MACRO feature is unsuitable for this purpose, because every time a MACRO is used, the invoking sentence has to be written. Hence, the only expression in a 'one line macro' may be as well written directly with the proper symbols. The MACRO feature is therefore in general used only when the MACRO definition contains more than one structural statement.

③⑥ A subroutine is invoked in the execution phase, that means after compilation and loading of the FORTRAN program. A MACRO on the contrary is active before the compilation. A MACRO is an order to write a part of a CSMP program, and only afterwards is the program compiled and executed. Even the sorting is done after activation of the MACRO, so that the statements that occur together in the MACRO, may be scattered all over the FORTRAN UPDATE.

③⑦ The initial value of RS can be calculated with Eqn (4.7). When numerator and denominator are both divided by Z, RS is written as  $B/(B + Z^{-1})$ . Here the inverse of the seed density is equal to the distance between the rows, so that RS is given by  $B/(B + DIST)$ . The initial value of B is much less than 1 cm per row in this example, so that RSI may also be written as  $BI/DIST$ . As pointed out in Exercise 33, DB need not be initialized very accurately so that the average value for DB during the first DELT is certainly good enough. Numerator and denominator are both proportional to DELT, so that DBI is independent of DELT as long as DELT is small enough. Thus no problems occur with METHOD RKS.

```

38) TITLE COMPETITION BETWEEN THREE BARLEY VARIETIES USING THE MACRO FEATURE
MACRO RSI,DBI=BEGIN(BTB,DIST)
      RSI=BI/(BI+DIST)
      BI=AFGEN(BTB,0.)
      DBI=(AFGEN(BTB,DELT)-BI)/DELT
ENDMAC
MACRO O,RS=GROWTH(RSI,DBI,BTB,OMTB)
      RS=INTGRL(RSI,(DB/B)*RS*(1.-SRS))
      B=AFGEN(BTB,TIME)
      DB=DERIV(DBI,B)
      O=RS*AFGEN(OMTB,TIME)
ENDMAC
INITIAL
      RSI1,DBI1=BEGIN(BTB1,DIST1)
      RSI2,DBI2=BEGIN(BTB2,DIST2)
      RSI3,DBI3=BEGIN(BTB3,DIST3)
DYNAMIC
      O1,RS1=GROWTH(RSI1,DBI1,BTB1,OMTB1)
      O2,RS2=GROWTH(RSI2,DBI2,BTB2,OMTB2)
      O3,RS3=GROWTH(RSI3,DBI3,BTB3,OMTB3)
PARAM DIST1=1.2,DIST2=1.2,DIST3=1.2
      SRS=RS1+RS2+RS3
FUNCTION BTB1=(0.,0.001),(30.,.04),(70.,5.)
FUNCTION BTB2=(0.,0.0005),(30.,.02),(70.,2.5)
FUNCTION BTB3=(0.,0.001),(30.,.04),(70.,5.)
FUNCTION OMTB1=(0.,0.),(70.,5600.)
FUNCTION OMTB2=(0.,0.),(70.,5600.)
FUNCTION OMTB3=(0.,0.),(70.,2800.)
TIMER FINTIM=70.,PRDEL=5.,OUTDEL=5.
PRINT O1,O2,O3,SRS
OUTPUT RS1,RS2,RS3
END
STOP
ENDJOB

```

```

TITLE COMPETITION BETWEEN THREE BARLEY VARIETIES USING THE PREPROCESSOR
INITIAL
      RSI'1,3'=BI'1,3'/(BI'1,3'+DIST'1,3')
      BI'1,3'=AFGEN(BTB'1,3',0.)
      DBI'1,3'=(AFGEN(BTB'1,3',DELT)-BI'1,3')/DELT
DYNAMIC
      RS'1,3'=INTGRL(RSI'1,3',(DB'1,3'/B'1,3')*RS'1,3*(1.-SRS))
      B'1,3'=AFGEN(BTB'1,3',TIME)
      DB'1,3'=DERIV(DBI'1,3',B'1,3')
      O'1,3'=RS'1,3*AFGEN(OMTB'1,3',TIME)
PARAM DIST'1,3'=1.2,1.2,1.2
      SRS=SUM1(RS'1,3')
FUNCTION BTB1=(0.,0.001),(30.,.04),(70.,5.)
FUNCTION BTB2=(0.,0.0005),(30.,.02),(70.,2.5)
FUNCTION BTB3=(0.,0.001),(30.,.04),(70.,5.)
FUNCTION OMTB1=(0.,0.),(70.,5600.)
FUNCTION OMTB2=(0.,0.),(70.,5600.)
FUNCTION OMTB3=(0.,0.),(70.,2800.)
TIMER FINTIM=70.,PRDEL=5.,OUTDEL=5.
PRINT O'1,3',SRS
OUTPUT RS'1,3'
END
STOP
ENDJOB

```

③⑨ Thorough stirring does not result in a uniform distribution of the *Paramecium* throughout the liquid medium, but in a random distribution. One-tenth of the solution therefore contains sometimes more or sometimes less than exactly one-tenth of the number of protozoa.

④⑩

④⑪

Variable	Dimension	Type of 'variable'
H	protozoon	state
AFOOD	loop	state
TIME	day	state
CONVF	prot.loop <sup>-1</sup>	param.
RDR	day <sup>-1</sup>	param.
FOOD	loop.volume <sup>-1</sup>	auxil.
RSW	volume.prot. <sup>-1</sup> .day <sup>-1</sup>	param.
MRDIG	loop.prot. <sup>-1</sup> .day <sup>-1</sup>	param.
CNRT	loop.day <sup>-1</sup>	rate

The example of the relational diagram in Fig. 40 is as schematic as possible.

A simplified integral statement for the net growth rate is:

$$H = \text{INTGRL}(IH, \text{CONVF} \times \text{CNRT} - \text{RDR} \times H)$$

if a relative death rate is accounted for and

$$H = \text{INTGRL}(IH, \text{CONVF} \times (\text{CNRT} - \text{MNF} \times H))$$

if maintenance is accounted for by a maintenance factor (MNF in loop.day<sup>-1</sup>.protozoa<sup>-1</sup>).

The two formulations are the same with

$$\text{RDR} = \text{CONVF} \times \text{MNF}$$




Fig. 48

④ The graph consists out of two line segments: a straight line through the origin and a horizontal 'saturation' level. MRDIG determines the height of the saturation level. The slope of the line through the origin is determined by RSW, and H is a multiplication factor for the height of the graph as a whole. CNRT equals zero when FOOD is zero. NGR equals zero when

$$\text{CNRT} \times \text{CONVF} = \text{DR}$$

or

$$H \times \text{AMIN1}(\text{MRDIG}, \text{RSW} \times \text{FOOD}) \times \text{CONVF} = H \times \text{RDR}$$

or

$$\text{AMIN1}(\text{MRDIG}, \text{RSW} \times \text{FOOD}) = \text{RDR}/\text{CONVF}$$

The maximum value of the left side of the expression is MRDIG. If the value of the right side is even larger than MRDIG, no value of FOOD exists for which NGR equals zero. In other words,

the rate of death could be larger than the rate of growth, even if the animals were eating at the top of their consumptive ability. Then rapid extinction would follow, so that there is no need to consider this situation. If, on the other hand, the expression on the right is less than MRDIG, the equation may be simplified to

$$H \times RSW \times FOOD \times CONVF = H \times RDR$$

or

$$FOOD = \frac{RDR}{RSW \times CONVF}$$

This equilibrium level of FOOD is independent of the number of protozoa H. This is logical because each animal catches its food on its own.

④3 In Exercise 41 it is said that the basic unit to measure volumes is the volume of the test tube, and not  $\text{cm}^3$ . This convention must be maintained throughout the simulation program. Numerically there is no objection to eliminate AFOOD and to write directly:

$$FOOD = \text{INTGRL}(L, \text{FEED} - \text{CNRTA} - \text{CNRTC})$$

However, the dimensions of the variables are then not consistent: FOOD sometimes means amount of food, as here in the integral, and sometimes it means density of food, as in the expression for CNRT.

④4 In this program discontinuous changes occur. Rates are calculated as contents divided by DELT. Such situations cannot be handled with METHOD RKS (see also Section 2.4).

④5 Four parameters must be estimated for each species. If the order of magnitude of none of the parameters is known, one may start by estimating four values for each parameter of which the largest is 10 000 times the smallest. Then in a first evaluation as many as  $4^4 = 256$  simulation runs are needed.

④6 Since the data scatter considerably your estimates may differ from those in the following table.

	<u><i>P.aurelia</i></u>		<u><i>P.caudatum</i></u>	
	0.5 loop	1 loop	0.5 loop	1 loop
$H_{eq}$	2500	4500	600	1290
GR.75	350	550	100	230
RGR	1.23	1.23	0.74	1.05
CONVF	2800	2200	800	920
RDR	0.46	0.39	0.57	0.61
MRDIG	0.00064	0.00078	0.00176	0.00191
RSW	0.00128	0.00156	0.00176	0.00191

Averaging of the values for 0.5 and 1 loop, and taking into account that *P.caudatum* is about four times larger than *P.aurelia*, leads to the following estimates of the parameters:

	<i>P.aurelia</i>	<i>P.caudatum</i>
CONVF	3000	750
RDR	0.43	0.59
MRDIG	0.0007	0.0028
RSW	0.0014	0.0018

Some of the reasons why these estimates may be considerably in error are:

- a the scatter of data;
- b the population size in the end is not always at  $H_{eq}$ , but varies throughout the day so that the two equations with  $H_{eq}$  in it are not strictly valid;
- c it is not certain whether in the beginning the saturation density of food is reached or not. The assumption that it is not so leads to a set of equations in which MRDIG is larger than RSW and which provides different values for RSW;
- d in the beginning the density of food varies also throughout the day, so that growth is not exactly exponential;
- e the whole concept, visualized in the simulation program may be wrong.

(47) Other parameters being equal, a difference in the relative death rate of 10% causes a difference of only about 8% in the maximum size of the populations for the monoculture. In the mixed culture, however, it is just this 10% difference that makes for survival of one species and extinction of the other; after several days the difference in the population size is much larger than 8%. The same argument holds for the other parameters. Our best estimates are given in Table 4. Almost the complete program is presented in the text, so that finalizing it, should not give any difficulties.

④⑧ The food consumption (CNRT) is proportional to the number of protozoa (H) and to the minimum of the maximum rate of digestion (MRDIG) or the food concentration times the rate of searching (FOOD x RSW). The food consumption is proportional to H, when FOOD exceeds MRDIG/RSW or when FOOD is constant. FOOD is constant in the beginning, because the consumption is small compared with supply. Since the death rate is also proportional to H, exponential growth results at the beginning of the experiment.

At the end of the experiment the population grows very fast just after feeding. As soon as the food level is below  $RDR/(RSW \times CONVF)$ , the death rate is larger than the growth rate (Exercise 42). When the food is depleted below this critical level, the population size goes through a maximum, and will be smaller at the end of the day than some time earlier. The relative death rate is about  $0.45 \text{ day}^{-1}$  and the relative sampling rate approximately  $0.1 \text{ day}^{-1}$ , so that death through natural causes is far larger than through sampling. It is interesting to remark that Gause did not only discard the sampled amount for practical reasons, but also because he was (unnecessarily) afraid that the natural death rate would be so small that one species would not replace the other in competition.

④⑨ According to the Poisson distribution function, the standard deviation is the square root of the sampled number so that the relative standard deviation is inversely proportional to this square root. The population of *P.aurelia* is about four times larger than of *P.caudatum*, so that its relative standard deviation is about half, as reflected in the scatter of the observations.

⑤⑩ a At  $14^\circ\text{C}$  the function DVRTB is interpolated between the points (12.,0.) and (26.,0.035); so that DVR is  $0.005 \text{ day}^{-1}$ . Accordingly, the development stage at 20 days is 0.1.

b DVR equals 0 at  $7^\circ\text{C}$  and  $0.0225 \text{ day}^{-1}$  at  $21^\circ\text{C}$ . The development stage at 20 days is  $20 \times 0.5 \times 0. + 20 \times 0.5 \times 0.0225$  or 0.225.

In both situations a and b, the average temperature is  $14^\circ\text{C}$ , but development is much faster in b because of the variation in the temperature, combined with a more than linear temperature response.

c At  $30^\circ\text{C}$  the rate of development is  $0.039 \text{ day}^{-1}$ , so that at 20 days a development stage of 0.78 is reached.

d At a temperature of  $40^\circ\text{C}$  DVR equals 0.041 and at  $20^\circ\text{C}$  0.020. Therefore, DVS after 20 days is  $20 \times 0.5 \times (0.04 + 0.020) = 0.61$ . This value is less than in

situation c, although the average temperature was the same in this period. Here the variability has caused a decrease in development, because of the less than linear temperature response in this region.

⑤1 With a constant relative death rate RDR, and no birth, the population as a function of time is

$$H = HI \times e^{-RDR \times TIME}$$

as derived in Section 2.1. When RDR has the dimension year<sup>-1</sup>, the total death during the first year amounts to

$$HI \times (1 - e^{-RDR})$$

so that the relative annual death rate equals

$$1 - e^{-RDR}$$

For small values of RDR this approaches RDR. When RDR is 0.02 the error is only 1%.

When RDR is zero and the relative birth rate equals RBR, the size of the population is

$$H = HI \times e^{RBR}$$

after one year.

The relative annual birth rate is then

$$e^{RBR} - 1.$$

The integration routine METHOD RECT must be used because a division by DELT occurs in the expression for the rates.

⑤2 The data of set 1 must be used. The time interval of integration is a half year because the IMPULS function works at 2.5, 7.5 etc. years. The birth and death rate data can also be used when time-intervals smaller than one year are applied but for large time-intervals they have to be recalculated on that basis. It is a good custom in the Netherlands to maintain graves for a limited period of time. Therefore, the question as to the number of graves may be relevant. To calculate the number

of graves another series of at least ten classes of five years must be introduced. The birth rate of the graves equals the death rate of the population and the 'relative death rate' of graves is zero.

Demographically, there is hardly any difference between death occurring during the first years of life and a decrease in birth rate corresponding with the death rate in excess of the 'normal' death rate during the first year. However, if this correction is made, it must be taken into account that the chances of dying during the first year are not the same for boys and girls, so that the sex ratio has to be corrected accordingly.

The total male and female population after 50 years equals  $10.49 \times 10^6$  and  $10.39 \times 10^6$ , respectively and the total number of graves are  $1.853 \times 10^6$ ,  $4.340 \times 10^6$  and  $7.528 \times 10^6$  when maintained for 10, 25 and 50 years.

The simulation program for the growth of the population is given in Fig. 41. The simulation of the number of graves is programmed in Exercise 61.

The birth rates and death rates per thousand are recalculated on a relative basis in a NOSORT section because statements of the form:

```
MRDR1 = MRDR1 * .....
```

cannot be sorted (Section 2.3).

It would have been also possible to rename the variables.

⑤3 GS =1 means that all seeds are germinated. Simulation beyond this point does not make sense, as far as germination is concerned.

The termination of the program is achieved by inserting a finish card: FINISH GS = 1.

```
TITLE GROWTH OF THE NETHERLANDS POPULATION
PARAM WRDR'1,19'=11.4,1.2,.3,.3,.4,.4,.6,1.,1.5,2.5,4.,5.5,8.,13., ...
      20.,50.,120.,250.,500.
PARAM MRDR'1,19'=15.6,1.8,.5,.5,.7,1.,1.2,1.5,2.2,4.,6.5,9.,11.5, ...
      16.,35.,70.,150.,300.,600.
INCON WI'1,19'=291.,584.,570.,548.,548.,487.,400.,380.,379.,377., ...
      353.,327.,310.,262.,226.,180.,110.,50.,33.
INCON MI'1,19'=305.,612.,597.,575.,576.,517.,429.,399.,382.,367., ...
      338.,306.,280.,223.,184.,150.,90.,40.,13.
PARAM RBR'1,16'=4*0.,.091.,.159.,.152.,.084.,.036.,.01,6*0.
PARAM SEXR=1.048
INITIAL
      FRBOY=SEXR/(1.+SEXR)
      FRGIRL=1./(1.+SEXR)
NOSORT
```

\*CONVERSION OF DEATHS PER THOUSAND PER YEAR TO RELATIVE DEATH RATES

MRDR'1,19'=-ALOG(1.-0.001\*MRDR'1,19')

WRDR'1,19'=-ALOG(1.-0.001\*WRDR'1,19')

\*INITIAL CONSTANTS WERE GIVEN IN THOUSANDS:

WI'1,19'=1000.\*WI'1,19'

MI'1,19'=1000.\*MI'1,19'

DYNAMIC

M1=INTGRL(MI1,MBR - M1\*MRDR1 - MFL1)

W1=INTGRL(WI1,WBR - W1\*WRDR1 - WFL1)

M'2,19'=INTGRL(MI'2,19',MFL'1,18'-MFL'2,19'-M'2,19'\*MRDR'2,19')

W'2,19'=INTGRL(WI'2,19',WFL'1,18'-WFL'2,19'-W'2,19'\*WRDR'2,19')

TBR=SUMX(W'1,16',RBR'1,16')

WBR=FRGIRL\*TBR

MBR=FRBOY\*TBR

WFL'1,19'=PUSHD\*(W'1,19' - WRDR'1,19'\*W'1,19'\*DELT)

MFL'1,19'=PUSHD\*(M'1,19' - MRDR'1,19'\*M'1,19'\*DELT)

PUSHD=PUSH/DELT

PUSH=IMPULS(2.5,5.)

TW=SUM1(W'1,19')

TM=SUM1(M'1,19')

TP=TM + TW

METHOD RECT

TIMER FINTIM=50.,DELT=.5,OUTDEL=5.,PRDEL=5.

PRINT TP,TW,TM,TBR,M'1,19',W'1,19'

OUTPUT TP,TW

END

STOP

ENDJOB

Fig. 49

(54) It is a rather arbitrary choice to give the value 1 to the stage of germination; it could just as well be 1000. Whatever the value, it is passed going through N classes. Accordingly each class covers the chosen germination value divided by N.

(55) The simulation program may read as follows.

PARAMETER N=10

H1=INTGRL(1000., - FLOW2)

H'2,10'=INTGRL(0.,FLOW'2,10' - FLOW'3,11')

FLOW'2,11'=H'1,10'\*PUSHD

PUSHD=INSW(GS - 1./N,0.,1.)/DELT

GS=INTGRL(0., VDV - PUSHD/N)

PARAMETER VDV=0.143

METHOD RECT

The seeds germinate at  $1./0.143=7.0$  days.

⑤6 The average germination period is not necessarily the moment when 50% is germinated. When G is the rate of germination, the mathematical definition of the average germination period is:

$$AGP = \frac{\int_0^{\infty} G \times T \times dT}{\int_0^{\infty} G \times dT}$$

To execute this calculation, the curve is divided into sections of one day and the formula

$$AGP = \frac{\sum_{n=0}^{27} G_n \times T_n}{\sum_{n=0}^{27} G_n}$$

is used, in which n is the number of the day.

Because the cumulative curve adds up to 100%, the denominator in this formula is 100. AGP appears to be 16.4 day.

Similarly, the formula for the variance (the square of the standard deviation) is:

$$VAR = \frac{\sum_{n=0}^{27} G_n \times (T_n - AGP)^2}{\sum_{n=0}^{27} G_n}$$

This value is 11.91, so that the standard deviation is 3.45 day.

⑤7 It is assumed that germination on the nth day means germination at the beginning of the nth day. The average germination periods in days are then:  $100 \times 5/100 = 5$ ,  $100 \times 10/100 = 10$ ,  $(50 \times 5 + 50 \times 10)/100 = 7.5$ ,  $(75 \times 5 + 25 \times 10)/100 = 6.25$ . If it is assumed that germination occurs during the nth day, 0.5 day must be added to these values. The dimension of the relative germination rate is  $\text{day}^{-1}$  and of the average germination period is days. The product is therefore dimensionless. Its value appears to be approximately one. The average germination period is

$$AGP = -\frac{1}{HI} \int_0^{\infty} T \times \frac{dH}{dT} \times dT = \frac{1}{HI} \int_0^{HI} T \times dH$$

The integral is the area below the H versus T curve, so that the expression can be replaced by

$$AGP = \frac{1}{HI} \int_0^{\infty} H \times dT = \frac{1}{HI} \int_0^{\infty} HI \times e^{-RDV \times T} \times dT =$$

$$= -\frac{1}{RDV} \times e^{-RDV \times T} \Big|_0^{\infty} = \frac{1}{RDV}$$

Hence  $RDV \times AGP = 1$ .

⑤⑧ The time constant of each integral is found by reducing its equation to (Section 2.3).

$$H_i = \text{INTGRL} (HI_i, H_i/TAU_i + \text{all other influences})$$

For all integrals TAU appears to be equal to REST, so that the time constant of the whole system is also equal to the residence time in each class.

⑤⑨ According to Eqn (6.2), F should be negative. Obviously the number of classes that is chosen is too large to obtain a relative dispersion of 0.25. With  $N = 16$  and  $F = 0$ , the proper dispersion is obtained.

⑥⑩

TIME	H1	H2	H3	H4
0	1	0	0	0
0.5 x REST	0.5	0.5	0	0
1.0 x REST	0.25	0.5	0.25	0
1.5 x REST	0.125	0.375	0.375	0.125

This table satisfies the binomial probability distribution function.

B equals  $TIME/(F \times REST)$  and f equals F, so that  $f \times B = TIME/REST$ . With  $f \times B$  small and constant and B increasing, the binomial approaches the Poisson distribution function. This situation is achieved here when the lowest value of F i.e.  $DELTA/REST$  is substituted in the expression for B, and DELTA approaches zero. Then B equals  $TIME/DELTA$  and approaches infinity. The expectation value and the variance of the Poisson distribution are  $f \times B$  and  $(1 - f) \times f \times B$  or in the other terms  $TIME/REST$  and  $(1 - F) \times TIME/REST$ . To convert the variance in terms of time, rather than class, the expression must be multiplied by  $REST^2$ . Then  $S^2$  becomes  $(1 - F) \times TIME \times REST$  which reduces into Eqn (6.2) when TIME is replaced by AGP. It is also recalled that for high expectation values the Poisson function approaches the Gaussian function with a variance equal to the

expectation value of the mean.

⑥1 The age-classes are now indeed 0-5, 5-10, 10-15 etc. so that set 2 with the class centres at 2.5, 7.5, ... years must be used. The residence time is 5, and the value of F is of course 1 because age-classes advance per definition without dispersion. The simulation program, including the number of graves is given in Fig. 42.

⑥2 The net flow rate of the first layer, NFL1, is calculated as FLW1 - FLW2. The choice of the signs means that FLW1 goes into the first (and topmost) layer and that FLW2 leaves it. Therefore the direction of the flow is downwards. Of course, upward flows are also possible but then they are negative. This rule for the sign must be taken into account when the expression for FLW is written.

⑥3 The flow into the first layer is governed by the temperature difference between the surface of the soil and the centre of the first layer. The distance between these levels is only half of the thickness of the compartment.

The unit of time is second, as follows from the definition of TMPS. Every time the argument of a sine has the value  $2\pi$  (= 6.2832), a full cycle is completed.

TAV represents the average temperature of the soil surface, and TAMPL is the amplitude of the sine wave.

⑥4

Variable	Dimension
TMP, TI, TMPS, TAV, TAMPL	$^{\circ}\text{C}$
TCOM	m
VHCAP	$\text{J m}^{-3} \text{ }^{\circ}\text{C}$
HC, HCI	$\text{J m}^{-2}$
COND	$\text{J m}^{-1} \text{ }^{\circ}\text{C}^{-1} \text{ s}^{-1}$
FLW, NFL	$\text{J m}^{-2} \text{ s}^{-1}$

It is extremely important to be aware of the units, as the numerical value of the properties depends on the units. It is recommended to use a consistent set of units: the international system with kg, m, s.

⑥5 The easiest method is to specify the conductivity and the heat capacity as a function of depth in an AFGEN function. A new variable for depth, DPT must then be introduced, whereby DPT1 equals  $0.5 \times \text{TCOM}$ , etc. Also TCOM can be varied with the number of the layer. TCOM1, TCOM2, etc. must then be specified.

TITLE GROWTH OF THE NETHERLANDS POPULATION II  
PARAM WRDR'1,19'=4.0,.8,.3,.3,.4,.5,.8,1.2,2.,3.2,4.7,6.7,10.5, ...  
16.5,35.,85.,180.,380.,760.  
PARAM MRDR'1,19'=6.0,.7,.5,.6,.9,1.,1.4,1.8,3.1,5.2,7.8,10.7,13.7,...  
25.5,52.,110.,200.,400.,900.  
INCON WI'1,19'=582.,587.,553.,543.,554.,420.,380.,381.,378.,376., ...  
330.,323.,298.,226.,226.,150.,70.,25.,13.  
INCON MI'1,19'=611.,613.,580.,569.,583.,452.,405.,393.,371.,362., ...  
314.,297.,262.,184.,184.,120.,60.,20.,3.  
PARAM RBR'1,16'=3\*0.,.022,.137,.188,.113,.055,.016,.002,6\*0.  
PARAM SEXR=1.048

INITIAL

FRBOY=SEXR/(1.+SEXR)  
FRGIRL=1./(1.+SEXR)

NOSORT

\* CONVERSION OF DEATHS PER THOUSAND PER YEAR TO RELATIVE DEATH RATES

MRDR'1,19'=-ALOG(1.-0.001\*MRDR'1,19')  
WRDR'1,19'=-ALOG(1.-0.001\*WRDR'1,19')

\* INITIAL CONSTANTS WERE GIVEN IN THOUSANDS:

WI'1,19'=1000.\*WI'1,19'  
MI'1,19'=1000.\*MI'1,19'

DYNAMIC

MO=INTGRL(0.,MBR - MFLO)  
WO=INTGRL(0.,WBR - WFLO)  
M1=INTGRL(MI1,MFLO - MFL1 - DRM1)  
W1=INTGRL(WI1,WFLO - WFL1 - DRW1)  
M'2,19'=INTGRL(MI'2,19',MFL'1,18'-MFL'2,19'-DRM'2,19')  
W'2,19'=INTGRL(WI'2,19',WFL'1,18'-WFL'2,19'-DRW'2,19')  
TBR=SUMX(W'1,16',RBR'1,16')  
WBR=FRGIRL\*TBR  
MBR=FRBOY \* TBR  
WFLO=WO/2.5  
MFLO=MO/2.5  
WFL'1,19'=PUSHD\*(W'1,19' - DRW'1,19'\*DELT)  
MFL'1,19'=PUSHD\*(M'1,19' - DRM'1,19'\*DELT)  
PUSHD=PUSH/DELT  
PUSH=IMPULS(2.5,5.)  
DRM'1,19'=M'1,19'\*MRDR'1,19'  
DRW'1,19'=W'1,19'\*WRDR'1,19'  
TDRW=SUM1(DRW'1,19')  
TDRM=SUM1(DRM'1,19')  
TM19=SUM1(M'1,19')  
TW19=SUM1(W'1,19')  
TM=TM19 + MO  
TW=TW19 + WO  
TP=TM + TW  
TDR=TDRM + TDRW + MFL19 + WFL19  
GO=INTGRL(0.,TDR - GO/2.5)  
G1=INTGRL(0.,GO/2.5 - FLG1)  
G'2,10'=INTGRL(0.,FLG'1,9'-FLG'2,10')  
FLG'1,10'=G'1,10'\*PUSHD  
TG10=GO + G1 + G2  
TG25=TG10 + G3 + G4 + G5  
TG50=TG25 + G6 + G7 + G8 + G9 + G10

TIMER FINTIM =50.,DELT=0.5,PRDEL=5.,OUTDEL=5.

METHOD RECT

PRINT TM,TW,TP,TBR,M'1,19',W'1,19',MO,WO,TG10,TG25,TG50

END

STOP

ENDJOB

Fig. 50

⑥⑥ This is very dangerous because the responses to temperature humidity, wind and probably radiation are likely to be non-linear (Compare with Exercise 50).

⑥⑦ \* WEATHER  
 FUNCTION TEMPT=(0.,14.),(12.,35.),(24.,14.)  
 FUNCTION WINDT=(0.,1.),(6.,1.),(14.,4.),(19.,2.),(24.,1.)  
 FUNCTION WETT=(0.,1.),(7.99,1.),(8.,0.),(19.99,0.),(20.,1.),(24.,1.)  
 FUNCTION LITET=(0.,-1.),(5.99,-1.),(6.,1.),(20.,1.),(29.01,-1.), ...  
 (24.,-1.)  
 FUNCTION RAIN=(0.,0.),(24.,0.)  
 FUNCTION LAIT=(0.,3.),(140.,3.)  
 HOUR=24.\*AMOD(TIME,1.)  
 LAI=AFGEN(LAIT,TIME)  
 WET=FCNSW(WETX + RAIN, 0.,0.,1.)  
 WETX=AFGEN(WETT,HOUR)  
 TEMP=AFGEN(TEMPT,HOUR)  
 WIND=AFGEN(WINDT,HOUR)  
 LITE=AFGEN(LITET,HOUR)  
 DRY=1.-WET  
 RAIN=AFGEN(RAIN, HOURS)

⑥⑧ The expressions are similar because the growth rate of the lesions is linearly dependent on the difference between their maximum area and actual area and because the maximum area of the lesions is the same.

\* GROWTH OF LESIONS  
 NIL'1,6'=INTGRL(NILI'1,6',FNIL'1,6' - FNIL'2,7')  
 FNILI=RTN  
 NVL=INTGRL(0.,FNIL7)  
 FNIL'2,7'=2.\*NIL'1,6'  
 AL=INTGRL(0.,RAL)  
 RAL=PAL\*(MALS\*NVL - AL)  
 PAL=AFGEN(PALT,TEMP)  
 FUNCTION PALT=0.,0.,10.,14,18.,33,23.,8,30.,8,35.,14,40.,0.  
 PARAM MALS=1.E-8

The total residence time of the lesions in the non-visible stage is  $6 \times 0.5 = 3$  days. Because multiplication takes less computing time than division, the contents of each class is multiplied by  $2 \text{ day}^{-1}$  rather than divided by 0.5 day.

Some lesions are already visible after 1.5 day because of the dispersion during passage through the six compartments. If we assume that DELT is small compared with 0.5 day, the standard deviation of lesion appearance is calculated with Eqn (6.1). The curve for total lesion area (AL) is sigmoid for two reasons: the dispersed lesion appearance and the proportionality of growth with the difference of maximum and current area.

(69) \* FORMATION OF GREEN STALKS  
 ROP=MOA\*RAL  
 ROP1=ROP + SROP  
 SROP=SUM1(EMPT'1,4')  
 ROP'2,4'=OP'1,3'\*16.\*WET  
 OP'1,3'=INTGRL(0.,ROP'1,3'-ROP'2,4'-EMPT'1,3')  
 OP4 =INTGRL(0.,ROP4 - ROX - ROG - EMPT4)  
 EMPT'1,4'=DRY OP'1,4' INVD  
 PARAM MOA=300.E10  
 POG=INSW(LITE,AFGEN(POGD,TEMP),AFGEN(POGL,TEMP))\*WET  
 POX=INSW(LITE,AFGEN(POXD,TEMP),AFGEN(POXL,TEMP))\*WET  
 FUNCTION POGL=(0.,0.),(14.,.04),(18.,.12),(23.,1.4),(30.,1.2),(35.,0.)  
 FUNCTION POXL=(0.,0.),(14.,.04),(18.,.12),(23.,1.4),(35.,0.)  
 FUNCTION POGD=(0.,0.),(14.,.1),(18.,.27),(23.,.27),(30.,1.33), ...  
 (35.,.67),(40.,0.)  
 FUNCTION POXD=(0.,0.),(14.,.02),(18.,.03),(23.,.18),(30.,.88), ...  
 (35.,1.54),(40.,0.)  
 ROG=POG\*OP4  
 ROX=POX\*OP4

Note that the division with DELT is replaced by multiplication with INVD, a parameter which is in the initial section defined as 1/DELT and that 16 is the inverse of the residence time of 0.0625 day. The dimension of ROG equals number of green stalks per ha soil surface per day. POG and POX in the light at 21 °C are the same, so that only half of the stalks that could potentially develop, reach the green stage.

(70) The number of spores after time T equals

$$S_T = S_0 \times e^{-RBETR \times T}$$

so that

$$\begin{aligned}
 RBETR &= - (\ln(S_T/S_0))/T = - (\ln(1 - 0.63))/5 \\
 &= 1/5 \text{ h}^{-1} = 4.8 \text{ day}^{-1}
 \end{aligned}$$

which may according to the function BEATT, be caused by a rainfall of 16 mm h<sup>-1</sup>.

(71) \* FORMATION OF SPORES ON GREEN STALKS  
 GST'1,3'=INTGRL(0.,RGST'1,3'-RGST'2,4'-DGST'1,3'-BGST'1,3')  
 DGST'1,3'=DRY\*GST'1,3'\*INVD  
 RGST1=ROG  
 RGST'2,3'=GST'1,2'\*16.\*WET  
 RGST4=RGS  
 RGS=GST3\*INSW(LITE,AFGEN(PGSD,TEMP),AFGEN(PGSL,TEMP))\*WET  
 FUNCTION PGSL=(0.,0.),(14.,.15),(18.,1.44),(23.,.32),(30.,0.),(40.,0.)  
 FUNCTION PGSD=(0.,0.),(14.,.06),(18.,.14),(23.,.14),(30.,.44),(40.,0.)  
 BGST'1,3'=RBETR\*GST'1,3'  
 RBETR=AFGEN(BEATT,RAIN)  
 FUNCTION BEATT=(0.,0.),(0.25,.08),(0.75,.32),(6.25,2.),(18.8,5.6), ...  
 (25.,6.7)  
 NGST=SUM1(GST'1,3')

The dimension of RGS is spores on green stalks per hectare soil surface per day

72 \* FORMATION OF SPORES ON DRIED STALKS  
 DGST=SUM1(DGST'1,3')  
 DDST=SUM1(DDST'1,4')  
 RDST1=DGST + DDST + RSR  
 RDST5=RDS  
 DST'1,4'=INTGRL(0.,RDST'1,4'-RDST'2,5'-DDST'1,4'-BDST'1,4')  
 RDST'2,4'=DST'1,3'\*16.\*WET  
 DDST'1,4'=DRY\*DST'1,4'\*INVD  
 BDST'1,4'=RBETR\*DST'1,4'  
 RDS=DST4\*INSW(LITE,AFGEN(PDSD,TEMP),AFGEN(PDSL,TEMP))\*WET  
 FUNCTION PDSL=(0.,0.),(14.,.17),(18.,1.75),(23.,.25),(30.,0.), ...  
 (40.,0.)  
 FUNCTION PDSD=(0.,0.),(14.,.07),(18.,2.95),(23.,2.2),(30.,.53), ...  
 (35.,0.),(40.,0.)  
 NDST=SUM1(DST'1,4')

The rate of spore removal (RSR) must still be calculated.

73 Again

$$S_T = S_0 \times e^{-RWASH \times T}$$

with T in days. Hence

$$RWASH = -(\ln(1 - .86))/(3/24) = 15.7 \text{ day}^{-1}$$

$$74 \quad RBLOW = -\ln(1 - .05)/(3/24) = 0.408 \text{ day}^{-1}$$

75 \* SPORE DISPERSAL  
 RSP=RDS + RGS  
 STSP=INTGRL(0.,RSP - RSR)  
 RSR=SPRR + SPRD  
 SPRR=RWASH\*STSP  
 SPRD=RBLOW\*STSP\*DRY  
 RWASH=RAIN\*2.62  
 RBLOW=0.102\*AMAX1(1.,WIND\*WIND)  
 RASP=SPRR\*REFF + SPRD\*WEFF  
 REFF=AFGEN(REFFT,RAIN)  
 FUNCTION REFFT=(0.,0.2),(2.5,0.003),(10.,0.003)  
 WEFF=LAI\*0.01

Host exhaustion could be taken into account by reducing the leaf area (LAI) by the area of the lesions (AL). If the disease is so severe that AL is not small compared with LAI, the disease should also feed back on the growth of the crop.

```

) * GERMINATION AND PENETRATION
SF=INTGRL(0.,RASP - RKSP - RG - REXS - WSF)
WSF=RWASH*SF
RG=SF*AFGEN(PFTT,TEMP)*(1.-KILL)*WET
REXS=SF*AFGEN(PFXT,TEMP)*(1.-KILL)*WET
FUNCTION PFTT=(0.,0.),(10.,.4),(15.,1.8),(20.,4.6),(23.,7.), ...
(35.,3.7),(40.,0.)
FUNCTION PFXT=(0.,0.),(10.,0.),(15.,1.8),(20.,4.2),(23.,2.6), ...
(35.,3.7),(40.,0.)
RKSP=KILL*SF*INVD
PROCEDURE KILL=DESS(WET)
KILL=0,
IF((WETP-WET).GT. 0.01) KILL=1,
WETP=WET
ENDPRO
GT=INTGRL(0.,RG - RKGT - RTN - PEXT - WGT)
WGT=RWASH*GT
RKGT=KILL*GT*INVD
REXT=GT*AFGEN(PTXT,TEMP)*(1.-KILL)*WET
RTN=GT*AFGEN(PTNT,TEMP)*(1.-KILL)*WET
FUNCTION PTNT=(0.,0.),(18.,.48),(23.,.65),(30.,.25),(35.,0.),(40.,0.)
FUNCTION PTXT=(0.,0.),(18.,1.3),(23.,2.6),(30.,2.2),(35.,0.),(40.,0.)

```

Note that RG and some other rates are also multiplied by (1 - KILL) to avoid that the same spores or germs are killed upon desiccation and transferred at the same time.

77 Another rate (INVR) has to be added to the integral of the lesions on the foliage (SF). This rate equals

$$\text{INVR} = 1.E6*24*DRY*INSW(\text{LITE}, 0., 1.)*INSW(\text{TIME}-7.,1.,0.)$$

78

```

OUT1=DEBUG(10,0.)
OUT2=DEBUG(2,5.)
OUT3=DEBUG(2,5.5)

```

79  $\text{TEMP} = \text{AVTMP} + \text{AMPL} * \text{SINE}(6.2832 * (\text{TIME} + 8/24))$

80 For JS and 30 °C the number of classes is  $(2.0/0.5)^2 = 16$  (see Eqn 6.1), for all other situations the number of classes is higher. When the temperature varies, the relative dispersion is not constant either, so that the boxcar train with constant relative dispersion cannot be used. Then the method with controlled dispersion must be used, in which the dispersion can be varied by using a factor F. The number of classes must then be smaller than the number minimally required in the constant relative dispersion method. A good choice is three quarters of the lowest value, so that here N can be chosen as 12.

The fraction F is given by

$$F = 1 - N\left(\frac{s}{x}\right)^2$$

F would become negative, if N was too large in relation to s/x  
A simulation model for the hatching process is

```
TITLE HATCHING PROCESS WINTER EGGS
EGGO=INTGRL(0.,-FLOW1)
FLOW1=EGGO*2./(F*REST)
EGG'1,12'=INTGRL(EGGI'1,12',FLOW'1,12'-FLOW'2,13')
FLOW'2,13'=EGG'1,12'*PUSHDF
PUSHDF=PUSH*F/DELT
REST=HATP/12
HATP=AFGEN(HATPT,TEMP)
FUNCTION HATPT=(15.,18.),(18.,10.),(25.,6.3),(30.,5.0)
DISP=AFGEN(DISP,TEMP)
FUNCTION DISP=(15.,3.),(18.,1.6),(25.,1.06),(30.,1.25)
TEMP=AVTMP + AMTMP*SIN(6.2832*TIME/24.)
PARAM AVTMP=22.5, AMTMP=7.5
F=1.-12*(DISP/HATP)**2
PUSH=INSW(GS-1.,0.,1.)
GS=INTGRL(0.5,1./(F*REST)-PUSH/DELT)
INCON EGGI'1,12'=1000.,11 0.
EGG=SUM1(EGG'1,12')
LARV=INTGRL(0.,FLOW1)
OUTPUT LARV,EGG
METHOD RECT
TIMER FINTIM=20., DELT=0.01, OUTDEL=1.
END
STOP
ENDJOB
```

The relative rate of mortality is calculated with:

$$RMOR = \ln\left(\frac{y_1}{y_2}\right)/(t_2 - t_1)$$

in which  $y_1$  is the number of eggs at  $t_1$  and  $y_2$  the number at  $t_2$ . During development at 30 °C which lasts five days (Table 7) 50% dies. Thus the relative mortality ( $\text{day}^{-1}$ ) amounts to:

$$RMOR = \frac{\ln(2)}{5} = 0.139$$

81

\*\*\*LISTING OF PROGRAM\*\*\*

```
INDEX=0
TITLE GROWTH AND DEVELOPMENT P. ULMI
STORAGE Y(100)
FIXED N1,N2,INDEX,I
INCON EGGI=1000.
OUTE,EGG=BOXCAR(EGGI,RESE,SDE,RME,0.,4)
OUTL,JUV=BOXCAR(0.,RESJ,SDJ,RMJ,OUTE,6)
OUTS,JUVS=BOXCAR(0.,RESJS,SDJS,RMJS,OUTL,6)
AF=INTGRL(0.,OUTS)
PRINT EGG,JUV,JUVS,AF
METHOD RECT
TIMER FINTIM=40.,DELT=0.05,PRDEL=2.
TEMP=AVTMP+AMTMP*SIN(6.2832*TIME/24.)
PARAM AVTMP=22.5,AMTMP=7.5
RESE=AFGEN(RESET,TEMP)
SDE=AFGEN(SDET,TEMP)
RME=AFGEN(RMET,TEMP)
RESJ=AFGEN(RESJT,TEMP)
SDJ=AFGEN(SDJT,TEMP)
RMJ=AFGEN(RMJT,TEMP)
RESJS=AFGEN(RESJST,TEMP)
SDJS=AFGEN(SDJST,TEMP)
RMJS=AFGEN(RMJST,TEMP)
AFGEN RESET=15.,18.,18.,10.,25.,6.3,30.,5.
AFGEN SDET=15.,3.,18.,1.6,25.,1.06,30.,1.25
AFGEN RESJT=15.,10.1,18.,6.5,25.,4.,30.,2.5
AFGEN SDJT=15.,2.,18.,1.3,25.,0.8,30.,0.62
AFGEN RESJST=15.,5.5,18.,4.3,25.,2.4,30.,2.
AFGEN SDJST=15.,1.1,18.,0.85,25.,0.6,30.,0.5
AFGEN RMET=15.,0.01,18.,0.01,30.,0.01
AFGEN RMJT=15.,0.02,18.,0.02,25.,0.03,30.,0.04
AFGEN RMJST=15.,0.03,18.,0.02,25.,0.02,30.,0.03
END
STOP
```

82 Within the subroutine BOXCAR all classes have the same physiological properties. Of course it is possible to calculate an average rate of oviposition, but for each age distribution the average rate of oviposition is different. Therefore the boxcar train must be programmed according to the basic method given in Chapter 6.

⑧ With the following statements the extended program of exercise 81 also accounts for ageing of the females and for oviposition. The data of table 8 and table 9 are used to define the relations of oviposition rate and relative mortality rate with temperature.

```

AF' 1,5' = INTGRL(0.,RAF'1,5')
RAF1 = OUTS-DAF1-MORA1
DAF' 1,5' = PUAF * (AF'1,5'*DELX - MORA'1,5')
RAF' 2,5' = DAF'1,4'-DAF'2,5'-MORA'2,5'
MORA' 1,5' = AFGEN(MORT'1,5',TEMP)
PUAF = INSW(GSA-1.,0.,1.)
GSA = INTGRL(0.,5./LOAF-PUAF/DELT)
OUTE,EGG = BOXCAR(EGGI,HATP,DISP,RME,REPR,12)
REPR = SUMX(AF'1,5',REP'1,5')
REP' 1,5' = AFGEN(REPT'1,5',TEMP)
LOAF = AFGEN(LOAFT,TEMP)
AFGEN LOAFT = 10.,67.5,15.,61.,20.,28.,25.,20.,30.,14.6
AFGEN MORT1 = 10.,0.011,15.,0.011,20.,0.005,25.,0.005,30.,0.003
AFGEN MORT2 = 10.,0.033,15.,0.031,20.,0.049,25.,0.050,30.,0.027
AFGEN MORT3 = 10.,0.088,15.,0.089,20.,0.161,25.,0.245,30.,0.273
AFGEN MORT4 = 10.,0.15,15.,0.138,20.,0.360,25.,0.504,30.,0.709
AFGEN MORT5 = 10.,0.2,15.,0.15,20.,0.4,25.,0.6,30.,0.8
AFGEN REPT1 = 10.,0.6,15.,1.2,20.,1.9,25.,3.1,30.,4.2
AFGEN REPT2 = 10.,0.5,15.,1.2,20.,2.2,25.,3.7,30.,5.5
AFGEN REPT3 = 10.,0.5,15.,1.1,20.,1.8,25.,3.1,30.,3.8
AFGEN REPT4 = 10.,0.4,15.,1.0.,20.,1.4,25.,2.0,30.,1.4
AFGEN REPT5 = 10.,0.2,15.,0.8,20.,0.8,25.,1.0,30.,0.8

```

⑧④ This simulation model is:

```
TITLE PREY-PREDATOR
PARAM PREY=10., PRED=1.
STPRED=INTGRL(IST,INCST-DECST)
DECST=STPRED*RRDIG
RRDIG=AFGEN(RRDIGT,TEMP)
INCST=PREY*PRED*RRPRED
AFGEN RRDIGT=0.,0.,15.,.5,20.,1.,25.,2.
RRPRED=SUCR*COIN
SUCR=AFGEN(SUCRT,STPRED)
COIN=AFGEN(COINT,TEMP)
AFGEN SUCRT=0.,0.8,1.,0.01
AFGEN COINT=15.,0.01,20.,0.08,25.,0.2
TEMP=AVTMP + AMTMP*SIN(6.2832*TIME/24.)
PARAM AVTMP=20.,AMTMP=5.
TIMER FINTIM=20., DELT=.01, PRDEL=1.
PRINT STPRED
END
STOP
ENDJOB
```

The dimensions of the variables are:

Variables	Dimensions
PREY	$m^{-2}$
PRED	$m^{-2}$
RRDIG	$day^{-1}$
INCST	gut $day^{-1} m^{-2}$
DECST	gut $day^{-1} m^{-2}$
STPRED	gut $m^{-2}$
RRPRED	gut $m^2 day^{-1}$
SUCR	-
COIN	gut $m^2 day^{-1}$

⑧⑤ The relative space occupied by plants has been built up during growth, so that it contains historical information. In Fig. 10 the B curves for barley and oats cross at Day 55. At this moment the space occupied by barley is much larger because its B has always been larger before. In the acarine system, the gut content is in equilibrium with the density of prey, so that past information is forgotten.

⑧⑥ At very low density  $D_m$  the ratio  $(B_1 + 1/D_m)/(B_2 + 1/D_m)$  approaches one, so that both prey types have the same preference. The predator is then so hungry that it just accepts everything. For very high densities,  $k_{12}$  approaches  $B_1/B_2$  according to the

equation used. It is, however, possible that with a mixed prey population the gut content of the predator exceeds that attainable with the unattractive type alone. The predator then only eats attractive preys, so that  $k_{12}$  is infinitely large. Clearly this attempt to bypass the gut content then breaks down and cannot be used, so that the gut content must be explicitly modelled as a state variable.

# References

- Arnold, G.W. & C.T. de Wit (Eds), 1976. Critical evaluation of systems analysis in ecosystems research and management. Pudoc, Wageningen, 114 pages.
- Baeumer, K. & C.T. de Wit, 1968. Competitive interference of plant species in monocultures and in mixed stands. Neth. J. agric. Sci. 16: 103-122.
- Berger, R.D., 1970. Forecasting *Helminthosporium turcicum* attacks on Florida sweet corn. Phytopathology 60: 1284.
- Bergh, J.P. van den, 1968. An analysis of yields of grasses in mixed and pure stands. Versl. Landbouwk. Onderz. (Agr. Res. Rep.) 714, Wageningen.
- Brennan, R.D. & M.Y. Silberberg, 1968. The system/360 continuous modeling program. Simulation 11: 301-311.
- Donald, C.M., 1963. Competition among crop and pasture plants. Adv. Agronomy 15: 1-118.
- Faddeev, D.K. & V.N. Faddeeva, 1964. Computational methods of linear algebra. W.H. Freeman Co., San Francisco.
- Forrester, J.W., 1961. Industrial Dynamics. MIT-press, Boston.
- Fransz, H.G., 1974. The functional response to prey density in an acarine system. Pudoc, Wageningen, 149 pages.
- Gause, G.F., 1934. The struggle for existence. Williams and Wilkins, Baltimore.
- Goudriaan, J., 1973. Dispersion in simulation models of population growth and salt movement in the soil. Neth. J. agri. Sci. 21: 269-281.
- Goudriaan, J., 1977. Crop micrometeorology: a simulation study. Pudoc, Wageningen, 249 pages.
- IBM, 1975. Continuous System Modeling Program III. Program Reference Manual SH19-7001-2. Techn. Publ. Dept., White Plains, USA.
- Jansen, J.G.M., 1974. Simulation of germination of winter annuals in relation to microclimate and micro-distribution. Oecologia 14: 197-228.
- Jansen, J.G.M., 1974. Effect of light, temperature and seed age on the germination of the winter annuals *Veronica arvensis* and *Myosotis ramosissima* Rochel ex. Schult. Oecologia 12: 141-146.
- Lotka, A.J., 1925. Elements of physical biology. Williams and Wilkins, Baltimore.

- Milne, W., 1960. Numerical solution of differential equations. McGraw-Hill, New York.
- Rabbinge, R., 1976. Biological control of fruit-tree red spider mite. Pudoc, Wageningen, 228 pages.
- Royle, D.J., 1973. Quantitative relationships between infection by the hop downy mildew pathogen, *Pseudoperonospora humuli* and weather and inoculum factors. *Ann. appl. Biol.* 73: 19-30.
- Sandfach, J., 1970. An analysis of competition between some barley varieties. Danish Atomic Energy Commission, Risø Report 230: 1-114.
- Shaner, G.E., Newman, Stirm & Lower, 1972. EPIMAY, an evaluation of a plant disease display model. *Bull. Agr. Exp. Sta. Purdue Univ., West Lafayette, Indiana, USA.*
- Unesco, 1972. Expert Panel on the role of systems analyses and modeling approaches in the programme on man and biosphere. MAB report series No. 2, Paris.
- Volterra, V., 1931. Variations and fluctuations of the number of individuals in animal species living together. In R. N. Chapman, 'Animal Ecology', McGraw-Hill, New York.
- Waggoner, P.E., J.G. Horsfall & R.J. Lukens, 1972. EPIMAY, a simulator of southern corn leaf blight. *Bull. Conn. Agr. Exp. Sta. New Haven, Conn., USA.*
- Wit, C.T. de, 1960. On competition. *Versl. Landbouwk. Onderz. (Agric. Res. Rep.) 66.8, Wageningen.*
- Wit, C.T. de, R. Brouwer & F.W.T. Penning de Vries, 1970. The simulation of photosynthetic systems. In: *Prediction and Measurement of Photosynthetic Productivity. Proc. of the IBP/PP Technical Meeting, Třeboň, Sept. 1969: 47-70.*
- Wit, C.T. de & H. van Keulen, 1972. Simulation of transport processes in soils. Pudoc, Wageningen, 109 pages.
- Wit, C.T. de, P.G. Tow & G.C. Ennik, 1966. Competition between legumes and grasses. *Versl. Landbouwk. Onderz. (Agric. Res. Rep.) 687, Wageningen.*

# Appendix A

## THE PREPROCESSOR

THE PREPROCESSOR IS A TEXT HANDLING PROGRAM  
DEVELOPED AT: UNIVERSITY COMPUTING CENTRE, WAGENINGEN.  
IT IS AVAILABLE AT A SMALL CHARGE.

THE PREPROCESSOR ENABLES THE USER TO WRITE STATEMENTS  
THAT MUST BE REPEATED SEVERAL TIMES,  
FOR INSTANCE WITHIN A CSMP PROGRAM.  
THE FOLLOWING EXAMPLE SHOWS HOW A NOTATION WITH APOSTROPHES IS USED.

THE INPUT OF THE PREPROCESSOR IS: THE SOURCE PROGRAM  
AND THE OUTPUT IS: THE RESULTING CSMP

THIS OUTPUT IS WRITTEN ON A DISK FILE AND CAN BE HANDLED BY  
THE STANDARD CSMP COMPILER.

AN EXAMPLE OF A SOURCE PROGRAM:

```
C'2,5'=C'1,4'*S'1,4'/P'1,4'  
PARAM S'1,4'=1,2,3,4  
PARAM P'1,4'=3*0.5,1.5  
Y'1,4'=INTGRL(YI'1,4',RY'1,4'*SIN(P'1,4'))  
INCON YI'1,4'=4*0.1  
RY'1,4'=RGR*Y'1,4'*(1.-RED'1,4')  
RED'1,4'=AFGEN(RDTB'1,4',ALC)  
FUNCTION RDTB1=(0.,0.),(1.5,2.)  
FUNCTION RDTB2=(0.,0.),(3.,6.)  
FUNCTION RDTB3=(0.,0.),(0.75,0.5),(2.,2.5)  
FUNCTION RDTB4=(0.,0.),(3.,8.)  
ABCD'1,4'=ABCD('1,4')  
SOM =SUM1(Y'1,4')  
SOMKW=SUM2(Y'1,4')  
INPRO=SUMX(P'1,4',S'1,4')  
PRINT Y'1,4',SOMKW,ABCD'1,4'  
OUTPUT ABCD'1,4'  
END  
STOP  
ENDJOB
```

## THE RESULTING PROGRAM

```
C2=C1*S1/P1  
C3=C2*S2/P2  
C4=C3*S3/P3  
C5=C4*S4/P4
```

```

PARAM S1=1,S2=2,S3=3,S4=4
PARAM P1=0.5,P2=0.5,P3=0.5,P4=1.5
Y1=INTGRL(YI1,RY1*SIN(P1))
Y2=INTGRL(YI2,RY2*SIN(P2))
Y3=INTGRL(YI3,RY3*SIN(P3))
Y4=INTGRL(YI4,RY4*SIN(P4))
INCON YI1=0.1,YI2=0.1,YI3=0.1,YI4=0.1
RY1=RGR*Y1*(1,-RED1)
RY2=RGR*Y2*(1,-RED2)
RY3=RGR*Y3*(1,-RED3)
RY4=RGR*Y4*(1,-RED4)
RED1=AFGEN(RDTB1,ALC)
RED2=AFGEN(RDTB2,ALC)
RED3=AFGEN(RDTB3,ALC)
RED4=AFGEN(RDTB4,ALC)
FUNCTION RDTB1=(0.,0.),(1.5,2.)
FUNCTION RDTB2=(0.,0.),(3.,6.)
FUNCTION RDTB3=(0.,0.),(0.75,0.5),(2.,2.5)
FUNCTION RDTB4=(0.,0.),(3.,8.)
ABCD1=ABCD(1)
ABCD2=ABCD(2)
ABCD3=ABCD(3)
ABCD4=ABCD(4)
      SOM=Y1+Y2+Y3+Y4
      SOMKW=Y1**2+Y2**2+Y3**2+Y4**2
      INPRO=P1*S1+P2*S2+P3*S3+P4*S4
PRINT Y1,Y2,Y3,Y4,SOMKW,ABCD1,ABCD2,ABCD3,ABCD4
OUTPUT ABCD1,ABCD2,ABCD3,ABCD4
END
STOP
ENDJOB

```

THE FOLLOWING RESTRICTIONS MUST BE TAKEN CARE OF:

- A) THE RESULTING INDEXED VARIABLES MAY NOT CONTAIN MORE THAN 6 CHARACTERS
- B) THE STATEMENTS THAT CONTAIN APOSTROPHES CANNOT BE CONTINUED BY ...
- C) THE "OUTPUT" STATEMENT MAY NOT CONTAIN ADDITIONAL ARGUMENTS SPECIFYING THE RANGE
- D) SUM1,SUM2 AND SUMX MAY NOT BE NESTED OR BE PART OF AN EXPRESSION

# Appendix B

```

TITLE DEMONSTRATION OF THE USE OF SUBROUTINE BOXCAR
STORAGE Y(100)
FIXED N1,N2, INDEX,I
INCON ITOT1=100.,ITOT2=0.
DYNAMIC
    INDEX=0
PARAM RM1=0.,RM2=0.,LONG1=5.,LONG2=3.,SD1=1.,SD2=0.1,N1=10,N2=10
    OUT1,TOT1=BOXCAR(ITOT1,LONG1,SD1,RM1,IN1,N1)
    OUT2,TOT2=BOXCAR(ITOT2,LONG2,SD2,RM2,IN2,N2).
    IN1=0.
    IN2=OUT1
TIMER FINTIM=15., DELT=0.01, PRDEL=1.
PRINT TOT1,TOT2
METHOD RECT
END
PARAM SD1=0.,SD2=0.
END

STOP
    SUBROUTINE BOXCAR(TOTALI,RT,SD,RM,RIN,N,
        & OUT,TOTAL)
COMMON
C    INITIALIZATION
C
C    IF(TIME.GT.0.) GO TO 1
C    DEVELOPMENT STAGE
C    Y(INDEX+1)=0.5
C
C    PRECLASS
C    Y(INDEX+2)=TOTALI
C
C    N CLASSES
C    DO 2 II=1,N
C    2 Y(II+INDEX+2)=0.
C
C    1 INDEX=INDEX+1
C    PUSH =1.
C    TEST FOR DEVELOPMENT STAGE
C    IF (Y(INDEX).LT.1.) PUSH=0.
C    F    =1,-N*(SD/RT)**2
C    IF(F.GT,N*DELT/RT) GO TO 5
C    WRITE(6,800)
C    800 FORMAT(' NUMBER OF CLASSES TOO LARGE; F TOO SMALL OR NEGATIVE')
C    CALL EXIT
C    5 CONTINUE
C    INTEGRATION OF RATE OF DEVELOPMENT
C    Y(INDEX)=Y(INDEX)+N*DELT/(RT*F)-PUSH
C
C    INDEX=INDEX+1
C    TOTAL=Y(INDEX)
C    FL =2.*TOTAL*(1,-RM*DELT)*N/(RT*F)
C    INTEGRATION OF PRECLASS
C    Y(INDEX)=TOTAL+(RIN-TOTAL*RM-FL)*DELT
C    IF (Y(INDEX).GE.0.) GO TO 3
C    FL =FL+Y(INDEX)/DELT

```

```

      Y(INDEX)=0.
3  PUSH =PUSH*(1./DELT-RM)*F
      DO 4 II=1,N
      INDEX=INDEX+1
      OUT  =Y(INDEX)*PUSH
      TOTAL=TOTAL+Y(INDEX)
      FLN  =FL-Y(INDEX)*RM-OUT
      IF (ABS(FLN).LT.1.E-35) GO TO 4
C   INTEGRATION OF CLASS
      Y(INDEX)=Y(INDEX)+FLN*DELT
4   FL =OUT
      RETURN
      END
ENDJOB

```

#### SOME NOTES FOR THE USE OF SUBROUTINE BOXCAR

- A) EACH BOXCARTRAIN CONTAINS N CLASSES PLUS A PRECLASS. THE DEVELOPMENT STAGE OF THE BOXCARTRAIN IS RECORDED IN AN ADDITIONAL ARRAY ELEMENT. THEREFORE N+2 STORAGE PLACES FOR Y ARE REQUIRED FOR EACH BOXCAR TRAIN OF N CLASSES.
- B) THE STATEMENT "INDEX=0" MUST BE PUT AT THE BEGINNING OF THE DYNAMIC SEGMENT.
- C) THE INFLOW "INI" OF THE FIRST BOXCARTRAIN CAN USUALLY BE IDENTIFIED WITH THE BIRTH RATE. THE INFLOWS OF THE FOLLOWING BOXCARTRAINS EQUAL THE OUTFLOWS OF THEIR PREDECESSORS.
- D) IN COMBINATION WITH SUBROUTINE BOXCAR ONLY METHOD RECT MAY BE USED.
- E) THE OUTPUT ARGUMENTS OF THE SUBROUTINE ARE;
- |       |   |  |
|-------|---|--|
| "OUT" | = | THE RATE OF OUTFLOW OF THE BOXCARTRAIN |
| "TOT" | = | THE TOTAL CONTENT OF THE BOXCARTRAIN   |
- THE INPUT ARGUMENTS OF THE SUBROUTINE ARE;
- |        |   |   |
|--------|---|---|
| "ITOT" | = | THE INITIAL CONTENT OF THE BOXCARTRAIN  |
| "LONG" | = | THE AVERAGE RESIDENCE TIME IN THE WHOLE BOXCARTRAIN,<br>NOT INCLUDING MORTALITY |
| "SD"   | = | THE STANDARD DEVIATION OF THE RESIDENCE TIME IN THE<br>WHOLE BOXCARTRAIN        |
| "RM"   | = | THE RELATIVE RATE OF MORTALITY  |
| "IN"   | = | THE RATE OF INFLOW INTO THE BOXCARTRAIN   |
| "N"    | = | THE NUMBER OF CLASSES WITHIN THE BOXCARTRAIN                                    |
- F) THE VARIABLE NAMES "F", "II", "PUSH", "FL", "FLN", "RT", "SD", "RM", "RIN", "N", "OUT", "TOTAL", "INDEX" SHOULD NOT BE USED IN THE MAIN PROGRAM BECAUSE THEY ARE USED IN THE SUBROUTINE.

# Index

- Acarina 115  
AFGEN 13  
age-classes 70  
age 67  
AINT 93  
Amblyseius 115  
AMIN1 52  
AMOD 93  
analogous 3
- binomial distribution 64  
biocides 115  
BOXCAR 118, 171  
boxcar train 80  
breakthrough curve 83
- cohort 67  
competition 34, 123  
control, biological 115  
controlled dispersion 84  
CSMP 7, 11  
CSSL 7
- diffusion 87  
digital 4  
discontinuous 20, 52  
dispersal 102  
dispersion 66, 78  
DEBUG 109  
DELTA 9  
demographic 70  
density of sowing 35  
DERIV 40, 142  
deterministic simulation 60  
development 66, 68  
dominant eigenvector 77  
driving variables 4  
DYNAMIC 19, 47
- eigenvalue 77  
eigenvector 77  
END 10  
ENDJOB 10  
ENDMAC 45  
environment of system 2  
EPIMAY 91  
error of integration 18  
estimation of parameters 54  
evaluation 111  
exponential growth 8
- FCNSW 93  
fecundity 122  
feedback 17  
FINISH 28  
FINTIM 10  
forcing variables 4  
Forrester's conventions 23  
FORTRAN 7, 11  
fudging 110  
FUNCTION 13  
functional response 123  
fungus disease 91
- GAUSS 62  
germination 78, 106  
germination period 81  
GT (greater than) 62  
gut content 122
- historical information 67  
Helminthosporium 91  
hybrid 4
- IF 62  
IMPULS 53  
INCON 10  
incubation 94

INDEX 43  
INITIAL 19, 47  
initialization 72, 108  
integration method 18  
INSW 98  
INTGRL 10  
  
KEEP 19  
  
length of variable name 143  
lesions 94  
life cycle 115  
logistic growth 29  
Lotka-Volterra equations 30  
LT (less than) 62  
  
MACRO 43  
matrix 76  
METHOD 10  
microclimate 92  
mimicking 85  
mite 115  
mixed culture 22, 31  
mixed prey predation 126  
model 2  
monoculture 22, 31  
mutual dependence 5  
  
NOSORT 19  
  
OUTDEL 10  
OUTPUT 10  
output variables 4  
oviposition 120  
  
Panonychus 115  
Paramecium 49  
PARAMETER 10  
pest management 59  
poisson distribution 59  
potential growth rate 36  
PRDEL 88  
PRINT 40  
Predator 115  
preference 123  
preprocessor 43, 169  
prey 115  
probability functions 61  
PROCEDURAL 61  
PROCEDURE 106  
  
rate variables 4  
RECT 10  
relational diagram 23  
relative crowding  
    coefficient 33  
relative dispersion 80  
relative reproductive  
    rate 32  
relative space 39  
relative yield (total) 32  
replacement series 31, 123  
residence time 29, 82  
RKS 19  
RKSFX 20  
RNDGEN 62  
Runge-Kutta 19  
  
Saccharomyces 22  
sampling 52  
Schizosaccharomyces 22  
Simpson 19  
simulation 3  
simulation language 6  
SIN 14  
sensitivity analysis 110  
SORT 19  
sorting routine 6  
state determined 4  
state variables 4  
stochastic modelling 59  
stoichiometric relation 136  
STOP 10  
structure of system 2  
SUBROUTINE 47  
SUM1 43  
SUMX 71  
system 2  
systems ecology 1, 4  
  
tabulated function 13  
temperature sum 67  
threshold temperature 68  
TIME 14

Time constant 16  
time-interval of  
integration 17  
TIMER 10  
TITLE 10  
TRAPZ 18  
  
units 37, 51, 156  
UPDATE 11  
  
verification 126  
  
waste product 22  
  
ZZ - - - variable 45

This book introduces the reader stepwise to one of the main aspects of systems ecology: analyses and simulation of state-determined systems. This is done by treating in detail various ecological systems, ranging from simple exponential growth, then a plant epidemic, and prey and predatory mite interaction.

Ecological, mathematical and programming aspects are interwoven; exercises form an integral part of the text for use on a second reading. The simulations are presented in one of the most used continuous-system simulation languages (CSMP) in such a way that the programs can be understood without any prior knowledge of programming techniques.

The mathematics is presented at a level that can be followed by biologists without a mathematical background and the ecology is treated in such a way that mathematicians should not feel lost.